

로보스타 로봇  
N1 시리즈  
프로그래밍 설명서



- 취급 설명서
- 조작 및 운용 설명서
- 프로그래밍 설명서
- 유니호스트 설명서
- GAIN 설정
- 알람코드 설명서

**Robostar**

[www.robostar.co.kr](http://www.robostar.co.kr)

ROBOSTAR ROBOT  
N1 Series  
PROGRAMMING MANUAL



- INSTRUCTION MANUAL
- OPERATION MANUAL
- PROGRAMMING MANUAL
- UNI-HOST MANUAL
- GAIN SETUP MANUAL
- ALARM CODE MANUAL

**Robostar**

[www.robostar.co.kr](http://www.robostar.co.kr)

---

Copyright © ROBOSTAR Co., Ltd 2012

이 사용 설명서의 저작권은 주식회사 로보스타에 있습니다.

어떠한 부분도 로보스타의 허락 없이 다른 형식이나 수단으로 사용할 수 없습니다.

사양은 예고 없이 변경 될 수 있습니다.

---

## 제품 보증에 관하여

(주) 로보스타의 제품은 엄격한 품질 관리로 제조되고 있으며, 로보스타의 전 제품의 보증 기간은 제조일로부터 1년간입니다. 이 기간 내에 로보스타 측의 과실로 인한 기계의 고장 또는 정상적인 사용 중의 설계 및 제조상의 문제로 발생하는 고장에 한해서만, 무상으로 서비스를 합니다.

다음과 같은 경우에는 무상 서비스가 불가능합니다.

- (1) 보증 기간이 만료된 이후
- (2) 귀사 또는 제 3 자의 지시에 따른 부적당한 수리, 개조, 이동, 기타 취급 부주의로 인한 고장
- (3) 부품 및 그리스 등 당사의 지정품 이외의 것의 사용으로 인한 고장
- (4) 화재, 재해, 지진, 풍수해 기타 천재지변에 의한 사고로 발생하는 고장
- (5) 분료 및 침수 등 당사의 제품 사양 외의 환경에서 사용함으로 인한 고장
- (6) 소모 부품의 소모로 인한 고장
- (7) 사용설명서 및 취급 설명서에 기재된 보수 점검 작업 내용대로 실시하지 않음으로 인해 발생하는 고장
- (8) 로봇 수리에 드는 비용 이외의 손해

### (주) 로보스타 주소 및 연락처

- 본사 및 공장

경기도 안산시 상록구 사사동 119-38  
119-38, Sasa-dong, Sangnok-gu,  
Ansan-City, Gyeonggi-do, Republic of  
South Korea (426-220)

- 제 2공장

경기도 수원시 권선구 고색동 945  
960, Gosaek-dong, Gwonseon-gu,  
Suwon-City, Gyeonggi-do, Republic of  
South Korea (441-813)

- 서비스요청 및 제품문의

- 영업문의  
TEL. 031-400-3600  
FAX. 031-419-4249  
- 고객문의  
TEL. 1588-4428



[www.robostar.co.kr](http://www.robostar.co.kr)

# 사용 설명서의 구성

본 제품에 관한 사용 설명서는 다음과 같이 구성되어 있습니다. 본 제품을 처음 사용하는 경우 모든 설명서를 충분히 숙지하신 후 사용하시기 바랍니다.

## ■ 취급 설명서

제어기의 전반적인 내용에 대하여 설명합니다. 제어기의 개요, 설치 및 외부 기기와의 인터페이스 방법에 대해 설명합니다.

## ■ 조작 및 운용 설명서

제어기 사용의 전반적인 사용방법과 함께, 파라미터 설정, JOB 프로그램 편집, 로봇 구동 등에 대하여 설명합니다.

## ■ 프로그래밍 설명서

로보스타 로봇 프로그램인 RRL (Robostar Robot Language)에 대하여, 그리고 RRL에 의한 로봇 프로그램 작성 방법에 대하여 설명합니다.

## ■ 유니호스트 설명서

로보스타의 온라인 PC 프로그램인 '유니호스트'에 대하여 설명합니다.

## ■ GAIN 설정 설명서

시운전시 필요한 게인 설정 방법과 게인 값 변경에 따른 응답성에 대하여 설명합니다.

## ■ 알람코드 설명서

제어기 운용 중 발생 할 수 있는 알람 상황에 대하여, 발생 원인 및 조치 사항에 대하여 설명합니다.

# 목차

- 제1장 개요 .....1-2
  - 1.1 개요 .....1-2
- 제2장 명령어 일람.....2-1
  - 2.1 명령어 화면 표시 .....2-1
  - 2.2 명령어 일람표.....2-2
    - 2.2.1 프로그램 제어 관련 명령어.....2-2
    - 2.2.2 로봇 동작 관련 명령어.....2-3
    - 2.2.3 입출력 관련 명령어.....2-4
    - 2.2.4 로봇 동작 조건 관련 명령어.....2-5
    - 2.2.5 변수 선언.....2-6
    - 2.2.6 프로그램 제어 관련 명령어.....2-6
    - 2.2.7 시스템 변수.....2-7
    - 2.2.8 상수.....2-7
    - 2.2.9 연산자.....2-8
    - 2.2.10 문자열.....2-9
- 제3장 명령어 해설.....3-1
  - 3.1 MAIN, EOP (프로그램 시작/종료).....3-1
    - 3.1.1 프로그램 사용 예제.....3-1
  - 3.2 FOR, NEXT (반복 수행문) .....3-2
    - 3.2.1 프로그램 사용 예제.....3-2
  - 3.3 WHILE, ENDWL (조건 반복수행 명령문).....3-4
    - 3.3.1 프로그램 사용 예제.....3-4
    - 3.3.2 조건식 사용 예제 .....3-5
  - 3.4 IF, ENDIF (조건 분기 명령어) .....3-6
    - 3.4.1 프로그램 사용 예제.....3-7
    - 3.4.2 IF(조건식) THEN.....3-8
  - 3.5 LABEL, GOTO (분기 명령어) .....3-9
    - 3.5.1 프로그램 사용 예제.....3-9
  - 3.6 SUBR, RET (부실행문 명령어).....3-11
    - 3.6.1 프로그램 사용 예제.....3-11
  - 3.7 CALL, JCALL (호출 명령어) .....3-12

3.7.1	프로그램 사용 예제.....	3-13
3.8	STOP, EXIT (로봇 및 JOB 정지 명령어).....	3-14
3.8.1	프로그램 사용 예제.....	3-14
3.9	JMOV (PTP 이동 명령어).....	3-15
3.9.1	프로그램 사용 예제.....	3-15
3.10	LMOV (CP 이동 명령어).....	3-17
3.10.1	TOOL 방향을 일정 하계 유지 하면서 티칭 (SCARA 로봇).....	3-18
3.10.2	프로그램 사용 예제.....	3-19
3.11	CMOV (원형 보간 이동 명령어).....	3-20
3.11.1	프로그램 사용 예제.....	3-20
3.12	AMOV (원호 보간 이동 명령어).....	3-21
3.12.1	프로그램 사용 예제.....	3-21
3.13	IMOV, IMOV2 (증분 이동 명령어).....	3-22
3.13.1	프로그램 사용 예제.....	3-23
3.14	JNTSYN (PTP 동기 모드 명령어).....	3-24
3.14.1	프로그램 사용 예제.....	3-25
3.15	EECH (말단 장치 선택 명령어).....	3-26
3.15.1	프로그램 사용 예제.....	3-27
3.16	TIMOV (TOOL 좌표계 기준 증분 이동 명령어).....	3-28
3.17	HMOV (원점 이동 명령어).....	3-29
3.17.1	프로그램 사용 예제.....	3-29
3.18	PMOV (PALLETIZING 이동).....	3-30
3.18.1	TOOL 방향을 일정 하계 유지 하면서 티칭 (SCARA 로봇).....	3-31
3.18.2	Pallet 작업 예.....	3-32
3.18.3	프로그램 사용 예제 (1).....	3-35
3.18.4	프로그램 사용 예제 (2).....	3-37
3.18.5	프로그램 사용 예제 (3).....	3-40
3.19	PASS.....	3-42
3.19.1	해 설.....	3-42
3.19.2	프로그램 사용 예제 (1).....	3-42
3.19.3	프로그램 사용 예제 (2).....	3-43
3.20	WITH, ENDWT (동시처리 명령어).....	3-44
3.20.1	프로그램 사용 예제 (1).....	3-44
3.21	OUT, POUT (외부 출력 명령어).....	3-46
3.21.1	프로그램 사용 예제.....	3-48
3.22	IN, PIN (외부 입력 명령어).....	3-49
3.22.1	프로그램 사용 예제.....	3-51

3.23 CIN,CBIN,CWIN,CDIN,CFIN(필드 버스용 입력 명령어) .....3-52  
 3.23.1 프로그램 사용 예제.....3-53

3.24 COUT,CBOUT,CWOUT,CDOUT,CFOUT(필드 버스용 출력 명령어).....3-54  
 3.24.1 프로그램 사용 예제.....3-55

3.25 VEL(축 이동 속도 설정 명령어).....3-56  
 3.25.1 프로그램 사용 예제.....3-56

3.26 ACC,DEC(가감속 설정 명령어).....3-57  
 3.26.1 프로그램 사용 예제.....3-58

3.27 FOS,PFOS,(연속제적생성 명령어).....3-59  
 3.27.1 프로그램 사용 예제.....3-60

3.28 SFOS(연속제적 및 등속 생성 명령어).....3-62  
 3.28.1 연속 모션 속도 프로파일 패턴.....3-64  
 3.28.2 프로그램 사용 예제.....3-65

3.29 SVON,SVOF(서보 ON/OFF 명령어).....3-66  
 3.29.1 프로그램 사용 예제.....3-66

3.30 DLAY(시간지연 명령어).....3-68  
 3.30.1 프로그램 사용 예제.....3-68

3.31 OFFS(오프셋 지정 명령어) .....3-69  
 3.31.1 프로그램 사용 예제.....3-69

3.32 LIMIT(축 제한 명령어).....3-70  
 3.32.1 프로그램 사용 예제.....3-70

3.33 PLUP(PULL UP 동작 설정 명령어).....3-71  
 3.33.1 프로그램 사용 예제.....3-72

3.34 TOOL(축 제한 명령어) .....3-73  
 3.34.1 프로그램 사용 예제.....3-74

3.35 FIX(축 제한 명령어).....3-75  
 3.35.1 프로그램 사용 예제.....3-76

3.36 FORM(로봇 ARM 형상 지정 명령어) .....3-77  
 3.36.1 프로그램 사용 예제.....3-77

3.37 TRQ(충돌 감지 명령어).....3-78  
 3.37.1 프로그램 사용 예제.....3-78

3.38 TQL(축 출력 토크 제한 명령어).....3-79  
 3.38.1 프로그램 사용 예제.....3-79

3.39 INPOS(목표점 도달 지정 명령어).....3-80  
 3.39.1 프로그램 사용 예제.....3-80

3.40 MINIT(MAPPING 초기화 명령어).....3-81

3.41 MSTART(MAPPING 검출 명령어).....3-82




- 3.41.1 맵핑 센서 입력 포트 관련..... 3-83
- 3.42 MREAD(MAPPING 위치 데이터 갱신 명령어)..... 3-84
  - 3.42.1 프로그램 사용 예제..... 3-84
- 3.43 변수..... 3-86
  - 3.43.1 변수의 종류..... 3-86
  - 3.43.2 사용 방법..... 3-87
- 3.44 정수형(INT, I, II), 실수형 (REAL, F) 변수..... 3-88
  - 3.44.1 LOCAL 변수..... 3-88
  - 3.44.2 프로그램 사용 예제..... 3-88
  - 3.44.3 정수형 GLOBAL 변수 I, II..... 3-89
  - 3.44.4 프로그램 사용 예제..... 3-89
  - 3.44.5 실수형 GLOBAL 변수 F..... 3-90
  - 3.44.6 프로그램 사용 예제..... 3-90
- 3.45 POSITION 변수..... 3-91
  - 3.45.1 POS 변수..... 3-91
  - 3.45.2 POINT 변수..... 3-92
  - 3.45.3 프로그램 사용 예제..... 3-92
- 3.46 시스템 변수 (CNT, TMR, MVR, HERE)..... 3-93
  - 3.46.1 CNT, TMR 변수..... 3-93
  - 3.46.2 프로그램 사용 예제..... 3-93
  - 3.46.3 MVR 변수..... 3-94
  - 3.46.4 프로그램 사용 예제..... 3-95
  - 3.46.5 HERE 변수..... 3-96
  - 3.46.6 프로그램 사용 예제..... 3-96
- 3.47 RSTATE(로봇 상태 확인 명령어)..... 3-97
  - 3.47.1 프로그램 사용 예제..... 3-98
- 3.48 RERROR(알람 코드 확인 명령어)..... 3-99
  - 3.48.1 프로그램 사용 예제..... 3-99
- 3.49 GFTOGP(GF를 GP에 저장하는 명령어)..... 3-100
  - 3.49.1 프로그램 사용 예제..... 3-100
- 3.50 REMCMD (SYSTEM COMMAND)..... 3-101
  - 3.50.1 프로그램 사용 예제..... 3-101
- 3.51 상수..... 3-102
  - 3.51.1 프로그램 사용 예제..... 3-102
- 3.52 연산자..... 3-103
  - 3.52.1 배정 연산자..... 3-103
  - 3.52.2 산술 연산자..... 3-104

3.52.3	관계 연산자.....	3-105
3.52.4	논리 연산자.....	3-105
3.52.5	비트 연산자.....	3-105
3.53	내장 함수.....	3-106
3.54	ASC.....	3-107
3.54.1	프로그램 사용 예제.....	3-107
3.55	CHR.....	3-108
3.55.1	프로그램 사용 예제.....	3-108
3.56	FLUSH.....	3-109
3.56.1	프로그램 사용 예제.....	3-109
3.57	FTOS.....	3-110
3.57.1	프로그램 사용 예제.....	3-110
3.58	HTOS.....	3-111
3.58.1	프로그램 사용 예제.....	3-111
3.59	SLEFT.....	3-112
3.59.1	프로그램 사용 예제.....	3-112
3.60	SLEN.....	3-113
3.60.1	프로그램 사용 예제.....	3-113
3.61	SMID.....	3-114
3.61.1	프로그램 사용 예제.....	3-114
3.62	SPOS.....	3-115
3.62.1	프로그램 사용 예제.....	3-115
3.63	SRIGHT.....	3-116
3.63.1	프로그램 사용 예제.....	3-116
3.64	STRIN.....	3-117
3.64.1	프로그램 사용 예제.....	3-117
3.65	STROUT.....	3-118
3.65.1	프로그램 사용 예제.....	3-118
3.66	SVAL.....	3-119
3.66.1	프로그램 사용 예제.....	3-119
3.1	VCMD.....	3-120
3.1.1	프로그램 사용 예제.....	3-120
3.2	VVAL.....	3-121
3.2.1	프로그램 사용 예제.....	3-121
3.3	VDAT.....	3-122
3.3.1	프로그램 사용 예제.....	3-122
3.4	VCNT.....	3-123

3.4.1 프로그램 사용 예제.....3-123

● N1 제어기 버전

N1 제어기는 로봇 타입에 따라 2가지 운용 소프트웨어로 구분되며, 각각의 운용 소프트웨어는 N1 제어기 전원 ON시 Teach Pendant 화면에서 확인 가능합니다.



스카라 로봇, 직각 로봇, DeskTop, 동기로봇  
운용소프트웨어(RO)



반도체 이송용 로봇 운용소프트웨어 (TR)

● 제어기 버전에 따라 변경사항

	RO(Robot)	TR(Transfer Robot)
Global INT	500EA(0~499)	2000EA(0~1999)
Global FLOAT	500EA(0~499)	2000EA(0~1999)
Local Point	2000EA(0~1999)	2000EA(0~1999)
Global Point	1024EA(0~1023)	15000EA(0~14999)

본 매뉴얼에서 설명하고 있는 내용은 RO(Robot) 버전 기준으로 설명 하고 있으며 TR(Transfer Robot) 버전 사용시 변경되는 부분은 위 표를 참고하시기 바랍니다.



**주의**

- ▶ 로봇 구동전 반드시 운용소프트웨어와 로봇 타입을 확인 하시기 바랍니다.
- ▶ 사용하는 로봇과 N1제어기의 운용 소프트웨어가 일치되어야 정상적인 운용이 가능 합니다

## 제1장 개요

### 1.1 개요

1. 로봇언어는 Robostar에서 만든 전용 언어로 시스템 내에서 로봇 작업 프로그램을 작성할 때 사용 하는 명령어 입니다.
2. 로봇 언어는 크게 4개의 그룹으로 분류됩니다.
  - FLOW 그룹  
프로그램의 조건 또는 무조건 분기, 반복 횟수, 부 프로그램 호출 등을 설정하는 명령어로 구성되어 있습니다.
  - MOVE 그룹  
로봇의 동작 조건(PTP, 보간, Palletizing 등)을 설정하는 명령어로 구성되어 있습니다.
  - I/O 그룹  
외부 입, 출력을 제어(16bit, 1bit)하는 명령어로 구성되어 있습니다.
  - COND(=Condition) 그룹  
속도, 지연시간 등을 설정하는 명령어로 구성되어 있습니다.

작업 용도에 맞는 적절한 명령어를 사용하여 원하는 작업 프로그램을 작성 하십시오.

## 제2장 명령어 일람

### 2.1 명령어 화면 표시



F1 F2 F3 F4

그룹	선택키	FLOW	MOVE	I/O	COND
선택키		F1	F2	F3	F4
항목 이동	F1	MAIN	JMOV	OUT	VEL
	F2	FOR	LMOV	POUT	FOS
	F3	IF	CMOV	IN	DLAY
	F4	WHILE	AMOV	PIN	INT
	F1	EOP	IMOV	CIN	ACC
	F2	NEXT	PMOV	CWIN	PFOS
	F3	ELSE	HMOV	COUT	PLUP
	F4	ENDWL		CWOUT	REAL
	F1	SUBR	SVON	SYS	FORM
	F2	RET	SVOF		TOOL
	F3	CALL	WITH		PCLR
	F4	ENDIF	ENDWT		POS
	F1	LABL	MVR		LIMIT
	F2	GOTO	HERE		OFFS
	F3	JCALL	PASS		CNT
	F4		PCNT		INPOS
	F1	STOP			TMR
	F2	EXIT			NO
	F3				LEFT
	F4				RIGHT
	F1				FIX
	F2				TRQ
	F3				TQL
	F4				XCHG
	F1				DEC
	F2				
	F3				
	F4				

## 2.2 명령어 일람표

### 2.2.1 프로그램 제어 관련 명령어

명령어	기능	형식	사용예
MAIN	프로그램 시작	MAIN : EOP	MAIN VEL 200
EOP	프로그램 종료		JMOV P1 JMOV P2 EOP
FOR	스텝반복수행	FOR <변수>=<초기값> TO <종료값> [BY<증분량>] : NEXT	: FOR A=1 TO 5 BY 2
NEXT		* 변수는 정수형 변수만 사용	VEL 200 JMOV P1 NEXT :
WHILE	조건반복수행	WHILE<조건식> : ENDWL	: WHILE IN0==1
ENDWL		JMOV P1 JMOV P2 ENDWL :	
IF	조건판단수행	IF<조건식> THEN : (ELSE) : ENDIF	: IF IN3==1 THEN
ENDIF		GOTO A0 ELSE JMOV P1 ENDIF :	
LABL	분기점 위치 지정	LABL <레이블명> :	: LABL A1
GOTO	수행문 분기	GOTO <레이블명>	JMOV P1 JMOV P2 GOTO A1 :
SUBR	부실행문 지정	SUBR<부실행문명> :	: SUBR HON
RET	부실행문 복귀	RET	GOTO A0 ELSE JMOV P1 RET :
CALL	부실행문 호출	CALL <부실행문명>	
JCALL	JOB 파일 호출	JCALL<JOB 파일명>	
STOP	로봇 동작 정지	STOP	
EXIT	JOB 수행 정지	EXIT	

2.2.2 로봇 동작 관련 명령어

명령어	기능	형식	사용예
JMOV	현 위치에서 목표점으로 PTP이동	JMOV <포인트 변수>	MAIN VEL 200 JMOV P1
LMOV	현 위치에서 목표점으로 직선보간 이동	LMOV <포인트 변수>	LMOV P2 EOP
CMOV	현위치에서 경유점1, 2를 잇는 원을 그리며 이동	CMOV <경유점1> <경유점2>	: VEL 200 CMOV P1 P2
AMOV	현위치에서 경유점1과 목표점을 잇는 원호를 그리며 이동	AMOV <경유점1> <목표점>	JMOV P3 AMOV P4 P5 :
IMOV	현 위치에서 증분량으로 PTP 이동	IMOV <포인트변수> <포인트변수> 값 → 축값 증분량	: JMOV P1 IMOV P2
IMOV2	현 위치에서 증분량으로 직선보간 이동	IMOV <포인트변수> <포인트변수> 값 → 축값 증분량	IMOV2 P3 :
PMOV	현 위치에서 지정된 Palletizing 작업 수행	PMOV <작업 pallet 번호> <작업 기준점> <작업 pallet 번호>의 데이터 값 → 파라미터에 설정	PMOV P1 P10
HMOV	현위치에서 지정된 홈 위치로 이동	HMOV <HOME 위치 번호> <위치번호>의 데이터값 → ORG 파라미터에 설정	HMOV 1
JNTSYN	PTP 동기 모드 선택	JNTSYN<모드>	JNTSYN 1
EECH	말단 장치 선택	EECH<말단 장치>	EECH 1
TIMOV	TOOL 좌표계 기준으로 증분이동량 만큼 직선 보간 이동	TIMOV<위치형 변수>	TIMOV CURR
SVON	서보 ON	SVON → 모든 축 해당 SVON<지정 축> → 1(X), 2(Y), 3(Z), 4(W), 5(E1), 6(E2)	SVON SVOF 2
SVOF	서보 OFF	SVOF → 모든 축 해당 SVOF<지정 축> → 1(X), 2(Y), 3(Z), 4(W), 5(E1), 6(E2)	SVON 2
WITH	로봇 동작 중 다음 수행문 열을 동시처리	WITH : ENDWT	: WITH JMOV P1 MVR=0 WHILE MVR<60 IF IN1==1 THEN OUT0==1 ENDIF ENDWL ENDWT :
ENDWT			
PASS	pallet 작업 통과 지정	PASS <pallet NO> <통과할 작업물 순번>	PASS 1 3 PASS 2 1



2.2.3 입출력 관련 명령어

명령어	기능	형식	사용 예
<b>OUT</b>	1비트 단위로, 지정된 비트번호의 출력을 ON(=1), OFF(=0)	OUT<출력비트번호>=<0/1> [펄스유효시간] [->]  [펄스유효시간] 단위 : 10ms [펄스유효시간]이 없으면 : 출력신호 계속 유효 [펄스유효시간]이 지나면 이전상태로 복귀 [->] : 펄스유효시간을 주기로 갖는 주기파형 출력	OUT0=1 OUT0=1 100 OUT0=1 100 ->
<b>POUT</b>	포트 단위로, 지정된 포트 로 지정된 값을 출력한다.	POUT<출력포트번호>=<출력지정 값>  <출력포트번호> : 0 ~ 2 중 선택 0 -> OUT0 ~ OUT15 1 -> OUT16 ~ OUT31 2 -> OUT32 ~ OUT47	POUT0=0 POUT0=20 POUT1=0H000F
<b>IN</b>	1비트 단위로, 지정된 비트 번호의 ON(=1), OFF(=0) 값을 읽는다.	IN<입력비트번호>=<0/1> -지정된 비트 입력 조건이 만족 될 때까지 대기 <변수>=IN<입력비트 번호> -지정된 비트 입력 상태 값을 변수에 저장	INT AA IN10=1 AA=IN0
<b>PIN</b>	포트 단위로, 지정된 입력 포트의 값을 읽는다.	<변수>=PIN<입력포트 번호>  <입력포트번호> : 0 ~ 2 중 선택 0 -> IN0 ~ IN15 1 -> IN16 ~ IN31 2 -> IN32 ~ IN47	INT AA AA=PIN0
<b>CIN</b>	1비트 단위로, 지정된 필드 버스 비트 입력 번호의 ON(=1), OFF(=0) 값을 읽 는다	<변수>=CIN<입력비트번호> -지정된 비트 입력 상태 값을 변수에 저장	INT AA AA=CIN0
<b>CBIN</b> <b>CWIN</b> <b>CDIN</b>	포트 단위로, 지정된 필드 버스 입력 포트의 값을 읽는다.	<변수>=CWIN<입력포트 번호>	INT AA AA=CWIN0
<b>COUT</b>	1비트 단위로, 지정된 필드 버스 출력을 ON(=1), OFF(=0)	COUT<출력비트 번호>=<0/1>	COUT=1
<b>CBOUT</b> <b>CWOUT</b> <b>CDOUT</b>	포트 단위로, 지정된 필드 버스 입력 포트의 값을 읽는다.	CWOUT<출력포트번호>=<출력지정 값>	CWOUT=0HFF0FFFF
<b>SYS</b>	시스템 제어 명령	예약된 시스템 명령어 수행	

2.2.4 로봇 동작 조건 관련 명령어

명령어	기능	형식	사용 예
VEL	축 이동 속도의 백분율(%)을 설정	VEL <백분율 값(%)> 축 이동속도=정격속도 X 0.001  <b>정격속도는 MOTION 파라미터에서 설정</b>	MAIN VEL 200 ACC 70 DEC 70
ACC	가속시간의 백분율(%)을 설정	ACC <백분율 값(%)> 가속시간 = 정격가속시간 X 백분율 값 X 0.01  <b>정격가속시간은 MOTION 파라미터에서 설정</b>	JMOV P1 VEL 1000 LMOV P2 ACC 100
DEC	감속시간의 백분율(%)을 설정	DEC <백분율 값(%)> 감속시간 = 정격감속시간 X 백분율 값 X 0.01  <b>정격감속시간은 MOTION 파라미터에서 설정</b>	DEC 100 EOP
FOS	축 선단이 목표점에 도달하기 전에 다음 목표점으로 궤적변경 함.	FOS <거리비율(%)> 거리비율은 전체 이동거리의 백분율(%) JMOV, LMOV, AMOV에 적용	FOS 5 JMOV P1 FOS 0(해제)
PFOS	축 선단이 목표점에 도달하기 전에 다음 목표점으로 궤적변경 함.	PFOS <거리(mm)> LMOV에 적용	PFOS 5 LMOV P1 PFOS 0(해제)
SPOS	축 선단이 목표점에 도달하기 전에 다음 목표점으로 궤적변경 함	SFOS<거리>,<속도레벨>,<원호 각도> LMOV, AMOV, CMOV에 적용	SFOS 10 0 0 LMOV P0 LMOV P1 SFOS 0 0 0(해제)
DLAY	지연시간 설정	DLAY <지연시간> 지연시간 단위는 10ms (지연시간이 500이면 5초 지연됨)	JMOV P1 DLAY 20 JMOV P2
OFFS	목표점을 지정된 값만큼 이동시킴.	OFFS <편차 값>  <b>편차 값은 MDI모드로 입력</b> (예) P100 → X:0, Y:-100, Z:10, W:0, E1:20, E2:30	JMOV P1 OFFS P100 JMOV P2
LIMIT	각 축의 이동범위를 제한한다.	LIMIT <축 위치 값(최소)> <축 위치 값(최대)>	JMOV P1 LIMIT P3 P4 JMOV P2
PLUP	PULL UP 동작을 위한 Z축 위치값 설정	PLUP <Z축 위치값> Z축에만 적용 됨	PLUP 5 JMOV P1 PLUP 0(해제)
FORM	로봇 동작시 로봇 암의 형상을 지정 함.	FORM <형상>  수평 다관절 로봇에만 적용 됨. <형상> LEFT, RIGHT, NO 중 선택	FORM RIGHT JMOV P1 FORM NO(해제)
TOOL	로봇에 부착된 사용 툴을 선택	TOOL <툴번호>  <b>툴번호에 해당하는 데이터는 파라미터에 설정</b> <툴번호> : 0 ~ 3중 선택	TOOL 1 JMOV P1 TOOL 0 JMOV P2

명령어	기능	형식	사용 예
<b>PCLR</b>	PMOV의 Palletizing 카운터를 초기화	PCLR <pallet 번호> Pallet 번호 : 0~99까지 사용 가능 클리어 시 카운트(XCNT, YCNT,ZCNT)는 1로 변경됨	MAIN PCLR 1 PCLR 2 PMOV P1 P10
<b>INPOS</b>	로봇 동작 시 목표점 도달 정도를 설정	INPOS <목표점 도달 정도> 도달 정도는 Pulse 값으로 설정	INPOS 10 JMOV P2 INPOS 0(해제) JMOV P3
<b>FIX</b>	보간 동작에서 W축 동작선택	FIX (0/1) 0 : W축 자세 회전, 1 : W축 자세 고정 생략시 초기 값은 1로 고정 임.	FIX 0 LOMV P1 FIX 1
<b>TRQ</b>	충돌 감지시 토크 리미트 알람 발생	TRQ <축번호> <토크제한 값> 각축의 토크제한 값 이상일 때 알람 발생	TRQ 1 50 LMOV P1 LMOV P2
<b>TQL</b>	각축의 최대 추력 토크 값 제한	TQL <축번호> <토크제한 값> 각축을 제한 토크로 구동	TQL 1 50 LMOV P1 LMOV P2
<b>MINIT</b>	맵핑 기능 초기화	MINIT<센서 타입>	MINIT 1
<b>MSTART</b>	맵핑 기능 시작	MSTART<센서 입력핀><축><GP Index>	MSTART 3 3 GP10
<b>MREAD</b>	맵핑 데이터 갱신	MREAD<제한 시간>	MREAD 500

### 2.2.5 변수 선언

명령어	기능	형식	사용 예
<b>INT</b>	정수형 변수를 선언	INT <변수명>, REAL <변수명>, POS <변수명>, DEFSTR <변수명>  변수명은 6자 이내 변수명은 대문자 알파벳과 숫자로 구성 변수명은 숫자로 시작할 수 없다.	MAIN INT N REAL A,B POS MM, XA DEFSTR PACKET
<b>REAL</b>	실수형 변수를 선언		
<b>POS</b>	위치형 변수를 선언		
<b>DEFSTR</b>	문자열 변수를 선언		

### 2.2.6 프로그램 제어 관련 명령어

명령어	기능	형식	사용 예
	포인트형 변수는 포인트 파일작성 시 정의됨	변수명은 POINT 명령 참조	

### 2.2.7 시스템 변수

명령어	기능	형식	사용 예
<b>CNTn</b> <b>TMR0</b> <b>TMR1</b>	시스템정의변수 (카운터변수,타이머 변수)	CNT<펄스입력포트번호>=<초기값> TMR0=<초기값> ▶ 카운터변수는 펄스입력포트 번호를 입력하는 순간부터 값이 할당되고 그 후 매 펄스입력을 카운트 한다. ▶ 타이머 변수는 정수 값을 입력하는 순간부터 값이 할당되어 시스템 파라미터에 정의된 시간 간격으로 1씩 증가한다.	CNT0=2 TMR0=0 TMR1=-50
<b>MVR</b>	MOVE RATE	MVR <로봇이동구간의 백분율>	IF MVR<50 THEN
<b>HERE</b>	현재축 각도 값 변수	AP=HERE 현재축 각도 값을 저장하는 변수	AP=HERE
<b>RSTATE</b>	로봇 상태정보 읽기	<정수형 변수>=RSTATE(채널, STATE INDEX)	TEMP=RSTATE(1,6)
<b>RERROR</b>	알람 코드 확인	<정수형 변수>=RERROR(PAGE,INDEX)	ERR=RERROR(1,1)
<b>REMCMD</b>	시스템 명령어	REMCMD<채널><명령어>	REMCMD 1 3
<b>GFTOGP</b>	GF값을 GP로 저장	GFTOGP<Global Float Index> <Global Point Index>	GFTOGP F10 GP50

### 2.2.8 상수

	기능	형식	사용 예
	정수, 실수, 2진 정수, 16진 정수를 표시	[0H/0B]<숫자>  숫자 앞에 0H 또는 0B가 없으면 10진수 0H : 16진수, 0B : 2진수	

## 2.2.9 연산자

명령어	기능	형식	사용 예
$+$ , $-$ , $*$ , $/$ , $\%$	이항연산자 (가,감,승,제,modulus)		$A = B * C$
( )	우선 연산자		$A = B / (A + C)$
=	대입연산자		$A = B$
$\&$ , $ $ , $\sim$ , $\wedge$ , $\ll$ , $\gg$	비트연산자 (BAND, BOR, Complement, BXOR, left shift, right shift)		
$\&\&$ , $\ \ $ , $\wedge\wedge$ , $!$	논리연산자 (AND, OR, XOR, Negation)		
$>$ , $<$ , $>=$ , $<=$ , $!=$ , $==$	비교연산자(초과, 미만, 이상, 이하, 상이, 상동)		
ABS	절대값을 취함	ABS(-10.5)는 10.5	
DEG	라디안 값을 각도 값으로 변환	DEG(3.1416)는 180.0	
RAD	각도 값을 라디안 값으로 변환	RAD(180.0)는 3.1416	
POW	지수함수	POW(2,4)는 16	
RND	소수부를 반올림하여 실수값을 정수 값으로 변환	RND(14.8)은 15	
LOG	상용지수함수		
SQRT	평방제곱근을 구함	SQRT(16)은 4	
SIN	사인 함수	SIN(RAD(30))는 0.5	
ASIN	아크 사인 함수	ASIN(0.5)는 0.4794	
COS	코사인 함수	COS(0)은 1.0	
ACOS	아크 코사인 함수	ACOS(0.5)은 1.0472	
TAN	탄젠트함수	TAN(RAD(45))는 1.0	
ATAN	아크탄젠트 함수	ATAN(-1.0)은 -0.7854	
ATAN2	제2 아크탄젠트 함수	ATAN2(Y,X) ATAN2(1,-1)은 2.3562	

2.2.10 문자열

명령어	기능	형식	사용 예
ASC	문자열 첫문자를 캐릭터 코드로 반환	정수형 변수=ASC(문자열)	INT AA AA=ASC("ABC")
CHR	정수를 문자로 변환	문자열 변수=CHR(정수)	DEFSTR AA AA=CHR(65)
FLUSH	입력, 출력 버퍼 클리어	FLUSH<클리어 버퍼 선택> ▶ 선택 범위: 1~3 1: 입력 버퍼 클리어 2: 출력 버퍼 클리어 3: 입력, 출력 버퍼 클리어	FLUSH3
FTOS	정수나 실수를 문자열로 변환	문자열 변수=FTOS(정수or실수) ▶ 정수 1234를 문자열 "1234"로 변환	DEFSTR AA AA=FTOS(1234)
HTOS	정수를 16진수의 문자열로 변환	문자열 변수=HTOS(정수) ▶ 정수 10을 16진수 문자열 "A"로 변환	DEFSTR AA AA=HTOS(10)
SLEFT	좌측부분 문자열 추출	문자열 변수=SLEFT(문자열, 숫자) ▶ 문자열의 왼쪽부터 숫자만큼 문자를 추출해 문자열 변수에 저장	DEFSTR AA AA=SLEFT("ABCDE",3)
SLEN	문자열 길이를 반환	정수형 변수=SLEN(문자열)	INT LEN LEN=SLEN("ABCDE",3)
SMID	문자열 지정위치부터 자릿수 만큼 문자열 추출	문자열 변수=SMID(문자열, 지정위치, 추출문자수) ▶ 문자열 위치는 맨 처음이 0부터 시작함	DEFSTR AA AA=SMID("ABCDEF",2,3)
SPOS	문자열1 내에서 문자열2가 들어가 있는 위치 반환	정수형 변수= SPOS(문자열1,문자열2)	INT PP PP=SPOS("ABCDEF","B") ▶ "B"의 위치 1반환
SRIGHT	우측부분 문자열 추출	문자열 변수=SRIGHT(문자열, 숫자) ▶ 문자열의 오른쪽부터 숫자만큼 문자를 추출해 문자열 변수에 저장	DEFSTR AA AA=SRIGHT("ABCDE",3)
STRIN	구분자까지 문자열 입력	문자열 변수=STRIN(타임아웃시간) ▶ 타임아웃시간은 ms단위임	DEFSTR AA AA=STRIN(1000)
STROUT	문자열 출력	정수형 변수= STROUT(문자열) ▶ 정수형변수에 전송하지 못한 문자수 저장	INT RT RT=STROUT("ABCDEF")
SVAL	문자열을 숫자로 변환	변수=SVAL(문자열)	INT VA REAL VB VA=SVAL("1234") VB=SVAL("0.123")

## 제3장 명령어 해설

### 3.1 MAIN, EOP (프로그램 시작/종료)

기능 프로그램 시작 및 종료를 나타냅니다.

형식 MAIN  
...  
EOP

- 설명
- 1) **MAIN-EOP는 반드시 같이 사용하여** 주 프로그램 블록을 형성하며 한 개의 JOB 파일에서 한 개의 블록만을 작성 할 수 있습니다.
  - 2) EOP(End Of Program)는 주 프로그램 블록의 끝을 의미하며 JOB 수행 시 이 명령을 만나면 **JOB 수행이 종료**됩니다.  
단, JCALL에 의해 수행된 JOB은 그 JOB의 수행을 종료하고 **호출한 JOB으로 복귀**합니다.
  - 3) **EOP 다음 라인에 필요한 부 프로그램 블록(SUBR ~ RET)을 계속 작성**할 수 있습니다.

#### 3.1.1 프로그램 사용 예제

<b>MAIN</b>	.....	<b>주 Program 시작</b>
VEL 100	.....	축 이동 속도
WHILE 1	.....	조건반복수행 시작
JMOV P0	.....	P0로 이동
JMOV P1	.....	P1으로 이동
CALL ON	.....	부 프로그램 호출
ENDWL	.....	조건반복수행 종료
<b>EOP</b>	.....	<b>주 Program 종료</b>
SUBR ON	.....	부 프로그램 지정
OUT0=1 200	.....	1 비트 단위로, 지정된 비트에 출력
OUT1=1 200	.....	1 비트 단위로, 지정된 비트에 출력
OUT2=1 200	.....	1 비트 단위로, 지정된 비트에 출력
RET	.....	부 프로그램 복귀

### 3.2 FOR, NEXT (반복 수행문)

기능 변수 값이 만족할 때 까지 블록을 반복 수행합니다

형식 FOR <변수>=<초기값> TO <종료값> (BY <증분량>)  
 ...  
 NEXT

용어 <변수> : 정수형으로 선언된 변수이름을 사용합니다. 예) INT A,B,C,AA,...  
 <초기값> : FOR 블록을 반복 수행하기 직전 변수에 설정되는 정수값입니다.  
 <종료값> : FOR 블록의 반복 수행 횟수를 제한하기 위한 정수값입니다.  
 <증분값> : 변수의 값을 규칙적으로 일정하게 증가시키기 위해 사용되는 정수값입니다

- 설명
- 1) **FOR - NEXT 는 반드시 같이 사용**되어야 합니다.
  - 2) 정수형 변수에 초기값이 설정된 후 **종료값을 만족할때까지 FOR 블록을 반복 수행**합니다.
  - 3) FOR 블록을 반복 수행할때마다 정수형 변수의 값은 증분값에 의해 증가되고, 증분값이 생략되었을 경우에는 **자동적으로 1씩 증가**합니다.
  - 4) 종료값은 초기값보다 **항상 큰 값**이어야 하며 증분값은 **1이상의 값**이어야 합니다.  
 (초기값≤종료값, 증분값≥1)

#### 3.2.1 프로그램 사용 예제

- 1) FOR 블록을 이용하여, 1부터 10까지 홀수의 합을 연산하여 정수형 변수 SUM에 저장.

MAIN		
INT AA, SUM	.....	정수변수 AA, SUM 선언
SUM=0	.....	정수변수 초기화
<b>FOR AA=1 TO 10 BY 2</b>	.....	AA=1부터 +2씩 증가 정수 10까지 5회 수행
SUM=SUM+AA	.....	5회 수행 후 NEXT 다음 이동
<b>NEXT</b>		
EOP		



2) FOR 블록을 이용하여, 포인트 P1 ↔ P2 간의 20회 왕복운동

MAIN		
INT J	.....	정수변수 J선언
J=0	.....	정수변수 초기화
<b>FOR J=1 TO 20</b>	.....	FOR문 20회 반복동작 후 NEXT 다음 이동
JMOV P1	.....	JMOV P1
JMOV P2	.....	JMOV P2
<b>NEXT</b>		
EOP		

3) Position 변수에 티칭포인트를 저장한 후 이동

MAIN		
INT A	.....	정수변수 A선언
POS AP(11)	.....	위치변수 AP(11)선언 → AP(0),AP(1) ~ AP(10)
<b>FOR A=1 TO 10</b>	.....	FOR문 10회 반복동작 후 NEXT 다음 이동
AP(A)=P(A)	.....	포인트 변수 AP(A)에 P(1) ~ P(10) 저장
<b>NEXT</b>		
EOP		

4) Position 변수에 티칭포인트를 저장한 후 이동

MAIN		
INT J,K	.....	정수변수 J,K선언
POS TMP,AP(11)	.....	위치변수 AP(11)선언 → AP(0),AP(1) ~ AP(10)
<b>FOR K=1 TO 10</b>	.....	FOR문 10회 반복동작 후 NEXT 다음 이동
AP(K)=P(K)	.....	포인트 변수 AP(K)에 P(1) ~ P(10) 저장
<b>NEXT</b>		
IF P100.3==1.0 THEN		
<b>FOR J=1 TO 10</b>	.....	FOR문 10회 반복동작 후 NEXT 다음 이동
TMP=AP(J)	.....	AP(J)=AP(I)=P(1) ~ P(10)
TMP3=TMP+10		위치변수 TMP의 Z축 값에 10을 더함.
AP(J)=TMP		
<b>NEXT</b>		
ENDIF		
<b>FOR K=1 TO 10</b>	.....	FOR문 10회 반복동작 후 NEXT 다음 이동
JMOV AP(K)		저장된 AP(1) ~ AP(10)으로 PTP 이동
DLAY 100		
<b>NEXT</b>		
EOP		

### 3.3 WHILE, ENDWL (조건 반복수행 명령문)

기능 조건식이 만족 되는 동안만 블록을 반복 수행합니다

형식 WHILE <조건식>  
...  
ENDWL

용어 <조건식> : 수식의 진위를 판단할 수 있는 논리연산식이나 비교연산식을 말합니다.

- 설명
- 1) WHILE 문은 조건이 만족(조건식의 결과가 참, 또는 0이외의 값)되는 동안 WHILE 블록을 반복 수행합니다.
  - 2) WHILE - ENDWL은 반드시 같이 사용되어야 합니다.
  - 3) 조건식의 결과가 항상 참인 경우에는 무한 반복 수행을 하게 됩니다.

#### CAUTION

**명령어 입력시 주의 사항**

- (O) WHILE\_(IN0==1)&&(IN1==0))
- (O) WHILE\_IN0==1&&IN1==0
- (O) WHILE\_IN0==1  
&&IN1==0

#### 3.3.1 프로그램 사용 예제

- 1) 포인트 P0 ↔ P1을 반복이동

MAIN		
VEL 100		
<b>WHILE 1</b>	.....	블록 무한 반복 실행
JMOV P0	.....	
JMOV P1	.....	P0 ↔ P1 반복이동
<b>ENDWL</b>		
EOP		

3.3.2 조건식 사용 예제

WHILE(조건식)		설 명
상 수	WHILE 1	WHILE ~ ENDWL 블록을 무한 반복 실행
입 력	WHILE IN0==1	입력신호 IN0=1로 입력되는 동안만 블록 반복 실행
	WHILE ((IN0==1)&&(IN1==0))	입력신호 IN0=1, IN1=0 두 가지 조건이 동시에 만족하는 동안만 블록을 반복실행
	WHILE ((IN0==1)  (IN1==0))	입력신호 IN0=1, IN1=0 두 가지 조건 중 한가지만 만족하는 동안만 블록을 반복실행
	WHILE PIN0==0H000F	입력PORT 0번의 IN0, IN1, IN2, IN3이 모두 "1"인 동안만 블록을 반복 실행
	WHILE PIN0==32	입력 PORT 0번의 입력신호상태가 IN4가 "1"인 동안만 블록을 반복 실행
출 력	WHILE OUT0==1	출력 OUT0이 "1"인 동안만 블록을 반복실행
	WHILE POUT0==0H000F	출력 POUT0번의 OUT0, OUT1, OUT2, OUT3이 모두 " 1"인 동안만 블록을 반복실행
	WHILE POUT0==0B0000000011111111	출력 POUT0번의 OUT0, OUT1, OUT2, OUT3, OUT4, OUT5, OUT6, OUT7이 모두 "1"인 동안만 블록을 반복실행
변 수	WHILE TMP==1	변수 TMP가 "1"인 동안만 블록을 반복실행
	WHILE MVR>10	MVR변수 값이 "10"보다 큰 동안만 블록반복 실행
	WHILE TMR>10	TMR(TIMER)변수값이 "10"보다 큰 동안만 블록반복 실행
	WHILE CNT>10	CNT(COUNT)변수 값이 "10"보다 큰 동안만 블록반복 실행
	WHILE AP10.1==1.0	POSITION 변수 AP값의 1축 값이 "1.0"과 같은 동안만 블록을 반복 실행

### 3.4 IF, ENDIF (조건 분기 명령어)

기능 조건식의 조건 판단을 수행합니다.

형식 IF <조건식> THEN  
...  
(ELSEIF) <조건식> THEN  
...  
(ELSE)  
...  
ENDIF

용어 <조건식> : 수식의 진위를 판단할 수 있는 논리연산식이나 비교연산식을 말합니다.

- 설명
- 1) **IF - ENDIF 는 반드시 같이** 사용하여야 합니다.
  - 2) 조건식의 연산 결과가 **참(또는 0이외의 값)이면 THEN 이하의 문장들이 실행되고, 거짓(또는 0)이면 ELSE 다음의 문장들이 실행됩니다.**
  - 3) ELSE문은 프로그램 작성자가 **선택적으로 사용할 수** 있습니다.
  - 4) IF 블록 내부에서 다시 IF 블록을 사용하는 중첩 IF 블록을 사용할 수 있습니다.  
이때 주의할 것은 IF와 ENDIF가 논리적으로 쌍을 이룰수 있도록 작성해야 합니다.
- ※ IF의 개수와 ENDIF의 개수가 다를경우 "Syntax Error"가 발생 됩니다.  
※ ELSE 또는 ELSEIF 개수는 최대 10개로 제한 됩니다.

#### CAUTION

##### 명령어 입력시 주의 사항

- (O) IF\_((IN0==1)&&(IN1==0))
- (O) IF\_IN0==1&&IN1==0
- (O) IF\_IN0==1  
&&IN1==0

### 3.4.1 프로그램 사용 예제

- 1) 입력 IN0의 상태에 따라 이동할 포인트 결정

MAIN		
VEL 100		
<b>IF IN0==1 THEN</b>		
JMOV P1	.....	입력 IN0 가 1인 경우 포인트 P1로 PTP 이동
<b>ELSE</b>		
JMOV P2	.....	입력 IN0 가 0인 경우 포인트 P2로 PTP 이동
<b>ENDIF</b>		
EOP		

- 2) 입력 Port "1"(IN0 ~ 15), IN0의 상태 검사

MAIN		
INT AA, BB		
AA=PIN1	.....	Port1(IN0~15) 상태를 변수 AA에 저장
<b>IF (AA==0HFFFF) THEN</b>		AA값 판단 (IN0~15 모두 1인지 여부)
STOP	.....	"참" 일 경우, 로봇 스톱
<b>ELSE</b>		
BB=IN0	.....	"거짓" 일 경우, IN0의 상태를 BB에 저장
<b>IF (BB==1) THEN</b>		BB가 1인지 판단
DLAY 10	.....	
JMOV P0	.....	"참(IN0=1)" 일 경우, 포인트 P0로 이동
<b>ENDIF</b>		
<b>ENDIF</b>		
EOP		

- 3) 아래 프로그램은 동일한 내용이다.

(단, IN0~IN3중 동시에 가해지는 입력이 하나뿐이란 가정하에 동일 내용)

<pre> : LABL A0 IF IN0==1 THEN GOTO B0 ENDIF IF IN1==1 THEN GOTO C0 ENDIF IF IN2==1 THEN GOTO D0 ENDIF IF IN3==1 THEN GOTO A0 ENDIF :                 </pre>	=	<pre> : LABL A0 IF IN0==1 THEN GOTO B0 ELSE IF IN1==1 THEN GOTO C0 ELSE IF IN2==1 THEN GOTO D0 ELSE IF IN3==1 THEN GOTO A0 ENDIF ENDIF ENDIF ENDIF :                 </pre>
--	---	---

3.4.2 IF(조건식) THEN

IF(조건식)		설 명
입 력	IF_IN0==1_THEN	입력 IN0이 "1"인가?
	IF_((IN0==1)&&(IN1==0))_THEN	입력신호 IN0=1, IN1=0 두 가지 조건이 동시에 만족하는가?
	IF_((IN0==1) (IN1==0))_THEN	입력신호 IN0=1, IN1=0 두 가지 조건 중 최소 한가지 만족하는가?
	IF_PIN0==0H000F_THEN	입력PORT0번의 IN0, IN1, IN2, IN3이 모두 "1"인가?
	IF_PIN0==32_THEN	입력 PORT0번의 입력신호상태가 IN4가 "1"인가?
출 력	IF_OUT0==1_THEN	출력 OUT0이 "1"인가?
	IF_POOUT0==0H000F_THEN	출력 POUT0번의 OUT0, OUT1, OUT2, OUT3이 모두 "1"인가?
	IF_POOUT0==0B0000000011111111_THEN	출력 POUT0번의 OUT0, OUT1, OUT2, OUT3, OUT4, OUT5, OUT6, OUT7이 모두 "1"인가?
변 수	IF_TMP==1_THEN	변수 TMP가 "1"인가?
	IF_MVR>10_THEN	MVR변수 값이 "10"보다 큰가?
	IF_TMR>10_THEN	TMR(TIMER)변수 값이 "10"보다 큰가?
	IF_CNT>10_THEN	CNT(COUNT)변수 값이 "10"보다 큰가?
	IF_AP10.1==1_THEN	POSITION 변수 값(1축 값) "1"과 같은가?

### 3.5 LABL, GOTO (분기 명령어)

기능 분기점 위치 지정, 수행문 분기 이동 명령어 입니다.

형식 LABL <레이블명>  
GOTO <레이블명>

용어 <레이블명> : 영문자와 숫자로 구성된 8자리의 문자 스트링입니다.

**단, 작업 포인트를 표시하는 "P0, P1, ..."는 사용할 수 없습니다.**


예) LABL UNCLAMP1 (O), GOTO CHECK2 (O)

LABL P1 (X), GOTO P100 (X)

- 설명
- 1) GOTO명령은 **조건 없이 해당 레이블로 분기**하며 동일 블록 내에서만 가능합니다.  
예를 들면, MAIN 블록 내에서 SUBR블록으로, SUBR블록 내에서 다른 SUBR블록으로 분기할 수 없습니다.
  - 2) IF블록이나 FOR, WHILE블록을 중첩하여 사용할 때, 내부에서 외부로의 분기는 가능하지만, **반대의 경우는 불가능**합니다.

#### 3.5.1 프로그램 사용 예제

- 1) 두 포인트 P1, P0를 반복 이동



2) 입력 IN0의 상태에 따라 분기

MAIN		
VEL 100		
JMOV P0		
IF IN0==1 THEN		
<b>GOTO A0</b>	.....	입력 IN0이 1이면 LABL A0로 무조건 점프
ELSE		
<b>GOTO A1</b>	.....	아니면 LABL A1로 무조건 점프
ENDIF		
JMOV P1		
<b>LABL A0</b>	.....	LABL A0 분기지점
JMOV P2		
<b>GOTO A2</b>	.....	LABL A2로 무조건 점프
<b>LABL A1</b>	.....	LABL A1 분기지점
JMOV P3		
<b>LABL A2</b>	.....	LABL A2 분기지점
EOP		

**! CAUTION**

**명령어 입력시 주의 사항**

```

MAIN
VEL 100
JMOV P0
IF IN0==1 THEN
GOTO A0
ELSE
GOTO A1
ENDIF
JMOV P1
LABL A0
JMOV P2
GOTO A2
LABL A1
JMOV P3
CALL D0
LABL A2
EOP
SUBR D0
JMOV P5
GOTO A0
RET
    
```



### 3.6 SUBR, RET (부실행문 명령어)

- 기능 프로그램 중에서 특정의 동작을 반복하는 부분을 따로 작성하여, 필요에 따라 호출이 가능
- 형식 SUBR <부실행문명>  
...  
RET
- 용어 <부실행문명> : 영문자와 숫자로 구성된 8자리의 문자 스트링입니다. 단, 작업 포인트를 표시하는 "P0, P1, ..."는 사용할 수 없습니다.
- 설명
- 1) **SUBR - RET 는 반드시 같이** 사용하여야 합니다.
  - 2) SUBR 블록은 **CALL문에 의해서만 호출**될 수 있습니다.
  - 3) RET문은 SUBR블록의 실행을 종료하고, **CALL 다음 STEP으로 복귀**하도록 합니다.
  - 4) CALL문은 MAIN이나 SUBR블록에서 모두 사용할 수 있으며 재귀호출(Recursive Call)도 가능합니다.
  - 5) SUBR 블록은 반드시 **MAIN-EOP 블록 다음에 정의**하여야 합니다.

#### 3.6.1 프로그램 사용 예제

MAIN		
VEL 100		
JMOV P0		
<b>CALL A0</b>	.....	부실행문 "A0"을 CALL
<b>CALL A1</b>	.....	부실행문 "A1"을 CALL
EOP		
<b>SUBR A0</b>		
JMOV P1		
<b>RET</b>	.....	SUBR A0 블록의 실행을 종료하고, CALL 다음 스텝으로 복귀
<b>SUBR A1</b>		
JMOV P2		
<b>RET</b>	.....	SUBR A1 블록의 실행을 종료하고, CALL 다음 스텝으로 복귀

### 3.7 CALL, JCALL (호출 명령어)

기능 부실행문, JOB 파일을 호출합니다.

형식 CALL <부실행문명>  
JCALL <JOB파일명>

용어 <부실행문명> : 영문자와 숫자로 구성된 8자리의 문자 스트링입니다.  
단, 작업 포인트를 표시하는 "P0, P1, ..."는 사용할 수 없습니다.  
<JOB파일명> : 호출할 JOB 프로그램의 확장자(JOB)를 제외한 파일명입니다.

- 설명
- ※ **CALL**
    - 1) CALL문을 사용하려면 반드시 호출될 부실행문(SUBR)이 존재해야 합니다.
    - 2) CALL문의 부실행문명과 부실행문이 동일해야 합니다.
    - 3) SUBR블록에서 RET를 만나면 CALL 다음 STEP으로 복귀합니다.
  - ※ **JCALL**
    - 1) JCALL은 JOB을 서브루틴으로 사용하는 것입니다.
    - 2) 호출된 JOB에서 EOP를 만나면 호출한 프로그램의 JCALL문 다음으로 복귀합니다.
    - 3) JCALL은 재귀호출이 되지 않도록 프로그램을 작성해야 합니다.

재귀호출(Recursive Call)이 가능합니다.

주) CALL문 재귀 호출 **최대 8회** → 초과하면 "CALL DEPTH ERROR"  
JCALL문 재귀 호출 **최대 3회** → 초과하면 "JCALL DEPTH ERROR"

#### CAUTION

명령어 입력시 주의 사항


```

MAIN
VEL 100
LABL A0
JMOV P1
JMOV P2
CALL A0
GOTO A0
EOP
SUBR A0
OUT1=1
IN1=1
RET
    
```

(X) LABL명과 SUBR명이 같으면 ERROR 발생 함.

3.7.1 프로그램 사용 예제

1) 입력포트 0번의 상태에 따라 JOB 파일명을 "TEST1"~"TEST4"중 하나를 JCALL 함.



### 3.8 STOP, EXIT (로봇 및 JOB 정지 명령어)

- 기능     로봇 동작 정지(STOP), JOB 수행정지(EXIT)
- 형식     STOP  
          EXIT
- 설명     1) STOP 명령어는 이동중인 로봇을 정지시키고, 다음 STEP의 명령어를 처리합니다.  
          즉, 정지후 JOB을 계속 수행합니다.  
          2) EXIT 명령어는 JOB 수행을 중지시킵니다. 이때 제어기는 **“EXIT Instruction”**  
          알람을 발생 시킵니다.

#### 3.8.1 프로그램 사용 예제

- 1) 포인트이동 후 출력 OUT1을 ON, OFF

```

MAIN
VEL 500
JMOV P1
MVR=0 ..... “이동구간 백분율”을 초기화
WITH .....
JMOV P2 ..... 포인트 P2로 이동(JMOV만 가능함)
WHILE MVR<60 ..... 이동구간의 60%미만 동안에
IF IN0==1 THEN ..... IN0=1이면 “LABL BB”로 점프
GOTO BB
ENDIF
ENDWL
OUT0=1
ENDWT

LABL BB
STOP ..... 로봇이동 동작 정지
ENDWT ..... WITH문 종료 (반드시 삽입)
JMOV P3
OUT1=1
EOP
    
```


### 3.9 JMOV (PTP 이동 명령어)

- 기능    현위치에서 목표점으로 축보간 이동
- 형식    JMOV P<번호>  
          JMOV GP<번호>  
          JMOV <위치형변수>
- 용어    <번호> : 티칭한 위치 좌표 번호를 설정합니다.  
          P : JOB 별로 개별 사용 되는 LOCAL POINT  
          GP : 공통으로 사용 되는 GLOBAL POINT

	RO(Robot)	TR(Transfer Robot)
Local Point	0 ≤ 번호 ≤ 1999	0 ≤ 번호 ≤ 1999
Global Point	0 ≤ 번호 ≤ 1023	0 ≤ 번호 ≤ 1499

<위치형 변수> : POS형으로 선언된 변수이름을 의미합니다.

- 설명    PTP(Point to Point)란 **점에서 점으로의 이동**을 의미합니다.  
          이동하는 경로는 로봇의 자세에 의존하며 **직선동작에 한정되지는 않습니다.**



#### 3.9.1 프로그램 사용 예제

- 1) 포인트 티칭한 위치좌표로 이동

```

MAIN
VEL 100
JMOV P10                      .....            포인트 P10 으로 PTP 이동
JMOV P11                      .....            포인트 P11 으로 PTP 이동
JMOV P100                    .....            포인트 P100 으로 PTP 이동
EOP
    
```

2) 위치형 변수를 이용한 이동 (1)

MAIN		
POS XA	.....	포인트형 변수 XA 선언
XA=<400.0,50.0,10.4,10.0,0,0,1>	.....	XA의 속성값을 지정 <X,Y,Z,W,E1,E2,FORM>
VEL 100		
<b>JMOV P0</b>	.....	포인트 P10 으로 PTP 이동
<b>JMOV XA</b>	.....	XA 위치 좌표로 PTP 이동
<b>JMOV P100</b>	.....	포인트 P100 으로 PTP 이동
EOP		

3) 위치형 변수를 이용한 이동(2)

MAIN		
POS AP,AP1,AP2	.....	포인트형 변수 AP,AP1,AP2를 선언
AP=P10 + <10,10,10,10,0,0>	.....	포인트 P10의 각축값에 10을 더한 값을 저장
AP1=P11 - <10,10,10,10,0,0>	.....	포인트 P11의 각축값에 10을 뺀 값을 저장
AP2=P100	.....	포인트 AP2에 P100의 값을 저장
AP2.3=P100.3+10	.....	포인트 P100의 Z축값에 10을 더한 값을 저장
<b>JMOV AP</b>	.....	AP 위치 좌표로 PTP 이동
<b>JMOV AP1</b>	.....	AP1 위치 좌표로 PTP 이동
<b>JMOV AP2</b>	.....	AP2 위치 좌표로 PTP 이동
EOP		

### 3.10 LMOV (CP 이동 명령어)

- 기능     현위치에서 목표점으로 직선보간 이동
- 형식     LMOV P<번호>  
           LMOV GP<번호>  
           LMOV <위치형 변수>
- 용어     <번호> : 티칭한 위치 좌표 번호를 설정합니다.  
           P : JOB 별로 개별 사용 되는 LOCAL POINT  
           GP : 공통으로 사용 되는 GLOBAL POINT

VER.	RO(Robot)	TR(Transfer Robot)
Local Point	0 ≤ 번호 ≤ 1999	0 ≤ 번호 ≤ 1999
Global Point	0 ≤ 번호 ≤ 1023	0 ≤ 번호 ≤ 1499

<위치형 변수> : POS형으로 선언된 변수이름을 의미합니다.

설명     CP제어는 동작목표 위치에 도달하는 경로를 **직선이 되도록 보간제어**합니다.



#### ⚠ CAUTION

- ▶ CP제어를 실시하면 로봇의 위치에 따라 동작할 수 없는 경우가 있습니다. 이때 제어기는 "Unreachable Point" 또는 "Inverse Error" 등의 Alarm이 발생 됩니다.
- ▶ 특히 "FOS", "FIX", "TOOL" 명령어 사용 시 주의 하시기 바랍니다.
- ▶ SCARA Robot은 현재 FORM을 유지하여 보간 동작함
- ▶ 프로그램의 처음 동작명령에는 PTP 제어를 사용해야 합니다.

#### ⚠ CAUTION

- SCARA Robot에서 W축을 보간 동작 할 때
- ▶ 파라미터의 "OFFS", "LENG를 기계부의 Sticker에 명시된 값으로 정확하게 설정.
  - ▶ TOOL OFFSET이 있을 때 W축 보간 동작에는 두가지 방법이 있습니다.  
 W축 자세고정 (로봇명령어 "FIX 1") W축 자세 회전 (로봇 명령어 "FIX 0") (FIX 명령어 참조)
  - ▶ TOOL OFFSET이 있는 경우 작업 포인트 티칭에 주의 해야 합니다.  
 W축 자세 고정 : TOOL 방향을 일정한 방향으로 유지 하면서 티칭  
 W축 자세 회전 : TOOL 오프셋을 정확하게 설정 후 TOOL 선정 및 작업 포인트티칭을 하십시오.
  - ▶ 가급적 W축 자세를 고정하여 작업 할 수 있도록 TOOL 선정 및 작업 포인트 티칭을 하십시오.
  - ▶ 특히 W축 자세 회전은 연속된 보간 동작 중 원호를 그릴 때 (AMOV) 로봇 기계부가 제어기의 지령에 추종 하지 못하는 경우(Unreachable Point)가 발생 할 수 있습니다.

3.10.1 TOOL 방향을 일정 하게 유지 하면서 티칭 (SCARA 로봇)

Step 1.

```
<RBSA804A : EDIT> V: 50
A : 0.13      B : 0.12
Z : 0.11      W : 65

EXCH  CORD  PJUMP  FWRD
```

```
<RBSA804A : EDIT> V: 50
A : 0.13      B : 0.12
Z : 0.11      W : 65

EXCH  CORD PJUMP  FWRD
```

F2

```
<RBSA804A : EDIT> V: 50
X : 595.32    Y : 0.13
Z : 0.11      W : 65

EXCH  CORD  PJUMP  FWRD
```



```
<RBSA804A : EDIT> V: 50
X : 400.38    Y : 101.13
Z : 0.11      W : 65

EXCH  CORD  PJUMP  FWRD
```

티칭 매뉴얼을 참조하여 포인트 화면으로 이동 합니다..

「F2」를 입력 하여 좌표계를 원통 → 직교 로 변경 합니다.

직교좌표계 화면에서 JOG 동작을 실시 하면 W축자세를 유지한 상태로 이동 됩니다.

←	J1	→	X축
	X		
←	J2	→	Y축
J	J	K	
←	J3	→	Z축
O	Z	P	
←	J4	→	W축
T	RX	U	
←	J5	→	EX1축
Y	RY	Z	
←	J6	→	EX2축
	RZ		



### 3.10.2 프로그램 사용 예제

1) 포인트 티칭한 위치좌표로 이동

MAIN		
VEL 100		
<b>JMOV P10</b>	.....	포인트 P10 으로 PTP 이동
<b>LMOV P11</b>	.....	포인트 P11 으로 CP 이동
<b>LMOV P100</b>	.....	포인트 P100 으로 CP 이동
EOP		

2) 위치형 변수를 이용한 이동 (1)

MAIN		
POS XA	.....	포인트형 변수 XA 선언
XA=<400.0,50.0,10.4,10.0,0,0,1>	.....	XA의 속성값을 지정 <X,Y,Z,W,E1,E2,FORM>
VEL 100		
<b>JMOV P0</b>	.....	포인트 P10 으로 PTP 이동
<b>LMOV XA</b>	.....	XA 위치 좌표로 CP 이동
<b>LMOV P100</b>	.....	포인트 P100 으로 CP 이동
EOP		

3) 위치형 변수를 이용한 이동(2)

MAIN		
POS AP,AP1,AP2	.....	포인트형 변수 AP,AP1,AP2를 선언
AP=P10 + <10,10,10,10,0,0>	.....	포인트 P10의 각축값에 10을 더한 값을 저장
AP1=P11 - <10,10,10,10,0,0>	.....	포인트 P11의 각축값에 10을 뺀 값을 저장
AP2=P100	.....	포인트 AP2에 P100의 값을 저장
AP2.3=P100.3+10	.....	포인트 P100의 Z축값에 10을 더한 값을 저장
<b>JMOV AP</b>	.....	AP 위치 좌표로 PTP 이동
<b>LMOV AP1</b>	.....	AP1 위치 좌표로 CP 이동
<b>LMOV AP2</b>	.....	AP2 위치 좌표로 CP 이동
EOP		


### 3.11 CMOV (원형 보간 이동 명령어)

- 기능     현위치에서 경유점1, 2를 잇는 원을 그리며 이동
- 형식     CMOV P<경유점1 번호> P<경유점2 번호>  
 CMOV GP<경유점1 번호> GP<경유점2 번호>  
 CMOV <경유점1 위치형 변수> <경유점2 위치형 변수 >
- 용어     <번호> : 티칭한 위치 좌표 번호를 설정합니다.  
 P : JOB 별로 개별 사용 되는 LOCAL POINT  
 GP : 공통으로 사용 되는 GLOBAL POINT

VER.	RO(Robot)	TR(Transfer Robot)
Local Point	0 ≤ 번호 ≤ 1999	0 ≤ 번호 ≤ 1999
Global Point	0 ≤ 번호 ≤ 1023	0 ≤ 번호 ≤ 1499

<위치형 변수> : POS형으로 선언된 변수이름을 의미합니다.

- 설명     동작 목표위치에 도달하는 경로를 **원을 그리도록 보간 제어**합니다.  
 원형보간 동작을 할때에는 **3개의 포인트 좌표가 필요** 합니다.



#### ! CAUTION

- ▶ 잘못된 Point Teaching으로 Auto RUN을 실행시 동작할 수 없는 경우가 있습니다. 이때 제어기는 "Unreachable Point" 또는 "Inverse Error" 등의 Alarm이 발생 됩니다.
- ▶ 특히 "FOS", "FIX", "TOOL" 명령어 사용 시 주의 하시기 바랍니다.
- ▶ W축의 보간 동작에 대하여는 "LMOV"명령어를 참조 하십시오.
- ▶ CMOV, AMOV는 단독으로 사용 할 수 없으며 예제와 같이 JMOV를 추가 하십시오.

#### 3.11.1 프로그램 사용 예제

##### 1) 포인트 티칭한 위치좌표로 이동

```

MAIN
VEL 100
JMOV P1 ..... 포인트 P1 으로 PTP 이동
CMOV P2 P3 ..... 포인트 P2 P3를 경유하는 원형 보간 이동
EOP
    
```


### 3.12 AMOV (원호 보간 이동 명령어)

- 기능 현위치에서 경유점1, 2를 잇는 원호를 그리며 이동
- 형식 AMOV P<경유점1 번호> P<경유점2 번호>  
 AMOV GP<경유점1 번호> GP<경유점2 번호>  
 AMOV <경유점1 위치형 변수> <경유점2 위치형 변수 >
- 용어 <번호> : 티칭한 위치 좌표 번호를 설정합니다.  
 P : JOB 별로 개별 사용 되는 LOCAL POINT  
 GP : 공통으로 사용 되는 GLOBAL POINT

VER.	RO(Robot)	TR(Transfer Robot)
Local Point	0 ≤ 번호 ≤ 1999	0 ≤ 번호 ≤ 1999
Global Point	0 ≤ 번호 ≤ 1023	0 ≤ 번호 ≤ 1499

<위치형 변수> : POS형으로 선언된 변수이름을 의미합니다.

- 설명 동작 목표위치에 도달하는 경로를 **원호를 그리도록 보간 제어**합니다.  
 원호보간 동작을 할때에는 **3개의 포인트 좌표가 필요** 합니다.



#### ! CAUTION

- ▶ 잘못된 Point Teaching으로 Auto RUN을 실행시 동작할 수 없는 경우가 있습니다. 이때 제어기는 "Unreachable Point" 또는 "Inverse Error" 등의 Alarm이 발생 됩니다.
- ▶ 특히 "FOS", "FIX", "TOOL" 명령어 사용 시 주의 하시기 바랍니다.
- ▶ W축의 보간 동작에 대하여는 "LMOV"명령어를 참조 하십시오.
- ▶ CMOV, AMOV는 단독으로 사용 할 수 없으며 예제와 같이 JMOV를 추가 하십시오.

#### 3.12.1 프로그램 사용 예제

##### 1) 포인트 티칭한 위치좌표로 이동

```

MAIN
VEL 100
JMOV P1 ..... 포인트 P1 으로 PTP 이동
AMOV P2 P3 ..... 포인트 P2 P3를 경유하는 원호 보간 이동
EOP
    
```


### 3.13 IMOV, IMOV2 (증분 이동 명령어)

- 기능 현위치에서 증분량으로 축보간 이동  
 IMOV – PTP 이동  
 IMOV2 – CP 이동
- 형식 IMOV P<번호>  
 IMOV GP<번호>  
 IMOV <위치형변수>
- 용어 <번호> : 티칭한 위치 좌표 번호를 설정합니다.  
 P : JOB 별로 개별 사용 되는 LOCAL POINT  
 GP : 공통으로 사용 되는 GLOBAL POINT

VER.	RO(Robot)	TR(Transfer Robot)
Local Point	0 ≤ 번호 ≤ 1999	0 ≤ 번호 ≤ 1999
Global Point	0 ≤ 번호 ≤ 1023	0 ≤ 번호 ≤ 1499

<위치형 변수> : POS형으로 선언된 변수이름을 의미합니다.


- 설명 1) 지정된 위치만큼 **상대 이동**합니다.  
 2) IMOV는 **현재위치를 기준**으로 하여 지정된 위치만큼 이동하는 것입니다.



#### ! CAUTION

- ▶ IMOV를 위한 Point Teaching에서 위치좌표를 저장할 때 파라미터의 RANG"이내의 값으로 저장하지만 이 값은 원점(ORIGIN)을 기준으로 설정된 위치값입니다.
- ▶ 따라서 이전 Point를 기준으로 로봇이 이동(IMOV)할 때 "Range over error"가 발생할 수 있습니다.
- ▶ 파라미터 "RANG"설정을 조정하십시오.
- ▶ IMOV는 이전 위치를 기준으로 이동하므로 이전위치의 In Position정도에 따라 위치편차가 발생할 수 있습니다.  
 이때 작업프로그램에 "INPOS", "DLAY"등을 추가 하여 로봇이 이전 위치에 정확하게 In Position 완료 후 IMOV가 이루어 지도록 프로그램을 수정하시기 바랍니다.

3.13.1 프로그램 사용 예제



1) IMOV 사용 (포인트 P1 사용)

MAIN		
VEL 100		
JMOV P10	.....	포인트 P10 으로 PTP 이동
JMOV P11	.....	포인트 P11 으로 PTP 이동
<b>INPOS 10</b>	.....	INPOS 또는 DLAY 사용 하여 위치 편차 예방.
<b>IMOV P1</b>	.....	포인트 P11을 기준으로 <b>P1 만큼 PTP 이동</b>
EOP		

2) IMOV 사용 (위치형변수 AP 사용)

MAIN		
POS AP	.....	포인트형 변수 AP 선언
AP=<20.0,-30.0,0,0,0,0,1>	.....	AP의 속성값을 지정 <X,Y,Z,W,E1,E2,FORM>
VEL 100		
JMOV P10	.....	포인트 P10 으로 PTP 이동
<b>IMOV AP</b>	.....	포인트 P10을 기준으로 <b>AP 값 만큼 PTP 이동</b>
JMOV P100	.....	포인트 P100 으로 PTP 이동
EOP		

3) IMOV2 사용

MAIN		
VEL 100		
LMOV P10	.....	포인트 P10 으로 CP 이동
LMOV P11	.....	포인트 P11 으로 CP 이동
<b>INPOS 10</b>	.....	INPOS 또는 DLAY 사용 하여 위치 편차 예방.
<b>IMOV2 P1</b>	.....	포인트 P11을 기준으로 <b>P1 만큼 CP 이동</b>
EOP		

4) IMOV2 사용 (위치형변수 AP 사용)

MAIN		
POS AP	.....	포인트형 변수 AP 선언
AP=<20.0,-30.0,0,0,0,0,1>	.....	AP의 속성값을 지정 <X,Y,Z,W,E1,E2,FORM>
VEL 100		
LMOV P10	.....	포인트 P10 으로 CP 이동
<b>IMOV2 AP</b>	.....	포인트 P10을 기준으로 <b>AP 값 만큼 CP 이동</b>
LMOV P100	.....	포인트 P100 으로 CP 이동
EOP		

### 3.14 JNTSYN (PTP 동기 모드 명령어)

기능 PTP 동기 모드 선택

**JNTSYN 명령어는 ROSEP Robot 및 Ver 03.02.08 이후 버전에서만 사용 가능합니다**

형식 JNTSYN <형식>

0 : PTP 동기 모드 사용 안함

1 : PTP 동기 모드 사용 함

설명 1) 특이점 근처에서 직선보간모션(LMOV)을 할 경우, 특이점에서 급격한 속도 변화로 인해 Over Speed 알람이 발생 합니다.

이러한 문제점을 해결 하기 위해 ROSEP 로봇 타입의 경우 PTP 모션(JMOV)를 동기로 구동하여 특이점을 지나는 직선 모션이 가능 합니다.

2) "JNTSYN 0"으로 설정 되면, 일반적인 JMOV와 동일하게 동작 합니다.

3) "JNTSYN 1"으로 설정 되면 J2, J4, J5의 이동량을 J1의 이동량에 동기시켜 특이점을 통과하는 직선 모션을 합니다. ( J1(A), J2(B), J3(Z), J4(W1), J5(W2) )

4) 축별 이동거리를 비교하여 동기조건이 성립하는지 확인 합니다.

W1, W2가 동기조건을 만족하지 못하면 동기식에서 제외합니다.

A, B가 동기 조건을 만족하지 못하면 알람(1422: PTP Sched. Error)이 발생합니다.


예) P1 = <-25, 230, 0, -115, -115, 0>, P2 = <25, 130, 0, -65, -65, 0>

P2.1 - P1.1 = 50 = delta\_TH1

P2.2 - P1.2 = -100 = -2 \* delta\_TH1

P2.4 - P1.4 = 50 = delta\_TH1

P2.5 - P1.5 = 50 = delta\_TH1



[ROSEP ROBOT TYPE의 특이점 통과]

3.14.1 프로그램 사용 예제

MAIN	.....	Program 시작
VEL 10	.....	축 이동 속도
JMOV P1	.....	P1=<-25, 230, 0, -115, -115, 0>로 이동
WHILE 1	.....	조건반복수행 시작
JNTSYN 1	.....	PTP 동기 모드 사용 함
JMOV P2	.....	P2=<25, 130, 0, -65, -65, 0>로 이동
JMOV P1	.....	P1=<-25, 230, 0, -115, -115, 0>로 이동
JNTSYN 0	.....	PTP 동기 모드 사용 안함
ENDWL	.....	조건반복문 종료
EOP	.....	Program 종료

 CAUTION

- ▶ 특이점 구간에서 직선모션을 하기 위해서 PTP동기 모드로 이동 할 때 J1(A) 이동량에 비해 J2(B) 이동량이 2배가 되어야 합니다.( J1×2 == J2)  
 예) J1(A) 이동량이 50° 이면, J2(B) 이동량은 100° 입니다.  
 ※ 이동량이 2배가 아닌경우 **“PTP Sched. Error ”** 알람이 발생 합니다.
- ▶ J4(W1),J5(W2)의 이동량이 J1(A)이동량과 같지 않으면, PTP동기 모드가 아닌 개별축 이동(PTP) 모션으로 변경 됩니다.


### 3.15 EECH (말단 장치 선택 명령어)

기능 말단 장치(W1, W2) 선택  
**EECH 명령어는 ROSEP Robot 및 Ver 03.02.08 이후 버전에서만 사용 가능합니다**

형식 EECH <형식>

- 1: 첫 번째 W1축 및 TOOL 0번선택
- 2: 두 번째 W2축 및 TOOL 1번선택

설명 1) ROSEP Robot의 경우 말단 장치가 2개(W1, W2)로 구성되어 있어 XY좌표 변환 시 적용할 말단 장치를 선택 하는 명령어 입니다.  
 2) "EECH 1"으로 설정 하면, 로봇의 W1축 (Lower Arm)과 말단 장치의 Tool 0이 선택 됩니다.  
 3) "EECH 2"으로 선택이 되면 로봇의 W2축(Upper Arm)과 말단 장치의 Tool 1이 선택 됩니다.



[ROSEP ROBOT Arm 선택]



3.15.1 프로그램 사용 예제

MAIN	.....	Program 시작
VEL 10	.....	이동 속도 설정
WHILE 1	.....	조건반복수행 시작
JMOV P0	.....	포인트 P0으로 PTP 이동
JMOV P1	.....	포인트 P1으로 PTP 이동
EECH 1	.....	로봇의 W1 말단 장치 선택
LMOV P2	.....	포인트 P2으로 CP 이동
LMOV P1	.....	포인트 P1으로 CP 이동
JMOV P3	.....	포인트 P3으로 PTP 이동
EECH 2	.....	로봇의 W2 말단 장치 선택
LMOV P4	.....	포인트 P4으로 CP 이동
LMOV P3	.....	포인트 P3으로 CP 이동
ENDWL	.....	조건반복문 종료
EOP	.....	Program 종료

### 3.16 TIMOV (TOOL 좌표계 기준 증분 이동 명령어)

기능 현재 위치에서 TOOL 좌표계 기준으로 증분량만큼 직선 보간 이동  
**TIMOV 명령어는 ROSEP Robot 및 Ver 03.02.08 이후 버전에서만 사용 가능합니다**


형식 TIMOV 위치형 변수

<위치형 변수> : POS형으로 선언된 변수 이름을 의미 합니다.

XY 좌표를 기준으로 이동 하기 때문에 위치형 변수 이름은 X으로 시작 해야 합니다.

예) POS XP, XCUR

설명 1) TIMOV는 TOOL 좌표계 기준으로 현재 위치에서 지정된 위치(dx, dy, dz)만큼 상대 이동 합니다.



#### ⚠ CAUTION

- ▶ TIMOV를 위한 Point Teaching에서 위치좌표를 저장할 때 파라미터의 RANG"이내의 값으로 저장하지만 이 값은 원점(ORIGIN)을 기준으로 설정된 위치값입니다.
- ▶ 따라서 이전 Point를 기준으로 로봇이 이동(TIMOV)할 때 "Range over error"가 발생할 수 있습니다.
- ▶ 파라미터 "RANG"설정을 조정하십시오.
- ▶ TIMOV는 이전 위치를 기준으로 이동하므로 이전위치의 In Position정도에 따라 위치편차가 발생할 수 있습니다.  
 이때 작업프로그램에 "INPOS","DLAY"등을 추가 하여 로봇이 이전 위치에 정확하게 In Position 완료 후 TIMOV가 이루어 지도록 프로그램을 수정하시기 바랍니다.

### 3.17 HMOV (원점 이동 명령어)

기능 현위치에서 지정된 홈 위치로 이동

형식 HMOV <정수형 변수 또는 번호>

용어 <번호> : 파라미터에서 설정된 원점 좌표값 번호 입니다.  
(0 ≤ 번호 ≤ 3)

<정수형 변수> : INT 형으로 선언된 정수형 변수의 이름을 의미 합니다.

설명 1) ORG 파라미터의 HOME 에서 설정한 **원점 좌표 값으로 로봇을 PTP이동** 합니다.  
2) 원점 좌표값 번호는 0번 ~ 3번 (HMOV 0, HMOV 1, HMOV 2, HMOV 3) 까지 4개의 값을 가집니다.

#### CAUTION

- ▶ **HMOV 0**로 이동 하는 경우에는 **ORG 파라미터의 SEQ에서 설정한** 원점 순서에 따라 이동 합니다.
- ▶ **HMOV1 ~ HMOV 3**으로 이동 하는 경우에는 **SEQ와 무관하게 PTP 이동** 합니다.

#### 3.17.1 프로그램 사용 예제

1) 포인트 티칭한 위치좌표로 이동

```

MAIN
VEL 100
JMOV P1 ..... 포인트 P1 으로 PTP 이동
HMOV 0 ..... 원점 수행 SEQ로 HOME 0으로 이동
HMOV 1 ..... HOME 1 으로 PTP 이동
EOP
    
```

### 3.18 PMOV (Palletizing 이동)


- 기능     현위치에서 지정된 Palletizing 작업 수행
- 형식     PMOV <작업 Pallet 번호> <작업 기준점>
- 용어     <작업 Pallet 번호> : 파라미터에서 설정한 Pallet 번호를 입력 합니다.  
           (0 ≤ 번호 ≤ 99)  
           <작업 기준점> : Pallet 위치 4개의 티칭 포인트 중 처음 포인트 번호를 입력 합니다.  
           (0 ≤ 번호 ≤ 1999)  
           예) PMOV P0 P1 인 경우: 기준점 (P1), X방향(P2), Y방향(P3), Z방향(P4)
- 설명     1) **PTP 동작**으로 Palletizing 작업 위치로 이동 합니다.  
           2) PMOV 수행은 파라미터의 카운터(CNT)에서 지정한 Pallet상의 위치로 이동 하는 것  
           입니다.  
           3) Palletizing 작업을 수행하려면 두 가지 내용이 설정 되어 있어야 합니다.  
           A. 파라미터 모드에서 팔레트 정보 입력  
           - **최초 정보 입력하기 전에 Pallet 데이터를 초기화 하십시오.**  
           B. Pallet 상의 4개의 포인트를 티칭 하십시오

#### ! CAUTION

- ▶ 기계부가 수평 다관절 로봇(SCARA 로봇) 일때는 A, B arm의 "OFFSET" 값을 정확하게 설정 하십시오.
- ▶ Pallet 작업순서는 파라미터에서의 TYPE에 따라 결정됩니다.  
(취급 설명서를 참조 바랍니다.)

#### ! CAUTION

- ▶ 회전 축 (W축)이 있는 로봇에서 Z축 끝단(End effect)에 핸드(HAND)를 부착 하였을 때 팔레트 티칭 방법은 아래와 같이 핸드자세를 동일하게 한 상태로 티칭 해야 합니다.



3.18.1 TOOL 방향을 일정 하게 유지 하면서 티칭 (SCARA 로봇)

Step 1.

```
<RBSA804A : EDIT> V: 50
A : 0.13      B : 0.12
Z : 0.11      W : 65

EXCH  CORD  PJUMP  FWRD
```

```
<RBSA804A : EDIT> V: 50
A : 0.13      B : 0.12
Z : 0.11      W : 65

EXCH  CORD PJUMP  FWRD
```

F2

```
<RBSA804A : EDIT> V: 50
X : 595.32    Y : 0.13
Z : 0.11      W : 65

EXCH  CORD  PJUMP  FWRD
```

```
<RBSA804A : EDIT> V: 50
X : 400.38    Y : 101.13
Z : 0.11      W : 65

EXCH  CORD  PJUMP  FWRD
```


티칭 매뉴얼을 참조하여 포인트 화면으로 이동 합니다.

「F2」를 입력 하여 좌표계를 원통 → 직교 로 변경 합니다.


직교좌표계 화면에서 JOG 동작을 실시 하면 W축자세를 유지한 상태로 이동 됩니다.

←	J1	→	X축
	X		
←	J2	→	Y축
J	J	K	
←	J3	→	Z축
O	Z	P	
←	J4	→	W축
T	RX	U	
←	J5	→	EX1축
Y	RY	Z	
←	J6	→	EX2축
	RZ		

3.18.2 Pallet 작업 예




① 작업순서 → 포인트 티칭




② 파라미터 설정 값 중 DATA설정

Pallet NO = 01  
1.XWM=03    3.ZWM=01  
2.YWM=03

▶ P4 포인트티칭은?  
- 1단인 경우: P4=P1  
- 2단 이상인 경우:  
P4는 P1의 Z축값만 다름




① 작업순서 → 포인트 티칭




② 파라미터 설정 값 중 DATA설정

Pallet NO = 02  
1.XWM=03    3.ZWM=01  
2.YWM=03

▶ P12 포인트티칭은?  
- 1단인 경우: P12=P9  
- 2단 이상인 경우:  
P12는 P9의 Z축값만 다름




① 작업순서 → 포인트 티칭




② 파라미터 설정 값 중 DATA설정

Pallet NO = 20  
 1.XWM=03    3.ZWM=01  
 2.YWM=02

▶ P23 포인트티칭은?  
 - 1단인 경우: P23=P20  
 - 2단 이상인 경우:  
 P23는 P20의 Z축값만 다름




① 작업순서 → 포인트 티칭



② 파라미터 설정 값 중 DATA설정

Pallet NO = 99  
 1.XWM=04    3.ZWM=01  
 2.YWM=01

▶ P4 포인트티칭은?  
 - 1단인 경우: P4=P1  
 - 2단 이상인 경우:  
 P4는 P1의 Z축값만 다름



**CAUTION**


▶ pallet 작업패턴 NO. 1 ~ 3의 내용은 사용설명서의 "제5장 파라미터 모드에 대하여"의 내용 참조바람.



3.18.3 프로그램 사용 예제 (1)

1) 작업내용


→ 로봇이 부품공급 Pallet의 부품을 집어, 콘베어 파레트에 부품을 공급한다.



2) 로봇 사용자(USER) I/O

IN		OUT	
0	공급 Pallet 감지	0	핑거 클램프 시작
1	콘베어 Pallet 감지	1	핑거 언클램프 시작
2	핑거 클램프 감지	2	Pallet 작업완료
3	핑거 언클램프 감지	3	1 싸이클 완료
4		4	

3) Pallet 데이터 파라미터에 설정




4) 프로그램 작성

MAIN		
VEL 500	.....	속도 50% 설정
PLUP 5	.....	Z축 풀업 5mm로 설정
POUT0=0	.....	출력 ALL OFF
CALL HOFF	.....	핑거 언클램프
LABL A0		
IN0=1	.....	Pallet가 도착할 때 까지 대기
CALL PALT	.....	Pallet작업 서브루틴 호출
IN1=1	.....	콘베어Pallet가 도착할 때 까지 대기
JMOV P100	.....	콘베어 위치 P100로 이동
CALL HOFF	.....	핑거 언클램프
CALL PALT	.....	Pallet 작업 서브루틴 호출
JMOV P101	.....	콘베어 위치 P101로 이동
CALL HOFF	.....	핑거 언클램프
OUT3=1 100	.....	1사이클 작업완료
GOTO A0	.....	LABL A0로 점프
EOP		
	.....	Pallet에 소재집는 작업
SUBR PALT		
PMOV P1 P1		
CALL HON		
DLAY 100	.....	핑거 언클램프
RET		
SUBR HOFF		
OUT0=0		
OUT1=1		
IN3=1	.....	핑거 클램프
RET		
SUBR HON		
OUT1=0		
OUT0=1		
RET		

3.18.4 프로그램 사용 예제 (2)


- 1) 작업내용 → 로봇이 부품공급 Pallet의 부품을 집어 가공기에 공급 후, 가공완료된 부품을 다시 공급Pallet에 적재하는 작업방식



- 2) 로봇 사용자(USER) I/O

IN		OUT	
0	공급 Pallet 작업시작	0	Pallet 소재배출 작업완료
1	가공기 작업시작	1	가공기 소재공급 완료(가공시작)
2	공급핑거 클램프 감지	2	공급핑거 클램프 시작
3	공급핑거 언클램프 감지	3	공급 핑거 언클램프 시작
4	배출핑거 클램프 감지	4	배출핑거 클램프 시작
5	배출핑거 언클램프 감지	5	배출핑거 언클램프 시작
6	핸드 원위치 감지	6	Pallet 소재공급 작업완료
7	핸드 180 도 회전 감지	7	핸드 원위치 시작
8		8	핸드 180 도 회전 시작

- 3) Pallet 데이터 파라미터에 설정



4) 프로그램 작성

MAIN		
POS AP	.....	POS변수 선언 AP
AP=<0,0,-100,0,0,0>		
VEL 500	.....	속도 50% 설정
PLUP 5	.....	Z축 풀업 5mm 설정
JMOV P1	.....	대기위치 이동
POUT0=0	.....	출력신호 전체리셋
CALL H0	.....	로봇핸드 원위치(0도)
CALL HOFF1	.....	로봇핸드 공급핑거 언클램프
CALL HOFF2	.....	로봇핸드 배출핑거 언클램프
LABL A0	.....	레이블 A0설정
IN0=1	.....	Pallet 작업시작 신호대기
CALL DEPALT	.....	부실행문 "DEPALT" 호출
JMOV P1	.....	대기위치 이동
PLUP 0	.....	Z축 풀업 해제
IN1=1	.....	가공기 작업시작 신호대기
CALL H180	.....	부실행문 "H180" 호출
JMOV P20	.....	가공기축 포인트 P20→021로 이동
JMOV P21	.....	
CALL HON2	.....	로봇핸드 배출핑거 클램프
JMOV P20	.....	가공기축 상단포인트 P20로 이동
CALL H0	.....	로봇핸드 원위치(0도)
JMOV P21	.....	가공기축 하단포인트 P21로 이동
CALL HOFF1	.....	로봇핸드 공급핑거 언클램프
JMOV P20	.....	
JMOV P1	.....	포인트 P20→P1로 이동
OUT1=1 100	.....	가공시작
CALL H180	.....	부실행문 "H180" 호출
CALL PALT	.....	부실행문 "PALT" 호출
IMOV AP	.....	상대이동 (Z축 100mm 상승)
CALL H0	.....	로봇핸드 원위치
GOTO A0	.....	레이블 A0FH 점프
EOP	.....	"Palletizing 작업" 부실행문
SUBR DEPALT		
PLUP 5		
PMOV P2 P11		
CALL HON1		
DLAY 100		
RET		

SUBR PALT	.....	"Palletizing 작업" 부실행문
PLUP 5		
PMOV P3 P11		
CALL HOFF2		
DLAY 100		
RET		
SUBR HOFF1	.....	"로봇핸드 공급핑거 언클램프 작업" 부실행문
OUT2=0		
OUT3=1		
IN3=1		
RET		
SUBR HON1	.....	"로봇핸드 공급핑거 클램프 작업" 부실행문
OUT3=0		
OUT2=1		
IN2=1		
RET		
SUBR HOFF2	.....	"로봇핸드 배출핑거 언클램프 작업" 부실행문
OUT4=0		
OUT5=1		
IN5=1		
RET		
SUBR HON2	.....	"로봇핸드 배출핑거 클램프 작업" 부실행문
OUT5=0		
OUT4=1		
IN4=1		
RET		
SUBR H0	.....	"로봇핸드 원위치(0도) 작업" 부실행문
OUT8=0		
OUT7=1		
IN6=1		
RET		
SUBR H180	.....	"로봇핸드 회전(180도) 작업" 부실행문
OUT7=0		
OUT8=1		
IN7=1		
RET		

3.18.5 프로그램 사용 예제 (3)

1) 작업내용

복수개의 Pallet 작업시, 전원이 OFF/ON 해도 작업중이던 Pallet번호를 기억하여 계속 작업 가능.



프로그램 작성시 정수변수(INT 변수이름)를 선언해 주고 이변수명에 PIN9 값을 읽어들이 기억시키면 이전에 진행중이던 팔레트 번호가 기억됨.  
프로그램내에서 이 변수에 설정된 팔레트 번호를 비교하여 프로그램 작성.

▶ 예) 2개의 팔레트를 이용하는 작업

```

MAIN
INT PALNO          .....          정수변수선언 "PALNO"
PLUP 5
VEL 300
PALNO=PIN9        .....          PIN9의 값을 PALNO에 저장
IF PALNO==1 THEN  }                PALNO의 값이 "1"이면 NO.1 Pallet 작업중
GOTO PAL1
ELSE
IF PALNO==2 THEN  }                PALNO의 값이 "2"이면 NO.2 Pallet 작업중
GOTO PAL2
ENDIF
ENDIF

LABL LOOP
LABL PAL1
PMOV P1 P10
IF OUT1==0 THEN  }                NO.1 Pallet 작업완료가 아니면
GOTO PAL1        }                NO.1 Pallet 계속작업
ELSE
GOTO PAL2
ENDIF

LABL PAL2
PMOV P2 P20
IF OUT2==0 THEN  }                NO.2 Pallet 작업완료가 아니면
GOTO PAL2        }                NO.2 Pallet 계속작업
ELSE
GOTO LOOP
ENDIF
EOP
    
```

▶ 파라미터 Pallet 데이터 설정

3. OUT

Pallet No = 1  
1. Port : 01  
2. Time : 0200  
3. Type : 04

Pallet No = 2  
1. Port : 02  
2. Time : 0200  
3. Type : 04

- ▶ 전원이 꺼졌다 켜진 후에도 위 프로그램에서처럼 정수변수를 선언해 준 후 이곳에 PIN9를 저장하면 이전작업중이던 Pallet 번호를 기억할 수 있음

### 3.19 PASS

명령어	기능	형식
PASS	Pallet 작업중 지정한 작업물	PASS_<Pallet NO>_<통과할 작업물 순번>


#### 3.19.1 해설

- ▶ 관련 명령어: PMOV
- ▶ 1 Pallet당 PASS 사용개수: 20개  
→ 20개 이상 사용하면 "E150(Passing PLT over)"가 발생됨.
- ▶ 한 개의 JOB에 사용할 수 있는 PASS 명령어를 사용하는 PALT 수는 최대 5개까지 가능함  
→ 프로그램 예(2) 내용 참조

#### 3.19.2 프로그램 사용 예제 (1)

##### 1) 작업내용

→ 그림과 같이 Pallet 작업을 하려고 한다.

 표시부분만 작업함

1	2	3
4	5	6
7	8	9

##### 2) 프로그램 작성

```

MAIN
VEL 200
PASS 1 1
PASS 1 3
PASS 1 7
PASS 1 9
LABL A0
PLUP 10
PMOV P1 P10
DLAY 10
GOTO A0
EOP
    
```

} Pallet NO.1의 1번, 3번, 7번, 9번 작업물 작업않고  
통과



3.19.3 프로그램 사용 예제 (2)

MAIN		
VEL 200		
PLUP 10		
LABL A0		
<u>PASS 1 1</u>	}	Pallet NO.1의 1번, 3번 작업물 작업않고 통과
<u>PASS 1 3</u>		
<u>PMOV P1 P10</u>		
DLAY 10		
<u>PASS 2 2</u>	}	Pallet NO.2의 2번, 4번 작업물 작업않고 통과
<u>PASS 2 4</u>		
<u>PMOV P2 P10</u>		
DLAY 10		
<u>PASS 3 5</u>	}	Pallet NO.3의 5번, 6번 작업물 작업않고 통과
<u>PASS 3 6</u>		
<u>PMOV P3 P10</u>		
DLAY 10		
<u>PASS 4 1</u>	}	Pallet NO.4의 1번, 4번 작업물 작업않고 통과
<u>PASS 4 4</u>		
<u>PMOV P4 P10</u>		
DLAY 10		
<u>PASS 5 2</u>	}	Pallet NO.5의 2번, 3번 작업물 작업않고 통과
<u>PASS 5 3</u>		
<u>PMOV P5 P10</u>		
DLAY 10		
GOTO A0		
EOP		

### 3.20 WITH, ENDWT (동시처리 명령어)

기능     로봇 동작중 다음 수행문열을 동시처리 합니다.

형식     WITH

...

ENDWT

- 설명     1) WITH 문 내의 최초 **JMOV 명령어**를 수행하면서 이후 명령어를 병렬처리합니다.  
 2) MVR 변수를 사용하여 로봇이동거리를 백단위로 나누어 원하는 지점에서 조건식 등을 처리할 수 있습니다.


#### CAUTION

WITH문 내에서 조건식 만족에 의한 점프 명령어로 WITH ~ ENDWT 블록을 빠져 나가는 경우, **점프한 최초 스텝에 반드시 ENDWT를 삽입**해야 합니다.

#### 3.20.1 프로그램 사용 예제 (1)

1) 작업내용 →

P1에서 P2로 이동중 입력 IN0 신호가 ON되면 이동을 멈추고, P3로 이동.  
 (단, 센서입력신호는 P1에서 P2까지의 이동거리의 60% 까지만 검사한다.)



2) 로봇 사용자(USER) I/O

IN		OUT	
0	센서 감지	0	P2 도착완료
1		1	P3 도착완료

3) 프로그램 작성

MAIN		
VEL 500		
JMOV P1		
MVR=0	.....	“이동구간 백분율”을 초기화
<b>WITH</b>	.....	
JMOV P2	.....	포인트 P2로 이동(JMOV만 가능함)
WHILE MVR<60	.....	이동구간의 60%미만 동안에
IF IN0==1 THEN	.....	IN0=1이면 “LABL BB”로 점프
GOTO BB	.....	
ENDIF		
ENDWL		
OUT0=1		
<b>ENDWT</b>		
LABL BB		
STOP	.....	로봇이동 동작 정지
<b>ENDWT</b>	.....	<b>WITH문 종료 (반드시 삽입)</b>
JMOV P3		
OUT1=1		
EOP		

### 3.21 OUT, POUT (외부 출력 명령어)

기능 비트단위 또는 포트단위로 지정된 값을 출력한다.

형식 OUT<비트출력포트번호>=<0 또는 1> [펄스유효시간] [→ 또는 ~]  
 OUT<(정수형변수)>=<0 또는 1>  
 POUT<출력포트번호>=<데이터>  
 <변수>=POUT<출력포트번호>

용어 <비트출력포트번호> : 비트출력 포트번호를 설정합니다. (0 ≤ 비트출력번호 ≤ 95)  
 1) 사용자 출력(User Output) : OUT0 ~ OUT15  
 2) 확장1 출력(Option Output1) : OUT16 ~ OUT47 → 확장 I/O Card 1장 장착시 해당  
 3) 확장2 출력(Option Output2) : OUT48 ~ OUT79 → 확장 I/O Card 2장 장착시 해당  
 <펄스유효시간> : 해당 비트출력 포트에 0/1값을 출력(펄스출력)하는 유효시간.

[→ 또는 ~] :

- 펄스유효시간을 주기로 갖는 주기파형 출력
- 펄스출력, 주기파형 출력은 최대 32점(User, Option1, Option2 중 1개)까지 사용가능.
- Port를 중복 사용할 수는 없음. ( 예) User OUT 16점 + Option OUT 16점 )
- 1) "펄스 출력"과 "주기 파형출력"은 스텝 진행과 동시에 진행 됩니다.  
 즉, 다음 스텝이 이동명령(MOVE)인 경우 이동하면서 출력비트를 ON 또는 OFF 시킵니다.
- 2) "펄스출력"과 "주기파형출력"은 파라미터에서 사용할 Port를 정해야 합니다.  
 설정방법은 "파라미터 설명서를 참조하십시오.

<출력포트번호> : 출력포트 번호를 설정합니다. (0 ≤ 출력포트번호 ≤ 4)

- 0일 경우 : OUT0 ~ OUT15 를 지정
- 1일 경우 : OUT16 ~ OUT31 을 지정
- 2일 경우 : OUT32 ~ OUT47 을 지정
- 3일 경우 : OUT48 ~ OUT63 을 지정
- 4일 경우 : OUT64 ~ OUT79 를 지정

<데이터> : 해당 출력포트에 출력할 값 (2진수, 16진수가능)을 의미합니다.

- 예1) POUT0=10 일 경우 ( 10(10진수) => 0000 0000 0000 1010 (2진수) )  
 출력포트 1 번과 출력포트 3번을 ON(1) 시키며 다른 출력 포트는 OFF (0) 된다.
- 예2) POUT0=0H0F0F 일 경우 (0H0F0F (16진수) => 0000 1111 0000 1111 (2진수) )  
 출력포트 1 번과 출력포트 3번을 ON(1) 시키며 다른 출력 포트는 OFF (0) 된다.

**설 명**     **OUT<비트출력포트번호>=<0 또는 1> [<펄스유효시간>] [->]**

- 1) 지정된 펄스 유효시간만큼 해당비트출력 포트에 0/1값을 출력합니다.
- 2) 펄스 유효시간이 없으면 계속 유효하며 펄스유효시간이 지나면 이전상태로 복귀합니다.
- 3) 펄스 유효시간의 단위는 10ms 입니다.

**OUT<정수형 변수>=<0 또는 1>**

"비트출력포트번호"에 정수형 변수를 사용할 수 있습니다.


**POUT<출력포트번호>=<데이터>**

데이터 값을 해당 출력포트에 출력(16비트)합니다.


**! CAUTION**

- ▶ OUT, POUT과 "비트출력포트번호"," 출력포트번호"사이에 공란(Blank)이 들어가지 않도록 주의하시기 바랍니다.  
예) OUT\_15=1,POUT\_0=0H00FF (Syntax Error 발생)
- ▶ 출력포트 번호의 데이터값은 우측에서 좌측으로 0,1,2 ~ 15 의 순으로 증가 합니다.
- ▶ **내부 접점은 펄스 출력 제어가 되지 않습니다.**
- ▶ 확장 I/O 보드는 옵션 구매품 입니다.

**POUT0=0H0F0F**



**POUT0=0H1964**



3.21.1 프로그램 사용 예제

1) 펄스출력, 데이터출력

MAIN		
VEL 100		
JMOV P10		
<b>OUT11=1 100</b>	.....	출력 포트 11번을 1s 동안 ON(1) 시킴
LMOV P100		
<b>POUT0=0H1004</b>	.....	출력포트 0~15번에 0H1004(16진수)값으로 출력
EOP		

2) 비트출력포트번호를 변수로 사용

MAIN		
INT A	.....	정수형 변수 A를 선언
FOR A=0 TO 10	.....	0 ~ 10까지 반복 수행을 하며
OUT(A)=0	.....	OUT0,OUT1 ~ OUT10 의 출력을 모두 OFF(0)
NEXT		
EOP		


### 3.22 IN, PIN (외부 입력 명령어)

기능	지정된 비트단위 또는 포트단위의 입력에서 값을 받는다.
형식	IN<입력비트번호>=<0 또는 1> PIN<입력포트번호>=<정수값 또는 변수> <변수>=IN<입력비트번호> <변수>=PIN<입력포트번호>
용어	<비트입력포트번호> : 비트입력 포트번호를 설정합니다.(0 ≤ 비트입력번호 ≤ 95) 1) 사용자 입력(User Input) : IN0 ~ IN15 2) 확장1 입력(Option Input1) : IN16 ~ IN47 → 확장 I/O Card 1장 장착시 해당 3) 확장2 입력(Option Input2) : IN48 ~ IN79 → 확장 I/O Card 2장 장착시 해당 <입력포트번호> : 입력포트 번호를 설정합니다. PIN, WIN 인 경우 : 16bit 묶음(word)을 의미합니다. -. 0일 경우 : IN0 ~ IN15 를 지정 -. 1일 경우 : IN 16 ~ IN31 을 지정 -. 2일 경우 : IN 32 ~ IN 47 을 지정 -. 3일 경우 : IN 48 ~ IN 63 을 지정 -. 4일 경우 : IN 64 ~ IN 79 를 지정
설명	<b>IN&lt;비트입력포트번호&gt;</b> 지정된 비트입력포트의 ON/OFF상태(1또는 0)를 IN<비트입력포트번호>변수에 저장합니다. <b>PIN&lt;입력포트번호&gt;</b> 지정된 입력포트의 값(16비트)을 읽어서 PIN<입력포트번호>변수에 저장합니다. <b>IN&lt;비트입력포트번호&gt;=&lt; 0/1 &gt;</b> 지정된 비트입력포트가 ON/OFF상태(1또는 0)가 될 때까지 대기합니다. <b>PIN&lt;입력포트번호&gt;=&lt;16비트값&gt;</b> 지정된 입력포트의 값이 지정된 값(16비트)이 될 때까지 대기합니다.


 **CAUTION**

- ▶ IN,PIN과 "비트입력포트번호"," 입력포트번호"사이에 공란(Blank)이 들어가지 않도록 주의 하시기 바랍니다.
- 예) AA=IN\_15, AA=PIN\_0 (Syntax Error 발생)
- ▶ 입력포트 번호의 데이터값은 우측에서 좌측으로 0,1,2 ~ 15 의 순으로 증가 합니다.
- ▶ 확장 I/O 보드는 옵션 구매품 입니다.

PINO=0HFF0F




PINO=0H0009



입력 "0"번 포트의 입력중 "IN4~IN7" 사이의 신호만 체크하려면?

AA=0H00F0&PINO





3.22.1 프로그램 사용 예제

1) 입력 대기, Port입력을 정수변수처리 이동선택

MAIN		
INT D1	.....	정수형 변수 D1 선언
VEL 100		
WHILE 1		
JMOV P0		
<b>IN1=1</b>	.....	비트 입력 포트 1 번이 ON(1) 될때까지 대기
<b>IF IN0==1 THEN</b>	.....	비트 입력 포트 0번이 ON(1) 일경우
<b>D1=PIN0</b>	.....	입력포트 0~15 까지의 상태를 D1에 저장
ELSE		
<b>PIN0=0HFFFF</b>	.....	입력포트 0~15까지 모두 ON(1) 될때까지 대기
ENDIF		
IF D1==0HFF0F THEN	.....	D1의 상태 (즉 PIN0의 상태)를 0HFF0F와 비교
JMOV P1		
ELSE		
JMOV P2		
ENDIF		
ENDWL		
EOP		

2) BCD 유닛을 이용한 MODEL 비교

- BCD 입력은 IN0 ~ IN7까지 입력을 사용 한 예

MAIN		
INT M1,M2,MD	.....	정수형 변수 M1,M2,MD를 선언
WHILE 1		
<b>M1=PIN0 &amp; 0H000F</b>	.....	M1에 PIN0와 0H000F를 비교 하여 저장
<b>M2=PIN0&amp;0H00F0</b>	.....	M2에 PIN0와 0H00F0를 비교 하여 저장
<b>M2=(M2&gt;&gt;4)*10</b>	.....	M2의 값을 1의 자리로 쉬프트 후 10을 곱함 ㉠
<b>MD=M1+M2</b>	.....	연산된 M1과 M2의 값을 더함
<b>IF MD==1 THEN</b>		
JMOV P0		
ELSE		
<b>IF MD==11 THEN</b>		
JMOV P1		
ENDIF		
ENDIF		
ENDWL		
EOP		

 CAUTION

- ▶ 포트 입력의 경우 16진수 연산을 하기 때문에 BCD 유닛등으로 10을 입력 시, 제어기에서는 16으로 인식 합니다.  
이때는 상기 예제와 같이 쉬프트 연산 후 10진수 변환을 해주어야 정확히 계산 됩니다.

### 3.23 CIN,CBIN,CWIN,CDIN,CFIN(필드 버스용 입력 명령어)

기 능	필드 버스 카드에서 지정된 비트단위 또는 포트 단위의 입력 값을 받는다
형 식	<p>Bit 영역 명령어</p> <p>&lt;정수형 변수&gt;=CIN&lt;입력비트번호&gt;</p> <p>&lt;정수형 변수&gt;=CIN&lt;정수형 변수&gt;</p> <p>&lt;정수형 변수&gt;=CBIN&lt;입력포트번호&gt;</p> <p>Word 영역 명령어</p> <p>&lt;정수형 변수&gt;=CWIN&lt;입력포트번호&gt;</p> <p>&lt;정수형 변수&gt;=CDIN&lt;입력포트번호&gt;</p> <p>&lt;실수형 변수&gt;=CFIN&lt;입력포트번호&gt;</p>
용 어	<p>Bit 영역: 1Bit 단위로 읽고 쓰기를 할수 있는 영역 입니다.</p> <p>&lt;입력비트번호&gt;: 입력 비트번호를 설정합니다(0≤입력비트번호≤127)</p> <p>1) System 영역 : CIN0~CIN31(사용불가)</p> <p>2) User 영역 : CIN32 ~ CIN127</p> <p>&lt;입력포트번호&gt;: 입력포트 번호를 설정합니다.( 0≤입력포트번호≤15)</p> <p>1) System 영역 : CBIN0~CBIN3(사용불가)</p> <p>2) User 영역 : CBIN4~CBIN15</p> <ul style="list-style-type: none"> <li>- 포트번호 0: CIN0 ~ CIN7</li> <li>- 포트번호 1: CIN8 ~ CIN15</li> <li>- 포트번호 2: CIN16 ~ CIN23</li> <li>- 포트번호 4: CIN24 ~ CIN31</li> <li>- 포트번호 5: CIN32 ~ CIN39</li> </ul> <p>Word 영역: Word(16Bit)단위로 읽고 쓰기를 할수 있는 영역 입니다.</p> <p>&lt;입력포트번호&gt;: 입력포트 번호를 설정합니다.( 0≤입력포트번호≤15).</p> <p>1) User 영역: CWIN0 ~ CWIN15</p> <p>2) CDIN&lt;포트번호&gt;: Word 영역 값을 Double Word(32Bit)단위(정수)로 받는다.</p> <ul style="list-style-type: none"> <li>- 포트번호 0: CWIN0 ~ CWIN1</li> <li>- 포트번호 1: CWIN2 ~ CWIN3</li> <li>- 포트번호 2: CWIN4 ~ CWIN5</li> </ul> <p>3) CFIN&lt;포트번호&gt;: Word 영역 값을 Double Word(32Bit)단위(실수)로 받는다.</p> <ul style="list-style-type: none"> <li>- 포트번호 0: CWIN0 ~ CWIN1</li> <li>- 포트번호 1: CWIN2 ~ CWIN3</li> <li>- 포트번호 2: CWIN4 ~ CWIN5</li> </ul>

**설 명** CIN<입력비트번호>  
 지정된 비트 입력포트의 ON/OFF상태(1또는0)를 IN<비트입력포트번호>변수에 저장합니다.

CIN<정수형 변수>

“입력비트번호”에 정수형 변수를 사용할수 있습니다

CBIN<입력포트번호>

지정된 입력포트의 값(8Bit)을 읽어서 CBIN<입력포트번호>변수에 저장합니다.

CWIN<입력포트번호>

지정된 입력포트의 값(Word)을 읽어서 CWIN<입력포트번호>변수에 저장합니다.

CDIN<입력포트번호>

지정된 입력포트의 값(Double Word)을 읽어서 CDIN<입력포트번호>변수에 저장합니다.

CFIN<입력포트번호>

지정된 입력포트의 값(Double Word)을 읽어서 CFIN<입력포트번호>변수에 저장합니다.

**CAUTION**

- ▶ CIN,CBIN,CWIN,CDIN,CFIN과 “비트입력포트번호”, “입력포트번호”사이에 공란(Blank)이 들어가지 않도록 주의하시기 바랍니다.  
 예) AA=CIN\_15, AA=CBIN\_0 (Syntax Error 발생)
- ▶ 입력포트 번호의 데이터값은 우측에서 좌측으로 0,1,2 ~ 15 의 순으로 증가 합니다.
- ▶ 필드버스용 명령어는 PARA->PUB->HWCONF->COMM->FieldBus->MAP0|USER\_PEF일 경우만 적용 됩니다.

3.23.1 프로그램 사용 예제

```

MAIN
INT D1 ..... 정수형 변수 D1 선언
VEL 100
WHILE 1
JMOV P0
IF CIN50==1 THEN ..... 비트 입력 포트 51번이 ON(1) 일경우
D1=CBIN5 ..... 입력포트 40~47 까지의 상태를 D1에 저장
ELSE
D1=CBIN6 ..... 입력포트 48~45까지의 상태를 D1에 저장
ENDIF
IF D1==0H0F THEN ..... D1의 상태를 0H0F와 비교
JMOV P1
ELSE
JMOV P2
ENDIF
ENDWL
EOP
    
```

## 3.24 COUT,CBOUT,CWOUT,CDOUT,CFOUT(필드 버스용 출력 명령어)

기 능	필드 버스 카드로 비트단위 또는 포트단위로 지정된 값을 출력한다
형 식	<p>Bit 영역 명령어</p> <p>COUT&lt;출력비트번호&gt;=&lt;0또는1&gt;</p> <p>COUT&lt;정수형변수&gt;=&lt;0또는1&gt;</p> <p>CBOUT&lt;출력포트번호&gt;=&lt;데이터&gt;</p> <p>Word 영역 명령어</p> <p>CWOUT&lt;출력포트번호&gt;=&lt;데이터&gt;</p> <p>CDOUT&lt;출력포트번호&gt;=&lt;데이터&gt;</p> <p>CFOUT&lt;출력포트번호&gt;=&lt;데이터&gt;</p>
용 어	<p>Bit 영역: 1Bit 단위로 읽고 쓰기를 할수 있는 영역 입니다.</p> <p>&lt;출력비트번호&gt;: 출력 비트번호를 설정합니다.(<math>0 \leq \text{출력비트번호} \leq 127</math>)</p> <ol style="list-style-type: none"> <li>1) System 영역 : COUT0~COUT31(사용불가)</li> <li>2) User 영역 : COUT32 ~ COUT127</li> </ol> <p>&lt;출력포트번호&gt;: 출력포트 번호를 설정합니다.( <math>0 \leq \text{출력포트번호} \leq 15</math>)</p> <ol style="list-style-type: none"> <li>1) System 영역 : CBOUT0~CBOUT3(사용불가)</li> <li>2) User 영역 : CBOUT4~CBOUT15 <ul style="list-style-type: none"> <li>- 포트번호 0: COUT0 ~ COUT7</li> <li>- 포트번호 1: COUT8 ~ COUT15</li> <li>- 포트번호 2: COUT16 ~ COUT23</li> <li>- 포트번호 4: COUT24 ~ COUT31</li> <li>- 포트번호 5: COUT32 ~ COUT39</li> </ul> </li> </ol> <p>Word 영역: 1Word(16Bit)단위로 읽고 쓰기를 할수 있는 영역 입니다.</p> <p>&lt;출력포트번호&gt;: 출력포트 번호를 설정합니다.( <math>0 \leq \text{출력포트번호} \leq 15</math>).</p> <ol style="list-style-type: none"> <li>1) User 영역: CWOUT0 ~ CWOUT15</li> <li>2) CDOUT&lt;포트번호&gt;: Word 영역 값을 Double Word(32Bit)단위(정수)로 출력한다. <ul style="list-style-type: none"> <li>- 포트번호 0: CWOUT0 ~ CWOUT1</li> <li>- 포트번호 1: CWOUT2 ~ CWOUT3</li> <li>- 포트번호 2: CWOUT4 ~ CWOUT5</li> </ul> </li> <li>3) CFOUT&lt;포트번호&gt;: Word 영역 값을 Double Word(32Bit)단위(실수)로 출력한다. <ul style="list-style-type: none"> <li>- 포트번호 0: CWOUT0 ~ CWOUT1</li> <li>- 포트번호 1: CWOUT2 ~ CWOUT3</li> <li>- 포트번호 2: CWOUT4 ~ CWOUT5</li> </ul> </li> </ol>

설 명 COUT<출력비트번호>=<0또는1>  
ON/OFF상태(1또는0)를 해당 출력비트 번호에 출력한다.  
COUT<정수형변수>=<0또는1>  
"출력비트번호"에 정수형 변수를 사용할수 있습니다.

CBOUT<출력포트번호>=<데이터>  
데이터(8Bit) 값을 해당 출력포트에 출력한다.  
CWOUT<출력포트번호>=<데이터>  
데이터(1Word) 값을 해당 출력포트에 출력한다.  
CDOUT<출력포트번호>=<데이터>  
데이터(2Word) 값을 해당 출력포트에 출력한다.  
CFOUT<출력포트번호>=<데이터>  
데이터(2Word) 값을 해당 출력포트에 출력한다.

**! CAUTION**

- ▶ COUT,CBOUT,CWOUT,CDOUT,CFOUT과 "출력비트번호"," 출력포트번호"사이에 공란(Blank)이 들어가지 않도록 주의하시기 바랍니다.  
예) COUT\_15=1,   CBOUT\_5=AA (Syntax Error 발생)
- ▶ 출력포트 번호의 데이터값은 우측에서 좌측으로 0,1,2 ~ 15 의 순으로 증가 합니다.
- ▶ 필드버스용 명령어는 PARA->PUB->HWCONF->COMM->FieldBus->MAP0| USER\_PEF일 경우만 적용 됩니다.

3.24.1 프로그램 사용 예제

1) Bit 영역 출력

```

MAIN
INT A ..... 정수형 변수 A를 선언
FOR A=32 TO 42 ..... 32 ~ 42까지 반복 수행을 하며
COUT(A)=0 ..... OUT32,OUT33 ~ OUT42 의 출력을 모두 OFF(0)
NEXT
EOP
    
```

2) Word 영역 출력

```

MAIN
POS CURR
WHILE 1
CURR=HERE ..... 로봇 현재 좌표를 CURR에 저장
CFOUT0=CURR.1 ..... CURR 1번축 값을 필드버스(CWOUT0,CWOUT1)
로 출력
CFOUT1=CURR.2 ..... CURR 2번축 값을 필드버스(CWOUT2,CWOUT3)
로 출력
ENDWL
EOP
    
```

### 3.25 VEL (축 이동 속도 설정 명령어)

기능 축 이동 속도의 백분율(%)을 설정

형식 VEL <속도>

용어 <속도> : 로봇의 이동속도를 설정합니다. ( $0 \leq \text{속도} \leq 100.0\%$ )  
 로봇이동속도 =  $Mv(\text{각 축의 최대속도}) * 0.001 * \text{속도}$   
 ( $0 \leq \text{Mv 설정 속도} \leq \text{최대 RPM}$ ).

- 설명
- 1) 로봇이 이동할 때의 속도이며, **최대속도를 1000으로 설정**합니다.
  - 2) 프로그램에서 속도 설정을 하지 않은 경우 INIT\_V 파라미터에 설정된 속도로 **자동 설정** 됩니다.
  - 3) 속도값을 갖는 **변수(정수형 변수)**를 사용할 수 있습니다.

 **CAUTION**

- ▶ 속도값은 100.0% 입니다.  
VEL 1000을 입력 시 100%의 속도로 이동 됩니다. (VEL 100 => 10%)
- ▶ 기구부의 허용 최대 RPM을 초과해서 사용시 소음 및 파손의 위험이 있습니다.  
반드시 기구부에 부착된 라벨을 확인 후 사용 하시길 바랍니다.

#### 3.25.1 프로그램 사용 예제

- 1) 작업위치에 따라 속도변경

```

MAIN
WHILE 1
VEL 1000           .....   로봇 이동 속도 MV x 0.001 x 1000
JMOV P0
JMOV P1
VEL 200           .....   로봇 이동 속도 MV x 0.001 x 200
LMOV P2
LMOV P3
VEL 1000          .....   로봇 이동 속도 MV x 0.001 x 1000
JMOV P0
ENDWL
EOP
    
```


### 3.26 ACC, DEC (가감속 설정 명령어)

기능 가속/감속 시간의 백분율(%)을 설정

형식 ACC <가속시간비율>  
DEC <감속시간비율>

용어 <가속시간비율/감속시간비율> : (50 ≤ <가감속시간비율> ≤ 200%)  
 -. 가감속시간을 프로그램에서 증가시키거나 감소시킵니다.  
 -. 가감속시간 = At(가감속시간) \* 0.01 \* 가감속비율(%)  
 (0.5\*최대가감속시간 ≤ <가감속시간> ≤ 2\*최대가감속시간)

설명 1) 로봇의 가감속시간을 설정합니다.  
 2) JOB프로그램 내부에서 가감속시간비율을 설정하지 않은 경우에는 100%로 설정됩니다.  
 3) 가감속비율값을 갖는 변수(정수형 변수)를 사용할 수 있습니다.  
 4) 모든 축에 적용됩니다. (SCARA 로봇 : A, B, Z, W축 적용)

예) 파라미터 At : A축(0.3초), B축(0.3초), Z축(0.2초), W축(0.5초)  
 ACC(DEC) 50 실행   
 At : A축(0.15초), B축(0.15초), Z축(0.1초), W축(0.25초)  
 ▶ 파라미터 값 자체가 수정되는 것이 아니고 JOB프로그램 내에서만 적용

5) 파라미터 가감속 시간(At)으로 복귀는 "ACC(DEC) 100"을 사용하고 다음스텝부터 적용됩니다.  
 (예제 프로그램 참조)

#### CAUTION

- ▶ 파라미터에 설정된 가감속 시간은 기계부 구성에 따라 최적으로 설정된 값입니다.
- ▶ ACC/DEC명령어의 가감속 시간비율이 100보다 작게 설정될 때는 기계부의 소음 및 진동이 발생할 수 있으니 주의하여 사용하십시오.

### 3.26.1 프로그램 사용 예제

1) 작업위치에 따라 가감속시간 변경


- 파라미터 에서 At 값이 0.3초로 설정 되어있을 경우의 예

MAIN		
WHILE 1		
VEL 500		
<b>ACC 50</b>	.....	가감속 0.15초로 설정
JMOV P1		
<b>ACC 200</b>	.....	가감속 0.6초로 설정
LMOV P2		
LMOV P3		
<b>ACC 100</b>	.....	가감속 0.3초로 설정
JMOV P0		
ENDWL		
EOP		




### 3.27 FOS, PFOS, (연속궤적생성 명령어)

- 기능 축 선단이 목표점에 도달하기 전에 다음 목표점으로 궤적 변경함
- 형식 FOS <거리비율>  
PFOS <거리>
- 용어 <거리비율> : ( $0 \leq \text{<거리비율>} < 50\%$ )
- 로봇이 지정된 위치에 도착하기 전에 다음 위치로 이동시키거나 출력할 때의 시점을 설정합니다.
  - 거리비율은 두점(Point) 이동 거리에 대한 백분율값으로 설정합니다.
  - JMOV를 제외한 다른 모션에서는 50% 이상 사용금지  
( $0 \leq \text{<거리비율>} < 50$ )



위 그림은 P0에서 P1으로 이동할 때 각 FOS값에 따라 다음 위치로 이동하거나 출력할 때의 시점을 표시.

<거리> : 로봇이 지정된 위치에 도착하기 전에 다음 위치로 이동시키거나 출력할 때의 거리를 설정합니다. ( $0 \leq \text{<거리>} \leq 999\text{mm}$ ) 단위: 0.01mm



위 그림은 P0에서 P1으로 이동할 때 각 PFOS값에 따라 다음 위치로 이동하거나 출력할 때의 시점을 표시.

#### CAUTION


- ▶ 연속된 보간이동위한 포인트 티칭시 두 점 사이의 거리는 최소 5mm이상일 것 (단 100mm/s 이동 속도 인 경우)
- ▶ 이동중인 궤적의 FOS 비율이 다음 위치의 전체길이 보다 짧을경우 "Too much FOS" Alarm 이 발생 됩니다.
- ▶ 이동궤적이 PTP인 경우 궤적 변경이 되며 보간동작일 경우 궤적 삽입(원,호 또는 직선)이 됩니다.

- 설 명
- 1) FOS 나 PFOS가 설정되면, MOVE 명령은 지정된 위치에 도착하기 전 (설정된 거리 비율또는 거리값)에 **다음 위치로 이동**합니다.
  - 2) OUT 명령인 경우에는 이동중에 출력이 가능합니다.
  - 3) PFOS는 보간 **모션(LMOV, CMOV, AMOV)**에만 적용됩니다.
  - 4) 작업 프로그램에서 FOS또는 PFOS적용 **이전의 상태로 복귀**하기위해 **"FOS 0"** 또는 **"PFOS 0"**를 사용하고, 명령어는 1스텝 다음 명령어부터 적용됩니다.

```

MAIN
FOS 10
LMOV P0
LMOV P1
FOS 0
LMOV P2
LMOV P3
LMOV P4
EOP
    
```

FOS 0 명령어 적용은 LMOV P3 부터 됩니다



**! CAUTION**


- ▶ FOS,PFOS를 사용함으로 연속된 보간동작이 이루어 집니다.
- ▶ 원호(AMOV) 또는 원(CMOV) 동작할 때 잘못된 Point Teaching으로 Auto RUN중 "Unreachable Point", "Inverse Error" 가 발생할 수 있습니다.
- ▶ 특히, 크기가 작은 AMOV또는 CMOV 명령어 사용할 때 주의하시기 바랍니다.

3.27.1 프로그램 사용 예제

1) PTP 동작 (JMOV)

```

MAIN
VEL 100
JMOV P0
FOS 5
JMOV P1
FOS 0
JMOV P2
EOP
    
```



2) 로봇 이동중 출력신호 (OUT, POUT) On, Off

<pre> MAIN VEL 100 JMOV P0 <b>FOS 10</b> LMOV P1 LMOV P2 OUT0=1 <b>FOS 0</b> JMOV P3 EOP                 </pre>	<p>FOS 10</p>
---	---------------

3) 로봇이동 중 다음스텝 명령어 처리


<pre> MAIN WHILE 1 VEL 100 JMOV P0 <b>FOS 10</b> IF IN0==1 THEN JMOV P1 ELSE IN1=1 JMOV P2 ENDIF <b>FOS 0</b> ENDWL EOP                 </pre>	<p>FOS 10</p> <p>①위치에서 P1도착하기 10%전에 IF 명령어 처리 즉, 입력 "IN0"에 신호가 들어오면 ②(점선)와 같은 궤적을 그리며 이동</p>
--	--

### 3.28 SFOS(연속궤적 및 등속 생성 명령어)

**기능** 축 선단이 목표점에 도달하기 전에 다음 목표점으로 궤적을 변경하는 것은 FOS와 동일하지만, SFOS 명령어 적용 시점부터 SFOS 설정 값을 기준으로 등속을 유지 특성이 있습니다.

**형식** SFOS <거리> <속도 레벨> <원호 각도>

**용어** <거리> : 로봇이 지정된 위치에 도착하기 전에 다음 위치로 이동시키거나 출력할 때의 거리를 설정합니다. ( $0 \leq \text{<거리>} \leq 999\text{mm}$ ) 단위: 0.01mm



위 그림은 P0에서 P1으로 이동할 때 각 SFOS값에 따라 다음 위치로 이동하거나 출력할 때의 시점을 표시.

<속도 레벨>: SFOS 명령어가 적용되는 이동 구간에서의 속도를 설정 합니다.

설정 값이 높을 수록 속도는 빨리 집니다.( $0 \leq \text{<속도 레벨>} \leq 10$ )


설정 값 0인 경우: 레벨 5와 동일

<원호 각도>: 연결되는 두 궤적과 궤적 사이의 원호 삽입 각도를 입력 합니다.

입력된 각도 값보다 작은 경우 원호가 삽입되지 않고, 클 경우 첫번째 인자의 값으로 원호가 삽입 됩니다.

설정 값 0도인 경우: 1도 값과 동일

설정 값 180도인 경우: 궤적 사이에 원호가 삽입되지 않고, Too much fos 알람이 발생 하지 않습니다. 180도 설정 사용시 궤적 변형 및 로봇 이동 중 진동이 발생 할 수 있습니다.



- 설 명
- 1) SFOS가 설정되면, MOVE 명령은 지정된 위치에 도착하기 전 (거리값)에 **다음 위치로 이동**합니다.
  - 2) **MOVE 명령어 사이 DLAY 명령어 사용이 불가능 합니다.**
  - 3) 연속된 보간이동에서는 가감속이 없이 연속동작(등속이동)을 할 수 있습니다.
  - 4) **SFOS는 보간 모션(LMOV, CMOV, AMOV)에만 적용됩니다.**
  - 5) 작업 프로그램에서 SFOS 적용 **이전의 상태로 복귀**하기 위해 **"SFOS 0 0 0"** 사용하고, 다음 명령어부터 적용됩니다.

```

MAIN
SFOS 10 0 0
LMOV P0
LMOV P1
SFOS 0 0 0
LMOV P2
LMOV P3
LMOV P4
EOP
    
```

SFOS 0 명령어 적용은 LMOV P3 부터 됩니다.


**! CAUTION**

- ▶ SFOS를 사용함으로 연속된 보간동작이 이루어 집니다.
- ▶ 원호(AMOV) 또는 원(CMOV) 동작할 때 잘못된 Point Teaching으로 Auto RUN중 "Unreachable Point", "Inverse Error" 가 발생할 수 있습니다.
- ▶ 특히, 크기가 작은 AMOV또는 CMOV 명령어 사용할 때 주의하시기 바랍니다.

### 3.28.1 연속 모션 속도 프로파일 패턴


1) FOS 0인 경우

FOS 관련 명령어가 없는 경우 궤적 연결 부분에서 속도가 0으로 됩니다.




2) FOS 10 & PFOS 10

FOS 명령어 사용 시 삽입 구간(원호)의 속도는 삽입 구간의 진입 속도로 유지 됩니다.



3) SFOS 10 0 0

SFOS 설정 값과 속도 레벨로 다음 궤적의 최고 속도(Vf)가 결정 되어, 이동 속도가 유지 됩니다.



**CAUTION**

- ▶ 연결 궤적의 길이가 너무 작아 vf를 유지 할수 없는 경우, 등속이 유지 되지 않습니다.
- ▶ 이럴 경우 속도 레벨을 작게 하거나, SFOS 량을 구간내의 최소 길이로 설정 합니다.

3.28.2 프로그램 사용 예제

1) 직선 보간 모션(LMOV)

<pre> MAIN VEL 100 LMOV P0 <b>SFOS 150 0 70</b> LMOV P1 LMOV P2 <b>SFOS 0 0 0</b> LMOV P3 EOP                 </pre>	<p>SFOS 150 0 70 P1, P2 도착 15mm전에 다음 위치로 이동</p>
--	---

2) 원호 보간 모션(AMOV)

<pre> MAIN VEL 100 JMOV P0 LMOV P1 <b>SFOS 150 0 180</b> AMOV P2 P3 AMOV P4 P5 <b>SFOS 0 0 0</b> LMOV P6 EOP                 </pre>	<p>SFOS 150 0 180(0) P1, P2 도착 1.5mm전에 다음 위치로 이동</p>
---	--

### 3.29 SVON, SVOF (서보 ON/OFF 명령어)

기능 서보 ON, 서보 OFF

형식 SVON : 모든 축 서보 ON  
 SVON [<지정축>] : 지정 축 서보 ON  
 SVOF : 모든 축 서보 OFF  
 SVOF [<지정축>] : 지정 축 서보 OFF

설명 1) SVOF :  
 -. 로봇 동작중 **서보를 OFF** 시켜 로봇 Body를 Free(손으로 로봇을 이동시킬 수 있는 상태) 상태로 전환함.  
 -. 현재 구동중인 모든 축을 서보 OFF 시킴

2) SVON :  
 -. 로봇 Body를 서보 ON 시킴, SVOF와 관련  
 -. SVOF 후 SVON시키지 않은 상태에서 로봇동작(RUN)은 Dry RUN(로봇 Body는 움직이지 않고 JOB 프로그램이 1Step씩 실행) 상태임.  
 -. 현재 서보 OFF된 모든 축을 서보 ON 시킴

3) SVON(SVOF) <지정축> : 개별 축을 서보 ON/OFF 시킴. (범위:1~5)  
 SCARA 로봇 → A축 : 1, B축 : 2, Z축 : 3, W축 : 4  
 직각 로봇 → X축 : 1, Y축 : 2, Z축 : 3, W축 : 4, E1 : 5, E2 : 6

#### 3.29.1 프로그램 사용 예제

1) PTP 동작 (JMOV)

```

MAIN
VEL 100
JMOV P0
IF IN1==1 THEN
SVOF 4 ..... 4번째 축을 서보 OFF 시킴
ENDIF
JMOV P1 ..... 4번째 축 서보 OFF 상태에서 포인트 P1으로 이동
IF IN1==0 THEN
SVON 4 ..... 4번째 축을 서보 ON 시킴
ENDIF
JMOV P2 ..... 모든 축이 서보 ON 상태로 포인트 P2로 이동
ENDWL(<-삭제)
EOP
    
```



2) PTP 동작 (JMOV)

```

MAIN
VEL 500
JMOV P0
MVR=0
WITH
JMOV P1
WHILE MVR<100
IF IN0==1 THEN ..... P1으로 이동중 IN0=1이면
STOP .....
SVOF ..... 로봇 정지 후 서보 OFF
GOTO L1
ELSE
EXIT
ENDIF
ENDWL

LABL L1
ENDWT
IN0=1 ..... 입력 IN0 가 "1"이 된 후
SVON ..... 서보 ON
JMOV P2
EOP
    
```

 CAUTION

▶ 서브루틴에서 SVON, SVOF를 사용하면 "Syntax Error" 발생됨.


예)

```

MAIN
VEL 200
IN0=1
JMOV P0
CALL A0
EOP

SUBR A0
SVOF
OUT1=1
IN1=1
SVON
RET
    
```

프로그램 수정



```


MAIN
VEL 200
IN0=1
JMOV P0
SVOF
CALL A0
SVON
EOP

SUBR A0
OUT1=1
IN1=1
RET
    
```



### 3.31 OFFS (오프셋 지정 명령어)

- 기능     티칭 된 목표 지점을 지정 된 값만큼 임시 이동 시킵니다.
- 형식     OFFS P<번호>  
           OFFS GP<번호>  
           OFFS <위치형변수>
- 용어     <번호> : 티칭한 위치 좌표 번호를 설정합니다.  
           P – JOB 별로 개별 사용 되는 LOCAL POINT (0 ≤ 번호 ≤ 1999)  
           GP – 공통으로 사용 되는 GLOBAL POINT (0 ≤ 번호 ≤ 1023)  
           <위치형 변수> : POS형으로 선언된 변수이름을 의미합니다.
- 설명     1) JOB 프로그램의 **임의의 POINT에 Offset값을 각 직각좌표로 저장**.  
           OFFSET 명령어가 수행된 후, **이후 스텝의 이동명령(MOV)시 오프셋값 만큼 증감하여**  
           로봇이 이동합니다.  
           2) 실행되고 있는 **JOB 프로그램 내에서만** 유효합니다.  
           MAIN 프로그램의 OFFS값은 **Job Call(JCALL)된 프로그램에서 적용되지 않습니다.**



#### ! CAUTION

- ▶ JOB프로그램에 저장된 POINT가 파라미터의 "RANG"이내 에 위치하여 POINT Teaching할 때, "Range OVER" Error가 발생하지 않으나 JOB 프로그램이 실행될 때 OFFS에 의해 "Range OVER" Error가 발생할 수 있습니다.
- ▶ Point Teaching 할때에는 ANGLE 좌표로 입력하시기 바랍니다.
- ▶ OFFS 을 해제 하려면, <0,0,0,0,0> 의 값을 할당 해야 합니다.

#### 3.31.1 프로그램 사용 예제

<pre> PTP 동작 (JMOV) &lt;SCARA 로봇&gt; MAIN VEL 100 JMOV P10 ..... 포인트 P10 으로 PTP 이동 <b>OFFS P100</b> ..... P100에 저장된 값 으로 오프셋 적용 <b>LMOV P11</b> ..... P11 + P100 의 위치로 CP이동 &lt;110.23,122.37,7.96,120.00, 0, 0&gt; <b>LMOV P12</b> ..... P12 + P100 의 위치로 CP이동 &lt;130.23, 122.37, 20.96, 200.00, 0, 0&gt; EOP                 </pre>	<pre> P11 &lt;10.23, 22.37, 7.96, 120.00, 0, 0&gt; P12 &lt;30.23, 22.37, 20.96, 200.00, 0, 0&gt; P100 &lt;100, 100, 0, 0, 0, 0&gt;                 </pre>
--	---

### 3.32 LIMIT (축 제한 명령어)

- 기능    각 축의 이동범위를 제한한다.
- 형식    LIMIT P<번호1> P(번호2)  
           LIMIT <위치형변수> (위치형변수)
- 용어    <번호1> : 마이너스(-) 리밋값을 JOB 프로그램의 POINT에 설정.  
           (0 ≤ <번호> ≤ 1999)  
           (번호2) : 플러스(+) 리밋값을 JOB 프로그램의 POINT에 설정.  
           (0 ≤ <번호> ≤ 1999)
- 설명    1) JOB 프로그램내에서의 **소프트웨어 리밋을 설정**합니다.  
           2) 실행되고 있는 **JOB 프로그램내에서만** 유효합니다.  
           MAIN 프로그램의 LIMIT값은 **Job Call(JCALL)된 프로그램에서도 적용**됩니다.

#### 3.32.1 프로그램 사용 예제

1) PTP 동작 (JMOV) <SCARA 로봇>

```

MAIN
VEL 100
JMOV P0
LIMIT P100 P101
JMOV P1
EOP
    
```

JOB 프로그램내에서의 소프트웨어 리밋은 아래와 같이 적용됨.

A: -100도 ~ +100도  
 B: -100도 ~ +100도  
 Z: 0 ~ 100 mm  
 W:-300도 ~ +300도

수평다관절 로봇에서 리밋을 P100, P101에 설정할 경우

P100 위치: (A:-100.00 B:-100.00 Z:000.00 W:-300.00 E1:-100.00 E2:-100.00)  
 P101 위치: (A:100.00 B:100.00 Z:100.00 W:300.00 E1:100.00 E2:100.00)

### 3.33 PLUP (Pull up 동작 설정 명령어)

기능 Pull up 동작을 위한 Z축 값 설정

형식 PLUP <Z축값>

용어 <Z축값> : Z축 풀업 값을 설정 합니다.

( $0 \leq \text{Z축값} \leq \text{Z축 스트로크(STROKE)}$ )

- 설명
- 1) **Z축이 있는 로봇**에만 적용. Z축이 2개 이상 있을 경우 **첫번째 Z축만 적용** 됩니다.
  - 2) **PTP이동(JMOV, PMOV, IMOV)에만** 적용 되며, 보간이동에는 적용 되지 않습니다.
  - 3) Z축 풀업값을 가진 변수 (정수형 변수)를 사용 할 수 있습니다.
  - 4) Z축 풀업값 적용 해제는 **"PLUP 0"**를 사용 하고, 다음 스텝부터 적용 됩니다.
  - 5) Z축 풀업 값에 따른 로봇의 동작은 아래와 같습니다.


구분	로봇의 동작	해설
1. P1, P2 의 값이 모두 풀업 값 보다 큰 경우		<ol style="list-style-type: none"> <li>1. P1에서 Z축을 풀업 값 까지 상승</li> <li>2. Z축을 제외한 나머지 축을 P2의 풀업 값 위치로 이동</li> <li>3. Z축을 P2 위치까지 하강</li> </ol>
2. P1의 Z축 값은 풀업 값 보다 적고, P2의 Z축 값은 풀업 값 보다 큰 경우		<ol style="list-style-type: none"> <li>1. Z축을 제외한 나머지 축을 P2의 풀업 값 위치로 이동</li> <li>2. Z축을 P2 위치까지 하강</li> </ol>
3. P1의 Z축 값은 풀업 값 보다 크고, P2의 Z축 값은 풀업 값 보다 적은 경우		<ol style="list-style-type: none"> <li>1. P1에서 Z축을 P2의 Z축 위치 까지 상승</li> <li>2. P2 위치로 이동</li> </ol>
4. P1, P2 의 값이 모두 풀업 값 보다 적은 경우		<p><b>PLUP 명령과 상관 없이 P2로 이동</b></p>

3.33.1 프로그램 사용 예제

1) 포인트 티칭한 위치좌표로 이동

```

MAIN
VEL 100
PLUP 5 ..... Z축 풀업 값을 5mm 로 설정
JMOV P0
JMOV P1
PLUP 0 ..... Z축 풀업 값을 해제
JMOV P2
EOP
    
```



### 3.34 TOOL (축 제한 명령어)

기능	로봇에 부착된 사용 툴을 선택
형식	TOOL <툴번호>
용어	<툴번호> : 파라미터에서 "TOOL" 옵션값(dx, dy, dz)이 저장된 툴(TOOL)의 번호를 설정 ( $0 \leq \text{<툴번호>} \leq 3$ )
설명	1) 작업시 사용 하는 툴의 번호를 설정 합니다. 2) 툴의 형상은 시스템 파라미터에서 설정 합니다.

#### CAUTION

- ▶ 수평 다관절 로봇에서 TOOL 0의 옵션 값은 티치펜던트 JOG의 기준이 됩니다.
- ▶ 작업 프로그램에서 "TOOL 명령어"를 사용하지 않으면 "TOOL 0" 옵션값이 적용 됩니다.
- ▶ 가급적 "TOOL 0"은 TOOL 옵션 초기화용으로 사용하고 TOOL 옵션 적용은 1~3 중 하나를 사용 하십시오.

#### CAUTION

- ▶ 옵션이 있는 TOOL을 사용하는 작업에서 작업 POINT 티칭
  - 1) 정확한 TOOL 옵션 설정 (파라미터 참조)
  - 2) W축을 고정(FIX 1) 또는 회전(FIX 0) 중 1가지를 설정 후 작업 포인트 티칭 ("LMOV 명령어" 설명 참조)

#### CAUTION

- ▶ 연속된 보간 동작이 있는 작업(Sealing 또는 Dispensing)에서는 툴의 끝단으로 작업 POINT를 티칭 할 때 A(X), B(Y)축에 무리한 동작을 요구하는 티칭이 될 수 있습니다.
- ▶ 작업 POINT 티칭에 주의 하십시오.
- TOOL 제작 및 고정
  1. "LMOV 명령어"에서 설명한 W축 자세 고정이 가능하도록 TOOL 제작
  2. 툴 옵션이 dx,dy 둘 중 한쪽 방향으로만 존재 하도록 툴 고정

## 3.34.1 프로그램 사용 예제

## 1) 툴(TOOL)을 사용한 보간 이동

MAIN

VEL 300

**TOOL 1**

툴(TOOL)1번 을 사용 하여 보간동작 시작

FIX 0

LMOV P1

LMOV P2

OUT0=0

FIX1

**TOOL 0**

옵셋이 없는 툴 0번으로 변경

EOP



### 3.35 FIX (축 제한 명령어)

기능 보간 동작에서 W축 동작 선택


형식 FIX <형식>

용어 <형식>

1 → 로봇의 핸드 자세가 고정 (W축 자세 고정)


0 → 로봇의 핸드 자세가 회전 (W축 자세 회전)

- 설명
- 1) 로봇이 보간 동작을 할 때 로봇 핸드(W축)의 자세를 고정 또는 회전 합니다.
  - 2) PTP 동작에서는 FIX 설정과 관계 없이 교시점(티칭 된 POINT)으로 이동 합니다.
  - 3) "FIX 0"으로 설정 되면 로봇의 회전축(W축)은 교시점의 W축 좌표값으로 보간 이동 합니다.
  - 4) "FIX 1"로 설정 되면 로봇의 회전축(W축)은 시작점의 자세를 유지하면서 보간 이동 합니다.
  - 5) "FIX 1"로 설정 되면 도착축의 회전축(W축)좌표값은 최초 교시점과 다른 값을 가질 수 있습니다.
  - 6) 작업 프로그램내에서 FIX명령이 사용 되지 않으면 "FIX 1"로 설정 됩니다.
    - ※ 단 ROSEP Robot의 경우는 프로그램내에서 FIX명령이 사용 되지 않으면 "FIX 0"로 설정 됩니다.
  - 7) 작업 프로그램에서 FIX 적용 이전의 상태로 복귀하기 위해 "FIX 1"을 사용 하고 명령어는 2스텝 다음 명령어부터 적용 됩니다.



3.35.1 프로그램 사용 예제

W축 자세를 회전 시키면서 보간 동작 작업




```


MAIN
VEL 100
JMOV P0
JMOV P1
FIX 0 ..... W축 자세 회전
LMOV P2
AMOV P3 P4
LMOV P5
AMOV P6 P7
LMOV P8
AMOV P9 P10
LMOV P11
AMOV P12 P13
FIX 1 ..... W축 자세 고정
LMOV P1
EOP
    
```

### 3.36 FORM (로봇 ARM 형상 지정 명령어)

- 기능     로봇 동작시 로봇 ARM 형태 설정
- 형식     FORM <형태>
- 용어     <형태> : 수평 다관절 로봇 (SCARA) 에서 이동 할 때 ARM (A,B 축) 의 형상을 설정 (LEFT, RIGHT,NO)
- 설명     1) **수평 다관절 로봇(SCARA) 에만 적용**  
 2) 로봇이 이동 할 때 ARM의 형상을 설정 합니다.  
 3) 형태에는 3가지 종류(LEFT, RIGHT, NO)가 있습니다.



FORM RIGHT



FORM LEFT

#### CAUTION

- ▶ JMOV 명령어에서만 로봇 자세가 변경됩니다.
- ▶ 수평 다관절 로봇은 하나의 좌표 값에 대해 두가지 자세가 있습니다. (LEFT FORM, RIGHT FORM)
- ▶ LEFT FORM 으로 포인트 저장 하였어도 프로그램에서 "FORM RIGHT"가 지정 되면 저장된 POINT의 자세를 무시 하고 RIGHT FORM 으로 이동 합니다.
- ▶ FORM 명령어는 JMOV만 적용됩니다.

#### 3.36.1 프로그램 사용 예제

- 1) 포인트 티칭한 위치좌표로 이동

```

MAIN
VEL 100
JMOV P10
FORM LEFT           .....      ARM 형상을 LEFT로 설정
JMOV P11              .....      P11 위치로 이동
FORM RIGHT        .....      ARM 형상을 RIGHT로 설정
JMOV P11              .....      P11 위치로 이동
EOP
    
```

### 3.37 TRQ (충돌 감지 명령어)

기능 충돌 감지시 토크 리미트 알람 발생

형식 TRQ <지정축> <토크리미트>

용어 <지정축> : 1 → A(X)축, 2 → B(Y)축, 3 → Z축, 4 → W축, 5 → E1축, 6 → E2축  
 <토크리미트> : 범위 → 1~300[%] (300%는 모터 정격토크의 3배를 나타냄)

설명 1) 로봇동작중 지정된 각축의 토크리미트를 설정하여 그 이상의 토크가 파라미터 TOL로 설정한 시간이상으로 동작시 **Torque Limit Alarm 발생**. (파라미터 모드 참조)  
 2) 토크리미트값을 240 이상으로 설정시 토크값은 제한되지만 알람발생은 되지 않습니다.

#### 3.37.1 프로그램 사용 예제

```

MAIN
VEL 500
FOS 10
JMOV P0
WHILE 1
LMOV P1
TRQ 1 50 ..... A(X)축의 토크 리미트를 50%로 설정
TRQ 2 50 ..... B(Y)축의 토크 리미트를 50%로 설정
LMOV P2
LMOV P1
LMOV P3
ENDWL
EOP
    
```

P2, P1, P3로 이동중 파라미터 TOL로 설정한 시간 이상 외부 충격이 가해 졌을시 알람 발생

### 3.38 TQL (축 출력 토크 제한 명령어)

- 기능      각축의 최대 추력 토크값 제한.
- 형식      TQL <지정축> <토크리밋>
- 용어      <지정축> : 1 → A(X)축, 2 → B(Y)축, 3 → Z축, 4 → W축, 5 → E1축, 6 → E2축  
 <토크리밋> : 범위 → 1~300[%] (300%는 모터 정격토크의 3배를 나타냄)
- 설명      1) 통상 순간 토크는 정격의 약3배를 허용 하고 있으나, 이로 인해 모터 또는 기계의 강도에 문제가 생길 우려가 있을 경우에 한해 **최대 토크를 제한** 하기 위함.  
 2) 토크가 부족하여 일정시간 동안 위치에 도달 하지 못할 경우 "In Position Error"등의 알람이 발생 됩니다.

#### 3.38.1 프로그램 사용 예제

```

MAIN
VEL 500
FOS 10
JMOV P0
WHILE 1
LMOV P1
TQL 1 50 ..... A(X)축의 최대 출력 토크를 정격의 50%로 제한
TQL 2 50 ..... B(Y)축의 최대 출력 토크를 정격의 50%로 제한
LMOV P2
LMOV P1
LMOV P3
ENDWL
EOP
    
```

P2, P1, P3 로 이동시 TQL로 설정된 최대 출력토크 이하의 토크로 이동함.  
 토크가 부족하여 이동 하지 못할 경우 알람 발생

### 3.39 INPOS (목표점 도달 지정 명령어)

기능     로봇동작시 목표점 도달정도 표시.

형식     INPOS <목표점 도달정도지정>

용어     <목표점 도달정도지정> : 로봇이 목표점에 도달하였는지 확인하는 허용값을 설정.  
(단위 : 펄스량)

- 설명
- 1) INPOS 명령어를 사용하면 로봇이 이동명령을 수행한 후 모든 축이 목표점에 허용 정도 이내로 도달하지 않으면 다음명령을 수행하지 않습니다.
  - 2) 로봇 축의 현재 위치값과 목표 교시점과의 차이가 INPOS 명령에서 설정한 펄스 (Pulse)양 이내가 되면 도착점에 도달한 것으로 인정하고 다음 명령을 수행합니다.
  - 3) INPOS 명령어 적용해제는 "INPOS 0"를 사용하고, 다음 스텝 명령어부터 적용됩니다.

#### CAUTION

- ▶ INPOS 명령을 사용할 경우 현재위치를 정확히 찾기 위해 DELAY가 발생 되어 로봇 이동 Cycle Time이 증가 할 수 있습니다.
- ▶ FOS 동작이나 WITH 명령 수행중에는 INPOS 명령은 적용 되지 않습니다.

#### 3.39.1 프로그램 사용 예제

1) 툴(TOOL)을 사용한 보간 이동

```

MAIN
VEL 100
JMOV P10
INPOS 100           .....   도착 완료 허용 값을 100 PULSE로 설정
JMOV P1              .....   P1,P2위치로 이동한 후 각축의 현재값과 티칭된
JMOV P2              .....   값이 100PULSE 이내가 될때까지 대기
INPOS 0            .....   INPOS 명령 적용 해제
JMOV P3
EOP
    
```

### 3.40 MINIT(MAPPING 초기화 명령어)

기능	<u>맵핑 기능 초기화 명령어</u> <u>MINIT Ver 03.02.08 이후 버전에서만 사용 가능 합니다</u>
형식	MINIT <센서 타입>
용어	<센서 타입>: 맵핑 기능에 사용되는 센서 타입을 설정 합니다. 설정 값 1: NC 센서인 경우 설정 값 2: NO 센서인 경우
설명	1) 맵핑관련 제어기 내부 변수를 초기화 합니다. 2) 맵핑시 사용되는 센서 타입을 설정 합니다

#### CAUTION

- ▶ 맵핑 기능은 MINIT, MSTART, MREAD 3개의 명령어가 조합되어 하나의 기능을 수행 합니다.
- ▶ 정확한 맵핑 동작을 위해서는 위해서는 맵핑 초기화(MINIT), 맵핑 시작(MSTART), 맵핑 데이터 읽기(MREAD) 순서로 진행 해야 합니다.  
그렇지 않는 경우 "Latch Sequence Err" 알람이 발생 합니다.

### 3.41 MSTART(MAPPING 검출 명령어)

기능 맵핑 시작 명령어

이 명령어는 Ver 03.02.08 이후 버전에서만 사용 가능 합니다

형식 MSTART <축 지정> <GP 번호> <센서 입력핀>

용어 <축 지정> : 맵핑을 수행 할 축 할당

<GP 번호>: 맵핑 데이터를 저장할 글로벌 포인트 시작 번호 할당

<센서 입력핀> : 맵핑에 사용되는 센서 입력 핀 번호 할당

- 설명
- 1) 맵핑 수행시 사용되는 센서 입력핀 번호와 지정 축 그리고 맵핑 관련 데이터를 저장할 글로벌 포인트 시작 지점을 설정합니다.
  - 2) 맵핑 동작 중 감지된 상승 엣지와 하강 엣지 수를 글로벌 포인트 시작위치에 저장합니다.
  - 3) 맵핑 검출 가능한 **최대 개수는 50개** 입니다.

#### CAUTION

- ▶ 맵핑 기능을 사용 하기 위해서는 센서 입력 핀 번호와 축 지정 번호가 동일한 Servo Module 에 해당 되어야 합니다.  
그렇지 않는 경우 맵핑 기능이 되지 않습니다.

#### CAUTION


- ▶ 맵핑 기능은 MINIT, MSTART, MREAD 3개의 명령어가 조합되어 하나의 기능을 수행 합니다.
- ▶ 정확한 맵핑 동작을 위해서는 맵핑 초기화(MINIT), 맵핑 시작(MSTART), 맵핑 데이터 읽기(MREAD) 순서로 진행 해야 합니다.  
그렇지 않는 경우 "Latch Sequence Err" 알람이 발생 합니다.



### 3.4.1.1 맵핑 센서 입력 포트 관련

맵핑 기능을 정확하게 사용하기 위해서는 서보 모듈의 센서 입력 번호와 구동 축이 일치되어야 합니다.

각각의 서보 모듈은 2개의 센서 입력 핀을 갖고 있고, 아래 표 "서보 모듈별 센서 입력 번호"와 같이 할당 됩니다.



	서보 모듈 1	서보 모듈 2	서보 모듈 3
IN 1	1	3	5
IN 2	2	4	6

[서보 모듈별 센서 입력 번호]

핀 번호	기능
1	-
2	-
3	-
4	-
5	-
6	-
7	-
8	P24V(출력)
9	IN2(입력)
10	IN1(입력)
11	G24V(출력)
12	-
13	-
14	-
15	-

[서보 모듈 EXT IO 핀기능]

예) 스카라 로봇 Z축 맵핑 기능 사용 하는 경우

일반적으로 스카라 로봇이 Z축의 경우 서보 모듈 2에 연결되고, 맵핑 데이터가 GP100부터 저장, 센서는 IN1에 연결되는 경우

MSTART    3    GP100    3  
                   Z축    GPNT INDEX    센서 번호

### 3.42 MREAD(MAPPING 위치 데이터 갱신 명령어)

기능 맵핑 데이터 갱신 명령어.

이 명령어는 Ver 03.02.08 이후 버전에서만 사용 가능 합니다

형식 MREAD <제한 시간>


용어 <제한 시간> : 맵핑 데이터를 읽어오는 제한 시간.(단위 : ms)

- 설명
- 1) 맵핑 센서에서 검출된 위치 데이터를 MSTART에서 설정한 글로벌 포인트 시작 번호 다음부터 차례로 저장합니다.
  - 2) 최소 시간 설정은 예상 검출 포인트 개수 \* 20ms 이상으로 설정 해야 합니다.

#### 3.42.1 프로그램 사용 예제

##### 1) 카세트의 글라스 두께 측정

MAIN	.....	Program 시작
JMOV P0	.....	대기 위치 P0 이동
MINIT 1	.....	맵핑 초기화
MSTART 3 GP100 3	.....	센서 입력핀 3번과 Z축 선택
		맵핑 데이터는 GP100부터 시작
JMOV P1	.....	P1 위치로 이동
MREAD 500	.....	맵핑 데이터 GPNT에 저장
EOP		Program 종료



#### ⚠ CAUTION

- ▶ 맵핑 기능은 MINIT, MSTART, MREAD 3개의 명령어가 조합되어 하나의 기능을 수행 합니다.
  - ▶ 정확한 맵핑 동작을 위해서는 맵핑 초기화(MINIT), 맵핑 시작(MSTART), 맵핑 데이터 읽기 (MREAD) 순서로 진행 해야 합니다.
- 그렇지 않는 경우 "Latch Sequence Err" 알람이 발생 합니다.

2) 맵핑 데이터 저장 방법 설명

카세트에 글라스가 5개 있는 경우

```

<RGA80A : EDIT> V :      50
F: GOLD   GP: 100 US   B L
① A: 5.0   B : 5.0 ②
  Z: 0.0   W: 0.0
EXCH  CORD PJUMP FWRD
    
```

[상승 하강 엿지 개수 저장 화면]

- ① 상승 엿지 수
- ② 하강 엿지 수

```

<RGA80A : EDIT> V :      50
F: GOLD   GP: 101 US   B L
① A: 22.0  B : 24.0 ②
③ Z: 2.0   W: 23.0 ④
EXCH  CORD PJUMP FWRD
    
```

- ① 상승 엿지시 위치 값
- ② 하강 엿지시 위치 값
- ③ 두 위치 편차
- ④ 두 위치의 평균 위치 값

GP100의 첫번째 축에 센서의 상승 엿지 개수 5가 입력되고, 두번째 축에 센서의 하강 엿지 개수 5가 입력됩니다.

그 다음 맵핑을 수행하고 나서 MREAD 명령을 만나면 GP101의 첫번째 축에 첫번째 상승 엿지가 발생한 위치가 저장되고, GP101의 두번째 축에 첫번째 하강 엿지가 발생한 위치가 저장됩니다.

GP101의 세번째 축에는 하강 엿지 발생 위치와 상승 엿지 발생 위치 편차가 저장되고, GP101의 네번째 축에는 하강 엿지 발생 위치와 상승 엿지 발생 위치 합의 평균이 저장됩니다.

같은 형식으로 GP101~GP105까지 맵핑 데이터가 저장됩니다.

### 3.43 변수

#### 3.43.1 변수의 종류

변수는 일반 변수, POSITON 변수, 시스템 변수로 나눌 수 있습니다.

		LOCAL	GLOBAL
일반 변수	정수형	INT	I(번호)
	실수형	REAL	F(번호)
POSITION 변수		POS, P(번호)	GP(번호)
시스템 변수		CNT,TMR,MVR,HERE	

- LOCAL 변수 - 현재 구동중인 프로그램 내에서만 사용 가능 한 변수

- 1) INT : 정수형 변수로 산술연산, 입/출력 DATA 임시저장 등에 사용
- 2) REAL : 실수 형 변수로 산술 연산, POINT DATA 연산 등에 사용
- 3) POS : 위치형 변수로 단순 변수와 배열 변수로 구분
- 4) CNT : 입력 포트를 통해 들어오는 펄스 입력을 업카운트(UP COUNT) 하여 저장
- 5) TMR : 파라미터 설정값에 따라 초기 화면 값에서 1씩 증가
- 6) MVR : 두 POINT 사이를 이동 하는 경우, 전체 거리에 대한 현재 위치의 백분율 값
- 7) HERE : 로봇 현재 위치 값을 갖고 있는 변수
- 8) P(번호) : 사용자가 티칭한 위치 값을 가지고 있는 변수

- GLOBAL 변수 - 전체 프로그램 및 채널에서 공통으로 사용 가능 한 변수

- 1) I: 정수형 변수로 INT로 선언된 LOCAL 변수와 동일한 기능
- 2) F: 실수형 변수로 REAL로 선언된 LOCAL 변수와 동일한 기능
- 3) GP: 사용자가 티칭한 위치 값을 가지고 있는 변수

	RO(Robot)	TR(Transfer Robot)
Global INT	0 ≤ 번호 ≤ 499	0 ≤ 번호 ≤ 1999
Global FLOAT	0 ≤ 번호 ≤ 499	(0 ≤ 번호 ≤ 1999
Local Point	0 ≤ 번호 ≤ 1999	0 ≤ 번호 ≤ 1999
Global Point	0 ≤ 번호 ≤ 1023	0 ≤ 번호 ≤ 14999

### 3.43.2 사용 방법

- 사용위치
  - 프로그램에서 사용하는 일반변수는 반드시 프로그램 시작부분인 **"MAIN" 다음에 선언**
  - 부 프로그램(Subroutine)이나 프로그램의 시작부분 이외에서는 선언할 수가 없습니다.
- 구성
  - 일반변수의 **변수명**은 영문자와 숫자로 구성된 **8자리의 문자** 스트링입니다.
  - 변수의 첫자는 반드시 **영문자이거나 '\_'**입니다.(숫자 사용 불가)  
단, **"P", "GP", "I", "F"** 등은 Point 변수 및 GLOBAL 변수와 중복되므로 **사용할 수 없습니다**.
- 초기화
  - 변수는 **초기화 후 사용**해야 합니다.
  - CNT, TMR 변수는 변수사용 **이전 STEP에서 초기화**를 해야 합니다.
  - MVR 변수는 **MOVE 명령어가 시작되면 0으로 초기화** 됩니다.

### 3.44 정수형( INT, I, II ), 실수형 ( REAL, F ) 변수

#### 3.44.1 LOCAL 변수

기능 프로그램 내부에서 사용 가능한 정수형, 실수형 변수의 선언

형식 INT <변수명>,<변수명>,...  
 REAL <변수명>,<변수명>,...

용어 <변수명> :  
 - 영문자와 숫자로 구성된 8자리의 문자 스트링입니다.  
 - 변수의 첫자는 반드시 영문자 이거나 '\_'입니다.

설명 1) **MAIN 블록의 시작** 에서 선언하여야 하고, 부 프로그램내에서는 선언할수 없습니다.  
 2) INT로 선언된 변수는 정수형 변수 (정수값 범위 :  $-2.14 \times 10^9 \sim 2.14 \times 10^9$ )  
 3) REAL로 선언된 변수는 실수형 변수 (실수값 범위 :  $\pm 8.4 \times 10^{-37} \sim \pm 3.4 \times 10^{38}$ )

#### CAUTION

- ▶ "P", "GP", "I", "F" 등은 Point 변수 및 GLOBAL 변수와 중복되므로 사용할 수 없습니다. ("Syntax Error", "Duplicated Symbol" 등 ALARM 발생)
- ▶ 실수형 변수와 정수형 변수의 연산을 혼합 사용시 정수형으로 변경 됩니다.  
 $9.5 + 10 = 19$  (실수와 정수 연산시 오차 발생)  
 $9.5 + 10.0 = 19.5$  (실수 + 실수 연산으로 변경 사용)

#### 3.44.2 프로그램 사용 예제

##### 1) 정수형, 실수형 변수 사용

MAIN		
<b>INT J,BOUT</b>	.....	정수형 변수 J 와 BOUT를 선언
<b>REAL FF1</b>	.....	실수형 변수 FF1을 선언
WHILE 1		
<b>BOUT=0H000F</b>		
<b>J=1</b>	.....	변수 초기화
FF1=10.5		
FOR <b>J</b> =0 TO 10	.....	AP1 위치 좌표로 이동
JMOV P1	.....	AP2 위치 좌표로 이동
JMOV P2		
POUT0= <b>BOUT</b>	.....	정수형 변수 BOUT을 사용 하여 출력
NEXT		
<b>FF1=FF1+100.0</b>		실수형 변수 산술 연산
ENDWL		
EOP		

3.44.3 정수형 GLOBAL 변수 I, II

- 기능 INT로 선언된 정수형 변수와 동일한 기능을 가지며 전체 프로그램에서 공통으로 사용 가능한 변수 입니다.
- 형식 I<정수> : 사용 가능한 정수의 범위는 0~499 까지 입니다.  
 II<정수> : 사용 가능한 정수의 범위는 0~499 까지 입니다.  
 → I와 <정수>, II와 <정수> 사이에 공란(BLANK) 이 있으면 안됩니다.  
 → I와 (정수), II와 (정수변수) 사이에 공란(BLANK) 이 있으면 안됩니다.
- 설명
- 1) LOCAL 변수와 달리 **프로그램에서 선언하지 않고** 사용합니다.
  - 2) II<정수> 는 I<정수> 를 중복해서 사용한 것과 같습니다.  
 예) I35 가 20 이라면 II35 는 I20 과 같습니다.
  - 3) 프로그램에서 직접 I<정수> 나 II<정수>, I(정수), I(변수)의 값을 변경할 수 있습니다.
  - 4) VEL, ACC, DEC, FOS, PFOS, PLUP, DLAY, INPOS 에 사용할 수 있습니다.  
 예) VEL I20 : I20 에 저장된 값으로 속도(Velocity)를 설정합니다.
  - 5) IF - (ELSE) - ENDIF 블록과 WHILE-ENDWL 블록에서 사용할 수 있습니다.  
 예) IF I7 > A0 THEN                      예) WHILE II49 ≤ 100  
**※ FOR-NEXT 블록에서는 사용할 수 없습니다.**
  - 6) I<정수> 형은 포인트 전역변수(Global Point Variable)에 사용할 수 있습니다.  
 예) I9 가 3 이라면 PI9 는 P3 과 같습니다.  
**※ PII<정수> 형태는 사용할 수 없습니다.**

3.44.4 프로그램 사용 예제

MAIN		
<b>INT L,M,H</b>	.....	정수형 변수 L,M,H를 선언
<b>L=I0</b>	.....	L=300 할당
<b>M=I1</b>	.....	M=600 할당
<b>H=I2</b>	.....	H=1000 할당
VEL H	.....	속도를 H(1000) 으로 지정
JMOV P0		
<b>WHILE I10&lt;I11</b>	.....	I10<I11 을 만족하는 동안
FOS I4	.....	FOS I4 => FOS 10
LMOV PI25	.....	P(I25) 로 CP 이동
VEL L	.....	속도를 L(300) 으로 지정
AMOV PI26 PI27	.....	P(I26) 과 P(I27)을 경유하는 원호 보간
LMOV PI28	.....	P(I28) 로 CP 이동
<b>I25=I25+10</b>		
<b>I26=I26+10</b>	.....	1회 이동 후 I25,26,27의 값을 4씩 증가
<b>I27=I27+10</b>		
<b>I10=I10+1</b>	.....	카운터 I10을 1증가
ENDWL		
EOP		

I 변수 값
I0 = 300
I1 = 600
I2 = 1000
I10 = 1
I11 = 4
I25 = 1
I26 = 2
I27 = 3
I28 = 4

### 3.44.5 실수형 GLOBAL 변수 F

- 기능 REAL로 선언된 실수형 변수와 동일한 기능을 가지며 전체 프로그램에서 공통으로 사용 가능한 변수입니다.
- 형식 F<정수> : 사용 가능한 정수의 범위는 0~499 까지 입니다.  
 → F와 <정수> 사이에 공란(BLANK) 이 있으면 안됩니다.
- 설명
- 1) LOCAL 변수와 달리 **프로그램에서 선언하지 않고** 사용합니다.
  - 2) 프로그램에서 직접 F<정수>의 값을 변경할 수 있습니다.
  - 3) IF - (ELSE) - ENDIF 블록과 WHILE-ENDWL 블록에서 사용할 수 있습니다.  
 예) IF F7 > A0 THEN                      예) WHILE F49 ≤ 100  
 ※ **FOR-NEXT 블록에서는 사용할 수 없습니다.**

### CAUTION

▶ 실수형 변수와 정수형 변수의 연산을 혼합 사용시 **정수형으로 변경** 됩니다.  
 9.5 + 10 = 19 (실수와 정수 연산시 오차 발생)  
 9.5 + 10.0=19.5 (실수 + 실수 연산으로 변경 사용)

### 3.44.6 프로그램 사용 예제

```

MAIN
POS XP
XP=P1
FOR I0=0 TO 10
XP.1=F0           .....   위치형 변수 XP1 의 X축(1번째축)에 F0의 값을 대입
LMOV XP          .....   XP의 위치로 CP 이동
F0=F0+10.0      .....   F0 값을 10.0씩 증가
NEXT
EOP
    
```



### 3.45 POSITION 변수

#### 3.45.1 POS 변수


기능	위치형 변수 선언 (단순변수 및 배열변수)
형식	단순변수 → POS <변수명>, <변수명>, ... 배열변수 → POS <변수명>(크기), {<변수명>(크기)}, ...
용어	<<변수명> : -. 변수명은 <b>영문자와 숫자로</b> 구성된 8자리의 문자 스트링 및 배열의 크기입니다. -. 변수의 <b>첫자는 반드시 영문자</b> 입니다.("P","GP","I","F" 문자 단독사용 제외) (크기) : 배열변수의 크기는 괄호 안의 숫자(정수)로 나타냅니다. 배열 변수의 선언의 괄호 안의 숫자는 <b>방의 개수</b> 를 나타냅니다. <b>예) POS AA(3) 선언시 - 사용 가능한 변수는 AA(0), AA(1), AA(2) 3개.</b>
설명	1) <b>MAIN 블록의 시작</b> 에서 선언하여야 하고, 부 프로그램내에서는 선언할수 없습니다. 2) POS로 선언된 변수값은 각축의 각도, 리드값 또는 직교좌표값입니다. -. <b>X로 시작되는</b> 변수값( 예: XA, XA1, ... ) → 직교좌표값 변수 -. <b>X이외의 문자로</b> 시작되는 변수 → 각축의 각도, 리드값 변수, 각축의 증분량변수 3) POS로 선언된 변수는 각 <b>성분값은 &lt;변수명&gt;.&lt;번호&gt;</b> 로 사용할 수 있습니다. <b>예) POS XP1 → XP1.1 XP1.2 XP1.3 XP1.4...</b> 4) 각도값 POS 변수는 <b>최대 6가지 성분</b> 을 가지며, 변수명(단순변수, 배열변수)=<1축성분, 2축성분, 3축성분, 4축성분, 5축성분, 6축성분> 형식으로 초기화할 수 있습니다. 5) 직교좌표값 POS 변수는 <b>최대 7가지 성분</b> 을 가지며, 변수명(단순변수, 배열변수)= <X축성분, Y축성분, Z축성분, 4축성분, 5축성분, 6축성분, Arm Form값> 형식으로 초기화할 수 있습니다. → 기계부가 <b>SCARA로봇 일 때만</b> 적용됩니다. -. Arm Form 값 : 0 → LEFT form, 1 → RIGHT form, 2 → NO form 6) 각도값 POS변수와 직교좌표값 POS변수는 서로 연산 및 지정이 가능하며, 이 경우 저장되는 값의 형식에 따라 변환됩니다.

 CAUTION

- ▶ "P", "GP", "I", "F" 등은 Point 변수 및 GLOBAL 변수와 중복되므로 **사용할 수 없습니다.**  
( "Syntax Error" , "Duplicated Symbol" 등 ALARM 발생)
- ▶ 성분값에 **ARM FORM**이 없는 경우 각도값으로 판단되어 **XP**변수에 담을 때 **XY**좌표계로 변환
- ▶ 5,6번축은 확장성을 고려한 **사용 하지 않는 축** 입니다. **0.0**값을 입력 하여야 합니다.

### 3.45.2 POINT 변수

- 기능 Job Editor에서 사용자가 Teaching한 모든 Point를 사용
- 형식 LOCAL변수 → P(번호) (0 ≤ 번호 ≤ 1999)  
 GLOBAL변수 → GP(번호) (0 ≤ 번호 ≤ 1023)
- 설명 1) POS변수와 달리 초기에 변수 선언을 하지 않습니다.  
 2) POINT 변수는 각 성분값은 <포인트번호>.<번호>로 사용할 수 있습니다.



### 3.45.3 프로그램 사용 예제

- 1) POINT변수, POSITION변수를 활용한 감속 포인트 적용

MAIN		
<b>POS MM,XA,XB</b>	.....	위치형 변수 XA 와 MM, XB를 선언
<b>MM=&lt;10.1,10.2,10.3,10.4,0,0&gt;</b>	.....	MM을 초기화
<b>XA=&lt;400.0,50.0,10.0,0.0,0.0,1&gt;</b>	.....	XA를 초기화
<b>XB=XA</b>	.....	XB를 XA 값과 동일하게 초기화
<b>XB.3=XA.3-40.0</b>	.....	XB의 Z(3번째축)값을 40mm 감산
VEL 1000		
JMOV P0	.....	P0 위치로 PTP 이동
LMOV XB	.....	XB 위치 좌표로 CP 이동 (감속 포인트)
VEL 300		
LMOV XA	.....	XA 위치 좌표로 CP 이동
IMOV2 MM	.....	위치형 변수 MM값 만큼 증분 이동
VEL 1000		
JMOV P0		
EOP		

### CAUTION

- ▶ 변수명에 따라 성분값이 변환되어 저장됩니다.
- XP=<각도값> : 직각좌표계로 변환하여 저장됩니다.
- XP=<직각좌표값> : 성분값 그대로 저장됩니다.
- AP=<각도값> : 성분값 그대로 저장됩니다.
- AP=<직각좌표값> : 각도값으로 변환하여 저장됩니다.

### 3.46 시스템 변수 ( CNT, TMR, MVR, HERE )

#### 3.46.1 CNT, TMR 변수

- 기능 CNT : 카운터값 저장.  
 TMR : 타이머값 저장
- 형식 CNT<펄스입력 BIT 번호>=<초기값>  
 TMR0=<초기값>  
 TMR1=<초기값>
- 용어 <펄스입력 BIT 번호> : 입력으로 받고자 하는 입력 BIT 번호  
 <초기값> : 카운터(CNT), 타이머(TMR)가 시작되는 정수값
- 설명 1) 시스템 변수는 자료형을 선언하지 않고 사용합니다.  
 2) CNT 변수는 펄스 입력 BIT가 결정되는 순간 "0"이 저장되며  
 그 후 매 펄스입력(10m 이상의 펄스)을 카운트합니다. (0 ≤ 입력범위 ≤ 65,535)  
 3) TMR변수는 정수값을 입력하는 순간부터 값이 할당되어 시스템 파라미터에서 정의된  
 시간간격으로 1씩 증가합니다.  
 설정 값 적용은 파라미터 모드의 TMR을 참조 바랍니다.

 CAUTION

▶ CNT와 펄스입력포트번호 사이 ,TMR와 "0" 또는 "1" 사이에 공란(BLANK)이 들어가지 않도록 주의 하시기 바랍니다.  
 예) CNT\_0=0 (X), TMR\_1=-100 (X)

#### 3.46.2 프로그램 사용 예제

1) CNT, TMR 변수 사용

```

MAIN
VEL 100
CNT0=0 ..... 카운터 초기화
TMR1=-100 ..... 타이머 초기화
WHILE CNT0<20 ..... 입력 BIT "0"번에 펄스가 20개 입력될때가 LOOP
JMOV P0
IF(TMR1>0) THEN ..... 타이머가 1초 이상 이면
GOTO TEST (-100 ~> 0 : 1ms가 100개 발생)
ENDIF
DLAY 100
JMOV P1
ENDWL
LABL TEST
EOP
    
```

### 3.46.3 MVR 변수

- 기능 전체 이동거리에 대한 이동거리 비율
- 형식 MVR <전체 이동거리에 대한 이동거리의 백분율값>
- 설명
- 1) 두 Point(Pn, Pn+1)사이를 이동하는 경우 전체 이동거리에 대한 이동거리의 백분율값  
즉,  $MVR = (\text{현재 로봇 이동거리} / \text{로봇 전체이동거리}) * 100$
  - 2) 로봇동작중 조건분기 및 I/O병렬 처리에 사용.
  - 3) 이동구간을 100단위로 나누어 반드시 WITH문과 병행하여 사용.
  - 4) 두 Point (Pn, Pn+1)의 이동거리가 짧고 고속으로 운전할 때는 MVR값 설정에 유의 바랍니다.

#### ! CAUTION

▶ Pn Point에서 Pn+1 Point로 이동 중 90% 이후에 입력 "IN1"를 확인하는 경우

- . 로봇이 Pn+1위치의 90%정도 도달하였을 때는 다음 이동 POINT( Pn+2 )로 이동을 위한 연산을 시작합니다.

- . 즉,위의 그림과 같이 로봇이 90%위치에 있어도 **현재 #1 위치를 연산중**입니다.

★ 따라서 이동거리가 짧고 고속으로 운전할 때에는 Pn출발 후 90% 이후에는 입력 Bit'0'의 상태를 확인할 수 없는 경우도 발생합니다.  
이 경우 로봇 이동 속도를 낮추어 사용해 보십시오.

#### ! CAUTION

▶ 이동 중 신호가 입력 되었을 때 현재좌표 READ(HERE 명령어)기능 사용

- . 로봇 이동중 신호가 입력되었을 때 정지 후 현재좌표 READ기능은 로봇이동 속도에 따라 현재좌표 READ 한 값에 편차가 발생할 수 있습니다.

- . 이 경우 로봇의 이동 속도를 낮추어 사용하십시오.

### 3.46.4 프로그램 사용 예제

- 1) 이동중 입력Bit 상태에 따라 이동 궤적 수정.

```

MAIN
VEL 100
FOS 10
JMOV P100
MVR=0
WITH
JMOV P0
WHILE MVR<100 ..... P0포인트를 100% 이동중
IF MVR>10 THEN ..... 이동거리가 10% 위치에서
OUT1=1 50 ..... 출력을 0.5초 동안 ON(1)으로 유지
ENDIF
IN1=1
GOTO L1
ENDWL
LABL L1
STOP
ENDWT
JMOV P1
EOP
    
```

- 2) P1 포인트 출발하여 80%이후에 입력 "IN0"의 신호입력에 따라 P2 포인트로 이동

```

MAIN
VEL 100
WHILE IN1==1
PLUP 10
JMOV P0
MVR=0
WITH
JMOV P1
WHILE MVR<100 ..... P1포인트를 100% 이동중
IF (MVR>80)&&(IN0==1) ..... 이동거리가 80% 이고, IN0가 ON(1)된 경우
THEN
PLUP 0
JMOV P22
GOTO BB
ENDIF
ENDWL
ENDWT
IN0=1
PLUP 0
JMOV P2
LABL BB
ENDWT
OUT1=1 50
ENDWL
EOP
    
```

### 3.46.5 HERE 변수

기능	현재위치 값을 저장하는 변수
형식	HERE HERE<정수>
용어	<정수> : 채널 번호 (1 또는 2) -. 채널간 로봇의 위치를 읽어올 경우 사용 합니다.
설명	1) 로봇의 현재 위치 값을 각도값 또는 리드값 으로 저장 2) 다른 채널에 있는 로봇의 현재 위치를 저장.

### 3.46.6 프로그램 사용 예제

- 1) HERE 변수를 활용 하여 현재 위치를 READ

```

MAIN
VEL 100
POS CURR
JMOV P0
MVR=0
WITH
JMOV P1
WHILE MVR<100
IF MVR>50 THEN
CURR=HERE           .....      P1 포인트로 이동 중 50% 지점에서
                               현재 위치값을 위치형 변수 CURR에 저장
GOTO LLL
ENDIF
ENDWL
LABL LLL
ENDWT
JMOV P2
JMOV CURR
EOP
    
```

- 2) HERE<n> 변수 활용 타 채널 로봇의 현재 위치를 READ

```

MAIN
VEL 100
POS AP
JMOV P0
WHILE AP.1<10.0
AP=HERE2           .....      2번 채널 로봇의 현재 위치를 저장
DLAY 10
ENDWL
OUT0=1
JMOV P1
EOP
    
```

### 3.47 RSTATE (로봇 상태 확인 명령어)

기능     로봇의 상태를 읽어오는 명령어

형식     <정수형변수> =RSTATE(채널, STATE INDEX)

설명

INDEX	STATE NAME	내용
0	ALL	상태 정보 전체를 반환
1	ALARM	알람 상태 반환
2	READY	구동 준비 상태 반환
3	ORIGIN	원점 완료 상태 반환 (ABS Type 일경우 항상 ON)
4	INPOSITION	목표위치 도달 완료
5	RUN(JOB 구동중)	JOB RUNNING 상태 반환
6	SERVO ON	SERVO MOTOR ON/OFF 상태 반환
7	STOP(JOB 정지중)	JOB STOP 상태 반환
8	RUN MODE	Manual Run mode
9	SYSTEM MODE	System mode
10	TEACH MODE	Point Teaching mode
11	EMG	비상 정지 발생시 ON
12	PSEL	JOB LOAD 완료.
13	EECH	EECH 명령어 상태 반환
14	Reserved	예약된 영역 입니다.
15	MOVING(로봇 구동중)	해당 채널의 Robot 이동 중 상태 반환
16	TIMEOUT	String command Time out 상태 반환
17	1 AXIS MOVING FLAG	해당 채널의 첫번째 축 이동 중 상태 반환
18	2 AXIS MOVING FLAG	해당 채널의 두번째 축 이동 중 상태 반환
19	3 AXIS MOVING FLAG	해당 채널의 세번째 축 이동 중 상태 반환
20	4 AXIS MOVING FLAG	해당 채널의 네번째 축 이동 중 상태 반환
21	5 AXIS MOVING FLAG	해당 채널의 다섯번째 축 이동 중 상태 반환
22	6 AXIS MOVING FLAG	해당 채널의 여섯번째 축 이동 중 상태 반환
23	Reserved	예약된 영역 입니다.
.....	.....	
31	Reserved	

3.47.1 프로그램 사용 예제

1) CH1의 상태를 모두 확인 하여 I/O로 개별 출력

```

MAIN
INT TEMP
TEMP=RSTATE(1,0) ..... CH1의 상태 전체 확인
OUT10=(TEMP>>5)&0H01 ..... SVON 상태 출력
OUT9=(TEMP>>4)&0H01 ..... JOB RUN 상태 출력
OUT8=TEMP&0H01 ..... Alarm 상태 출력
EOP
    
```

2) CH2의 상태를 개별로 읽어 I/O로 출력

```

MAIN
OUT10= RSTATE(2,6) ..... CH2 SVON 상태 출력
OUT9= RSTATE(2,5) ..... CH2 JOB RUN 상태 출력
OUT8= RSTATE(2,1) ..... CH2 Alarm 상태 출력
EOP
    
```



### 3.48 RERROR (알람 코드 확인 명령어)

기능     로봇의 알람 코드를 읽어오는 명령어


형식     <정수형변수> =RERROR(PAGE NO, INDEX)

용어     PAGE NO : 알람 이력 번호

- Alarm History의 이력 번호이며, 하나의 Page에 10개의 Index를 저장합니다.
- Page 번호가 1번이면 가장 최근에 발생한 이력 이며, 10번이 가장 오래된 이력 번호 입니다.

INDEX: : 동시 발생된 알람 테이블 번호

- 동시에 발생할수 있는 Alarm Code는 최대 10개까지 입니다.
- Alarm code가 없을 경우에는 0을 반환 합니다.



#### 3.48.1 프로그램 사용 예제

1) Alarm Code Read

```

MAIN
INT ERR,ERR1
WHILE 1
ERR=RERROR(1,1)           .....      Page1 의 첫번째 Alarm code 확인
IF ERR != 0 THEN         .....      Alarm Code가 있으면
ERR1=RERROR(1,2)       .....      Page1 의 두번째 Alarm code 확인
ENDIF
ENDWL
EOP
    
```

### 3.49 GFTOGP (GF를 GP에 저장하는 명령어)

기능 GF(Global Float)의 값을 GP(Global Point) 변수에 저장하는 명령어

형식 GFTOGP <Global Float index> <Global Point index>

용어 Global Float Index : 0~499  
Global Point Index : 0~1023

	RO(Robot)	TR(Transfer Robot)
Global FLOAT	500EA(0~499)	2000EA(0~1999)
Global Point	1024EA(0~1023)	15000EA(0~14999)

설명 1) 저장된 GF 변수의 시작 번호와 저장할 GP 번호를 지정하면 시작 번호로부터 6개의 GF Data가 지정한 GP에 저장됩니다.

#### 3.49.1 프로그램 사용 예제

1) 로봇 현재 위치를 GP(Global Point)변수에 저장

MAIN		
POS CURR		
<b><u>CURR=HERE</u></b>	.....	로봇의 현재 위치 좌표 Read
<b><u>F10=CURR.1</u></b>	.....	로봇 첫번째축 좌표 F10번에 저장
<b><u>F11=CURR.2</u></b>	.....	로봇 두번째축 좌표 F11번에 저장
<b><u>F12=CURR.3</u></b>	.....	로봇 세번째축 좌표 F12번에 저장
<b><u>F13=CURR.4</u></b>	.....	로봇 네번째축 좌표 F13번에 저장
<b><u>F14=CURR.5</u></b>	.....	로봇 다섯번째축 좌표 F14번에 저장
<b><u>F15=CURR.6</u></b>	.....	로봇 여섯번째축 좌표 F15번에 저장
<b><u>GFTOGP F10 GP50</u></b>	.....	F10~F15값을 GP50번에 저장
EOP		

### 3.50 REMCMD (SYSTEM COMMAND)

- 기능 System Command
- 형식 REMCMD <채널> <Command>
- 용어 채널: Command를 수행할 채널  
Command :

command	명칭	설명
1	Servo ON	해당 채널의 로봇을 Servo ON 시킨다.
2	Servo OFF	해당 채널의 로봇을 Servo OFF 시킨다.
3	PSEL	선택된 JOB 프로그램 LOAD 시킨다.
4	START	선택된 JOB을 구동시킨다.
5	STOP	구동중인 JOB을 정지시킨다.
6	RESET	알람을 클리어 시킨다.
7	EMG STOP	로봇을 비상정지 시킨다.

#### 3.50.1 프로그램 사용 예제

##### 1) CH1 JOB 구동과 로봇 구동 Power ON

```

MAIN
INT CHECK
REMCMD 1 3 ..... CH1에 JOB 파일을 LOAD 한다.
CHECK=RSTATE(1,12) ..... 선택된 JOB LOAD 완료 정보 확인
IF CHECK==1 THEN
REMCMD 1 1 ..... CH1의 로봇을 Servo ON 시킴
CHECK=RSTATE(1,6) ..... CH1의 SERVO ON/OFF 상태 정보 확인
IF CHECK==1 THEN
REMCMD 1 4 ..... CH1의 JOB을 RUN 시킨다.
ENDIF
ENDIF
EOP
    
```

### 3.51 상수

- 기능 16진수, 2진수 표시.
- 형식 0H<숫자>  
0B<숫자>
- 용어 <숫자> : 숫자 앞에 0H 또는 0B가 없으면 10진수  
\* 0H : 16진수  
\* 0B : 2진수
- 설명 1) 0H<숫자>는 16진수이며, 숫자는 8자리까지 가능합니다.  
2) 0B<숫자>는 2진수이며, 숫자는 16자리(16비트)까지 가능합니다.  
3) 0H나 0B가 없으면, 10진 정수 또는 실수입니다.

#### 3.51.1 프로그램 사용 예제

MAIN	
INT AA	
VEL 100	
WHILE 1	
AA=PIN0 & 0H00FF	입력포트0번의 값을 16진수 00FF 와 비교
IF AA==1 THEN	
POUT0=0B00110100	출력포트0번에 2진수 00110100 출력
JCALL MD1	
ENDIF	
IF AA==1 THEN	
POUT0=0B01110100	출력포트0번에 2진수 01110100 출력
JCALL MD2	
ENDIF	
ENDWL	
EOP	

### 3.52 연산자

#### 3.52.1 배정 연산자

명령어	기능
=	오른쪽 수식을 평가하여 그 결과를 왼쪽변수의 변수형에 맞게 변환하여 배정합니다. (예: IN0=1)

	정수형	실수형	위치형	카운터형	타이머형
정수형	정수형	정수형	불가	정수형	정수형
실수형	실수형	실수형	불가	실수형	실수형
위치형	불가	불가	위치형	불가	불가
카운터형	정수형	정수형	불가	정수형	정수형
타이머형	정수형	정수형	불가	정수형	정수형

3.52.2 산술 연산자

명령어	기능
*,/,+,-,%	서로 다른 자료형에 대한 각 연산자의 연산결과는 다음표와 같습니다. 표에서 '불가'는 연산이 이루어질수 없는 수식의 경우를 말합니다.

▶ +, - 연산자

	정수형	실수형	위치형	카운터형	타이머형
정수형	정수형	실수형	불가	정수형	정수형
실수형	실수형	실수형	불가	실수형	실수형
위치형	불가	불가	위치형	불가	불가
카운터형	정수형	정수형	불가	정수형	정수형
타이머형	정수형	정수형	불가	정수형	정수형

▶ \* 연산자

	정수형	실수형	위치형	카운터형	타이머형
정수형	정수형	실수형	위치형	정수형	정수형
실수형	실수형	실수형	위치형	실수형	실수형
위치형	위치형	위치형	불가	위치형	위치형
카운터형	정수형	실수형	위치형	정수형	정수형
타이머형	정수형	실수형	위치형	정수형	정수형

▶ / 연산자

	정수형	실수형	위치형	카운터형	타이머형
정수형	정수형	실수형	불가	정수형	정수형
실수형	실수형	실수형	불가	실수형	실수형
위치형	위치형	위치형	불가	위치형	위치형
카운터형	정수형	실수형	불가	정수형	정수형
타이머형	정수형	실수형	불가	정수형	정수형

▶ % 연산자

정수간 연산만이 가능하며 나눗셈 연산에서 몫은 버리고 나머지를 결과 값으로 갖습니다.  
예) AA=10%3 일경우 AA의 값은 1

### 3.52.3 관계 연산자

명령어	기능
>, <, ≤, ≥, ==, !=	-. 주로 정수형과 실수형 자료를 대상으로 하며 IF 문과 WHILE 문의 조건식에 사용됩니다. -. 관계 연산의 결과는 논리값 (true,false 또는 0/1)입니다

### 3.52.4 논리 연산자

명령어	기능
&&,   , !	논리연산자는 논리값만을 대상으로 합니다. 즉 관계연산의 결과를 피연산자로 사용합니다. 논리연산의 결과는 논리값 입니다.

### 3.52.5 비트 연산자

명령어	기능
&,	논리연산자
<<, >>	이동연산자
~	1의 보수

### 3.53 내장 함수

함 수	기 능	사 용 예
EXP(X)	지수 $e^x$	EXP(3) 은 20.085
LOG(X)	상용로그 $\log_{10}X$	LOG(100) 은 2
LN(X)	자연로그 $\log_e X$	LN(15)는 2.708
SQRT(X)	제곱근 $\sqrt{x}$	SQRT(16)은 4
POW(x,y)	$x^y$	POW(2,10)은 1024
ABS(K)	정수 K의 절대값	ABS(-12)는 12
RND(X)	실수 X를 반올림하여 정수값	RND(97.62)는 98
SIN(X)	sine, x 단위 : radian	SIN(RAD(30))는 0.5
COS(X)	cosine, x 단위 : radian	COS(0)는 1
TAN(X)	tangent, x 단위 : radian	TAN(RAD(45))는 1.0
ASIN(X)	arcsine, $-\pi/2 \sim \pi/2$ 의 값 반환	DEG(ASIN(0.5))는 30
ACOS(X)	arccosine, $0 \sim \pi$ 의 반환	DEG(ACOS(0.5))는 60
ATAN(X)	arctangent, $-\pi/2 \sim \pi/2$ 의 값 반환	DEG(ATAN(1.0))은 45
ATAN2(Y,X)	제 2 arctangent, $-\pi \sim \pi$ 의 값 반환	DEG(ATAN2(-1,-1))은 -135
DEG(X)	radian 을 angle 값으로 변환	DEG(3.1416)은 180.0
RAD(X)	angle 을 radian 값으로 변환	RAD(180.0)는 3.1416



### 3.54 ASC

기능 문자열 첫문자를 캐릭터 코드로 반환

형식 정수형 변수 = ASC(문자열)

용어 문자열변수

- DEFSTR을 사용하여 선언한 변수를 의미하며, 문자열을 다루기 위해 사용됩니다.

문자열

- 문자열 변수나 문자열 상수를 의미합니다. 문자열 상수는 " "로 둘러싸인 문자들의 집합입니다.
- EX) "ABCDEF"

캐릭터 코드

- 문자에 해당하는 숫자(정수)를 말합니다.  
Ex) 문자'A'는 정수(65)이고 문자 'B'는 정수(66)입니다.
- 아스키 코드표를 참조 하십시오.

설명 1) 입력된 문자열의 첫문자를 반환합니다.

#### 3.54.1 프로그램 사용 예제

```

MAIN
DEFSTR AA
INT BB
AA = "XYZW" ..... 문자열 변수 AA에 "XYZW"대입
BB = ASC(AA) ..... 문자열 변수 AA에서 첫번째 문자 "X"를 캐릭터코드(88)로 반환
EOP
    
```

### 3.55 CHR

기능 아스키 코드 범위의 정수를 문자로 변환

형식 문자열변수 = CHR(정수)

용어 문자열변수

- DEFSTR을 사용하여 선언한 변수를 의미하며, 문자열을 다루기 위해 사용합니다.

정수

- 아스키 코드값 내의 정수(0~127)

설명 1) 입력된 정수에 해당하는 아스키 코드를 반환합니다.

#### 3.55.1 프로그램 사용 예제

```

MAIN
DEFSTR AA
AA = CHR(65) ..... 문자열 변수 AA에 아스키코드 65에 해당하는 "A"대입
EOP
    
```

### 3.56 FLUSH

기능    입력, 출력 버퍼 클리어

형식    FLUSH PORT 정수

용어    PORT: COM Port를 선택(1~2)

- 1: RS-232 Port를 선택
- 2: RS-485 Port를 선택
- 3: 입력, 출력 버퍼 클리어

정수: 클리어 버퍼 선택(1~3)

- 1: 입력 버퍼 클리어
- 2: 출력 버퍼 클리어
- 3: 입력, 출력 버퍼 클리어

설명    1) 시리얼 통신 송수신에 사용하는 입출력 버퍼를 클리어 합니다.

#### 3.56.1 프로그램 사용 예제

```

MAIN
DEFSTR AA
AA = STRIN(1,1000) ..... Serial 통신으로 입력을 받아 문자열 AA에 저장
FLUSH 1 1 ..... RS-232 Port 입력 버퍼 클리어
EOP
    
```

### 3.57 FTOS

기능     정수나 실수를 문자열로 변환

형식     문자열변수 = FTOS(정수 or 실수)

용어     문자열변수

- DEFSTR을 사용하여 선언한 변수를 의미하며, 문자열을 다루기 위해 사용합니다.

정수 or 실수

- 문자열로 변환할 정수나 실수

설명     1) 정수나 실수 수치 데이터를 문자열로 변환합니다.

EX) FTOS(1234) → "1234"

#### 3.57.1 프로그램 사용 예제

```

MAIN
DEFSTR AA,BB
AA = FTOS(1234) ..... 1234를 문자열 "1234"로 변환후 문자열 변수 AA에 대입
BB = FTOS(-123.456) ..... -123.456를 문자열 "-123.456"로 변환후 문자열 변수 BB에 대입
EOP
    
```

## 3.58 HTOS

기능 정수를 16진수의 문자열로 변환

형식 문자열변수= HTOS(정수)

용어 문자열변수

- DEFSTR을 사용하여 선언한 변수를 의미하며, 문자열을 다루기 위해 사용합니다.

정수

- 16진수 문자열로 변환할 정수

설명 1) 입력된 정수를 16진수 문자열로 변환합니다.

## 3.58.1 프로그램 사용 예제

```

MAIN
DEFSTR AA
AA = HTOS(10) ..... 정수 10을 16진수 문자열("A")로 변환하여 문자열 AA에 대입
EOP

```

### 3.59 SLEFT

기능    입력된 문자열의 좌측부분 추출

형식    문자열 변수 = SLEFT(문자열, 정수)

용어    문자열변수

- DEFSTR을 사용하여 선언한 변수를 의미하며, 문자열을 다루기 위해 사용됩니다.

문자열

- 문자열 변수나 문자열 상수를 의미합니다.
- 문자열 상수는 " "로 둘러싸인 문자들의 집합입니다.

정수

- 추출할 문자의 수

설명    1) 입력된 문자열을 왼쪽에서부터 입력된 정수만큼 추출

#### 3.59.1 프로그램 사용 예제

```

MAIN
DEFSTR AA,BB
AA = "XYZW"           ..... 문자열 변수 AA에 "XYZW" 대입
BB = SLEFT(AA,2)     ..... 문자열 변수 AA의 왼쪽부터 2문자("XY")을 변환
EOP
    
```

### 3.60 SLEN

기능     입력된 문자열의 길이를 반환

형식     정수형변수=SLEN(문자열)

용어     문자열

- 문자열 변수나 문자열 상수를 의미합니다. 문자열 상수는 " "로 둘러싸인 문자들의 집합입니다.
- EX) "ABCDEF"

설명     1) 입력된 문자열의 길이를 반환합니다.

#### 3.60.1 프로그램 사용 예제

```

MAIN
DEFSTR AA
INT LL
AA = "XYZW"           ..... 문자열 변수 AA에 "XYZW"대입
LL = SLEN(AA)       ..... 문자열 변수 AA의 길이(4)를 정수형 변수 LL에 대입
EOP
    
```

### 3.61 SMID

기능 문자열 지정위치부터 자릿수 만큼 문자열 추출

형식 문자열 변수 = SMID(문자열, 지정위치, 정수)

용어 문자열변수

- DEFSTR을 사용하여 선언한 변수를 의미하며, 문자열을 다루기 위해 사용합니다.

문자열

- 문자열 변수나 문자열 상수를 의미합니다. 문자열 상수는 " "로 둘러싸인 문자들의 집합입니다.

지정위치

- 문자열 추출할 시작 위치, 문자열의 시작은 0입니다.

정수

- 지정위치부터 읽어올 문자 개수

설명 1) 지정위치에서부터 사용자 입력 정수만큼 문자를 추출합니다.

#### 3.61.1 프로그램 사용 예제

```

MAIN
DEFSTR AA, BB
AA = "ABCDEFGF" ..... 문자열 변수 AA에 "ABCDEFGF"대입
BB = SMID(AA,2,3) ..... 문자열 변수 AA의 2번째위치부터 문자 3개("CDE")반환
EOP
    
```



### 3.62 SPOS

기능 문자열1에서 문자열2가 매칭되는 시작위치 반환

형식 정수형변수 = SPOS(문자열1, 문자열2)

용어 문자열

- 문자열 변수나 문자열 상수를 의미합니다. 문자열 상수는 " "로 둘러싸인 문자들의 집합입니다.
- EX) "ABCDEF"

설명 1) 문자열1에서 문자열2와 매칭되는 시작위치를 반환합니다.

#### 3.62.1 프로그램 사용 예제

```

MAIN
DEFSTR AA, BB
INT PP
AA = "XYZW"           ..... 문자열 변수 AA에 "XYZW" 대입
BB = "Z"             ..... 문자열 변수 BB에 "Z" 대입
PP = SPOS(AA, BB)    ..... 문자열 변수 AA에서 BB와 일치하는 위치 반환
EOP
    
```

### 3.63 SRIGHT

기능    입력된 문자열의 우측부분 추출

형식    문자열 변수 = SRIGHT(문자열, 정수)

용어    문자열변수

- DEFSTR을 사용하여 선언한 변수를 의미하며, 문자열을 다루기 위해 사용됩니다.

문자열

- 문자열 변수나 문자열 상수를 의미합니다. 문자열 상수는 " "로 둘러싸인 문자들의 집합입니다.

정수

- 추출할 문자의 수

설명    1) 입력된 문자열을 오른쪽에서부터 입력된 정수만큼 추출

#### 3.63.1 프로그램 사용 예제

```

MAIN
DEFSTR  AA, BB
AA = "ABCDEF"           ..... 문자열 변수 AA에 "ABCDEF" 대입
BB = SRIGHT(AA,2)      ..... 문자열 변수 AA의 오른쪽 두문자를("EF")반환
EOP
    
```

### 3.64 STRIN

기능 시리얼통신을 통하여 들어온 문자열을 읽어옴

형식 문자열 변수 = STRIN(PORT,정수)

용어 문자열변수

- DEFSTR을 사용하여 선언한 변수를 의미하며, 문자열을 다루기 위해 사용합니다.

PORT: COM Port를 선택(1~2)

- 1: RS-232 Port 선택
- 2: RS-485 Port 선택

정수

- 타임아웃시간.(단위: ms)
- 타임아웃시간만큼 시리얼 입력을 기다립니다. 타임아웃시간동안 데이터가 들어오지 않으면, 시스템 영역에 타임아웃 상태를 체크하고 다음 스텝명령을 수행합니다

설명 1) 사용자가 지정한 시간만큼 대기하면서 시리얼 통신을 통한 입력을 읽어옵니다.

#### 3.64.1 프로그램 사용 예제

```

MAIN
DEFSTR PACKET
INT LEN, OVERTM
PACKET = STRIN(1,1000) ..... 시리얼 통신을 통한 문자열 입력(입력 없을시 1s 대기)
LEN = SLEN(PACKET) ..... 입력받은 문자열 길이 저장
SS1=SYS1
OVERTM=(SS1>12) &0H01 ..... TIMEOUT 유무를 읽어옴
IF OVERTM==0&&LEN>0 THEN ..... TIMEOUT이 아니고, 입력받은 문자열이 존재한 경우
.....
ENDIF
EOP
    
```

### 3.65 STROUT

기능 시리얼 통신을 통하여 문자열 출력  
 형식 정수형변수 = STROUT(PORT,문자열)

용어 PORT: COM Port를 선택(1~2)  
 - 1: RS-232 Port를 선택  
 - 2: RS-485 Port를 선택

문자열

- 문자열 변수나 문자열 상수를 의미합니다. 문자열 상수는 " "로 둘러싸인 문자들의 집합입니다.

설명 1) 입력한 문자열을 시리얼 통신을 통해 출력을 내보내며, 전송하지 못한 문자수 반환  
 EX) "ABCDEFGH"의 문자를 전송하는 경우 성공시 반환값은 0  
 전부 전송하지 못한 경우 반환값은 8

#### 3.65.1 프로그램 사용 예제

```

MAIN
DEFSTR SENDPK
INT STRRET
SENDPK="ABCDEF" ..... SENDPK 문자열 변수에 "ABCDEF" 대입
STRRET=STROUT(1,SENDPK) ..... 시리얼 통신포트로 문자열("ABCDEF") 출력
IF STRRET !=0 THEN ..... 전송 실패 확인
.....
ENDIF
EOP
    
```

## 3.66 SVAL

기능 문자열을 숫자로 변환

형식 (정수형변수 or 실수형변수) = SVAL(문자열)

용어 문자열

- 문자열 변수나 문자열 상수를 의미합니다.
- 문자열 상수는 " "로 둘러싸인 문자들의 집합입니다.

설명 1) 입력한 문자열을 정수 또는 실수 데이터로 변환합니다.

## 3.66.1 프로그램 사용 예제

```

MAIN
INT AI
REAL BR
AI=SVAL("1234")           ..... 문자열 "1234"를 정수 1234로 변환
BR=SVAL("12.234")        ..... 문자열 "12.234"를 실수 12.234로 변환
EOP

```

### 3.67 VCMD

- 기능     Vision point 및 관련 변수를 초기화 합니다.
- 형식     VCMD index 작업종류
  
- 용어     Index: 0~9까지 입력 합니다.  
           입력된 Index는 Serial Port로 전송합니다.
- 작업종류: 1 또는 2를 입력 합니다.  
           Serial Port로 전송할 DATA를 선택 합니다.
  
- 설명     Serial Port로 전송되는 DATA format은 다음과 같습니다.  
           작업종류가 1인 경우: STX, 0xFF, 'V' 'Q' Index ETX LRC  
           작업종류가 2인 경우: STX, 0xFF, 'V' 'R' Index ETX LRC

#### 3.67.1 프로그램 사용 예제

```

MAIN
VCMD 5 2                                   ..... Serial Port로 STX 0xFF 'V' 'R' '5' ETX LRC를 전송합니다.
EOP
    
```

### 3.68 VVAL

기능 Vision Point로 받은 DATA를 반환 합니다.

형식 정수형 변수=VVAL

용어 정수형 변수: INT로 선언된 변수명입니다.

설명 Serial CMD로 받은 DATA를 반환합니다.  
 "VA","VB","VM","VN" 프로토콜 명령어를 참조 하십시오  
 (N1\_옵선 로보스타 프로토콜 설명서 참조)

#### 3.68.1 프로그램 사용 예제

```

MAIN
JMOV P1000 ..... 카메라 위치로 이동
CALL H_ON ..... 판정 요청
IF VVAL != 0 THEN ..... 판정 값을 반환 합니다.
CALL ERROR ..... 불량 서브루틴 호출
ELSE
CALL GOOD ..... 양품 서브루틴 호출
ENDIF
EOP
SUBR ERROR
JMOV P100 ..... 불량 위치 이동
RET
SUBR GOOD
JMOV P200 ..... 양품 위치 이동
RET
    
```

### 3.69 VDAT

기능 Vision CMD 데이터 수신을 지정한 시간만큼 대기

형식 VDAT 대기시간

용어 대기시간: 정수형 DATA 입니다.  
단위는 10ms 입니다.

설명 Serial CMD DATA가 들어 올 때까지 대기 합니다.  
"VA","VB","VM","VN" 프로토콜 명령어를 참조 하십시오  
(N1\_옵선 로보스타 프로토콜 설명서 참조)

#### 3.69.1 프로그램 사용 예제

```

MAIN
JMOV P1000 ..... 카메라 위치로 이동
CALL H_ON ..... 판정 요청
VDAT 100 ..... 1초 동안 Vision CMD 수신 대기
IF VVAL != 0 THEN
CALL ERROR ..... 불량 서브루틴 호출
ELSE
CALL GOOD ..... 양품 서브루틴 호출
ENDIF
EOP
SUBR H_ON
OUT2=1 ..... 작업물 판정 요청
RET
SUBR ERROR
JMOV P100 ..... 불량 위치 이동
RET
SUBR GOOD
JMOV P200 ..... 양품 위치 이동
RET
    
```



### 3.70 VCNT

기능      전송 받은 Vision Point Data를 개수를 반환 합니다.

형 식      정수형 변수=VCNT

용 어      정수형 변수: INT로 선언된 변수명입니다.

설 명      Serial CMD DATA가 들어 올 때까지 대기 합니다.  
 "VA","VM","VN" 프로토콜 명령어를 참조 하십시오  
 (N1\_옵선 로보스타 프로토콜 설명서 참조)

#### 3.70.1 프로그램 사용 예제

```

MAIN
INT COUNT
JMOV P1000           .....   카메라 위치로 이동
CALL H_ON           .....   판정 요청
COUNT=VCNT         .....   Vision Point Data 반환
IF COUNT>0 THEN
JMOV P2000           .....   Vision Point로 이동
ELSE
JMOV P10             .....   대기 위치 이동
ENDIF
EOP
SUBR H_ON
OUT1=1              .....   요청 신호 출력
RET
    
```

Rev.	수정일자	내용	수정자	S/W Version
V.1	2012.07.30	초판 인쇄		
V.2	2012.12.31	내용 추가 및 수정	Kimjs	

N1 ROBOT CONTROLLER

---

# CONTROLLER MANUAL

FIRST EDITION JULY 2012

ROBOSTAR CO, LTD

ROBOT R&D CENTER

---